
第 1 章 イントロダクション 

制約計画法（Constraint Programming）は、離散変数の非線形制約式で表現される関係に対処するの

に、特にうまく機能する問題解決のアプローチです。そして、以下では、しばしば、「有限ドメイン制約

計画法（finite domain Constraint Programming）」と同じ意味で「制約計画法」、もしくは、「CP」を'使用

します。 

CP は、過去において、「(資源制約式のある、もしくは、ない)生産スケジュール作成問題」、「タスクの

順序付け、および、割当て問題（sequencing and assignment of tasks）」、「従業員計画と予定表問題

（workforce planning and timetabling）」、「周波数割当問題（frequency assignment）」、「積込、カット問

題（loading and cutting）*」、「グラフの塗りわけ問題（graph coloring）」のような多様な問題の解法とし

て成功裏に適用なれてきています。 

CP の強みは、それを、制約関係（constraint relations）の持つオリジナルな意味を保つことので

きる制約式の記述のために、ハイレベルな意味論を使うことにあります。このようなハイレベルの制約

式は「グローバル制約式（global constraints）と呼ばれます。 制約式を、数式の形に変換すること

は必要ありません。よくあることですが、数式の形に変換する過程で、問題の構造の多くが失われて

しまいます。制約式に内在する、固有な問題に関する知識は、ソリューションアルゴリズムで利用され、

アルゴリズムの効率をよくします。 

1.1 Xpress-Kalis 

「Xpress-Kalis」、別称、「Kalis for Mosel」により、Artelys 社の Kalis Constraint Programming ソ

ルバーへのアクセスができます。Xpress-Kalis を通して、Kalis の制約計画の機能が、Mosel 環境で

利用可能になり、ユーザは、Mosel 言語で CP モデルを定式化して、それを解くことが出来ます。 

Xpress-Kalisは、「有限ドメインソルバー」と「連続(浮動小数点)変数のソルバー」を結合します。スケジ

ュール作成問題の定式化を支援するために、このソフトウェアは、タスクと資源を表す特定のモデル

作成オブジェクトの集合体を定義します。この特定のモデル作成オブジェクトの集合体は、自動的に、

ある(暗黙的な)制約関係をセットアップして、このタイプの問題に専門化された、内蔵されているサー

チ戦略をトリガーします。こうして、標準のスケジュール作成問題は、単に、対応するタスク、および、

資源のオブジェクトをセットアップすることだけで、定義され、解くことができます。 

Mosel IO ドライバーを使うことで、メモリ内でのデータ転送、モジュール mmodbc を通してのデータベー

スへの ODBC アクセスを含む、「Mosel 環境のすべてのデータハンドリング・ファシリティ」は、kalisに

より、なんら、余分な努力なしで使えます。 

                                                   
* 参照：http://www.dashoptimization.com/home/cgi-bin/example.pl?id=mosel_book_4 

Mosel 言語は、ループ、また、より複雑なアルゴリズムを実行するのに必要なサブルーチンなどのよう

な、典型的なプログラミングのための構成要素をサポートしています。Mosel は、また、Xpress-Kalis を

Xpress-Optimizer に結合し、CP と LP/MIP を使って、共同で問題を解くことが出来るようにするための、

異なったソルバーを結合するプラットホームとしても使用できます。(Dash Whitepaper Multiple models 

and parallel solving with Mosel を参照。）このマニュアルでは、Mosel を使って行うモデル作成とプログ

ラミングの基礎について説明しますが、必要な場合、さらに高度なフィーチャーについても触れます。

その使用法についての完全なドキュメンテーション、より徹底的なイントロダクション論については、

Mosel language reference manual、および､ Mosel user guide を参照してください。 

1.1.1 製品バージョンについてのノート 

このマニュアルに記載の例は、「Xpress-Kalis の release 2007.1.0」、「Xpress-MP Release 2007A 

beta version of Mosel (1.7.9)」、および､「version 1.17.50 of Xpress-IVE」を使って書か

れています。これらの例が、他のバージョンでランされると、得られるアウトプットは、こと

なる形式で出力されるかもしれません。特に、その後の CP ソルバーのアルゴリズムの改善、

Xpress-Kalis のディフォルト設定の変更により、制約式、サーチの動きは影響されるかもしれ

ません。また、IVE インタフェースは、新機能が加えられて行くと、今後のリリースで、少し、

変わって行くかもしれません。 

1.2 ソフトウェアのインストレーション 

Xpress-Kalis、Xpress-Mosel、Xpress-Kalis を使うには、これらのソフトウェアをインストールして、ライ

センスを得なければなりません。CP モデルを扱う場合、Windows ユーザーは、グラフィカル環境

Xpress-IVE を追加インストールすると便利ですが、これは必須条件ではありません。 

Mosel (および､IVE)をコンピュータにインストールするには、Xpress-MP の配布のときに、同時に配布

されるインストールガイドに従ってください。次いで、Xpress-Kalisを、Xpress-Kalisの配布のときに、同

時に配布されるインストラクションに従って、インストールしてください。 

1.3 制約計画法（Constraint Programming）の基本概念 

制約計画法(CP)の問題は、その決定変数によって定義され、また、これらの決定変数はドメインと制

約式を持っています。問題定義は、通常、分岐戦略（branching strategy）によって終わります。分岐戦

略は、列挙戦略（enumeration strategy）、サーチ戦略（search strategy）とも呼ばれています。 

CP は、可変ドメイン（variable domain）の概念、すなわち、「決定変数が取る値の集合」を使います。

「有限ドメイン制約計画法（finite domain Constraint Programming）」では、これらは、整数の集合（set）、

または、インターバル（interval）です。 

CP の各制約式は、通常、例えば､、グラフ理論などのような、他の領域の計算結果（result）に基づく、

それ自身のソリューションアルゴリズムを備えています。ひとたび、制約式が確定されると、それは、

その変数集合を解かれた状態、すなわち、ソリューションアルゴリズムが、変数のドメインから、実行

不可能と判定した値も取り除いた状態で保持します。 

CP 問題の制約式は、「制約式プロパゲーション（constraint propagation）」と呼ばれるメカニズムによ

ってリンクされています。すなわち、ある変数のドメインが変更されると、これは、この変数についての

すべての制約式の再評価をトリガーします。これにより、今度は、他の変数の変化を惹起したり、

Figure1.1 の例に示されているように、最初の変数のドメインの一層の減少をトリガーしたりします。(変

数の元々のドメインは、2 つの制約式の追加で減少します。 最後のステップで、2 番目の制約式の影

響は、最初の制約式に伝播され、その再評価をトリガーします。) 

 (a) (b) 

  

Figure 1.1: Example of constraint propagation, (a) finite domain (discrete) variables, (b) continuous variables. 

CP 問題は、変数に、制約式と境界（bound）を加えることによって、徐々に、築きあげられてゆきます。

CP 問題を解くことは、最初の制約式のステートメントを書くことから始まります。そして、制約式の関係

を犯す値は、それに関る変数のドメインから取り除かれてゆきます。新たに加えられた制約式の結果

は、すぐに全体の CP 問題に伝播されるので、一般に、問題からこの制約式を、後で変更したり、削除

したりすることはできません。 

場合によっては、制約式を解くこと、および、伝播のメカニズムの組み合わせだけで、ある問題インス

タンス（problem instance）が実行不能であることを証明するの十分であることもあります。しかし、大抵

の場合、変数のドメインを、ただ一つの値（整合性のあるインスタンス化や実行可能ソリューション）に

まで減少させるための列挙（enumeration）を加えたり、そのような解が存在しないことを証明したりす

ることが必要です。さらに、最も良い目的関数の値(最適解)を持つ実現可能解をサーチするために、

目的関数（もしくは、コスト関数）やサーチを定義することもできます。 

1.4 この文書の内容 

このドキュメントは、Xpress-Kalis の使い方のイントロダクションです。制約計画法の基本的な概念も

説明されますが、読者の皆さんが、CP のテクニックについての、何らかの一般的な理解があることを

前提としています。また、Xpress-Kalis が Xpress-Mosel のモジュールなので、このドキュメントでは、

提示される例の理解に必要であると思われる場合は、Mosel 言語のフィーチャーについても説明して

います。 

モデル例によって、Xpress-Kalis で定義される以下の新しい「タイプ」の使い方を例示します。 

 

· 決定変数： 有限ドメイン（finite domain）変数と浮動小数点（floating point）変数 

· 制約式： absolute value and distance、all-different、element、generic binary、linear、

maximum と minimum、occurrence、logical relations 

· 列挙（Enumeration）： 事前に定義された分岐スキームとユーザサーチ戦略 

· スケジュール作成： タスクと資源を表す統合モデルオブジェクト 

ここでは、CP モデルの一般構造、および、データハンドリングを含む、Mosel の使い方の基礎を説明し

ます。次いで、Xpress-Kalis の異なる制約関係タイプの使い方を例示するために、一連の小さい問題

が続きます。次の章は、もっと体系だった方法で、列挙戦略を定義するための、可能な、複数の方法

について説明しますが、それらのいくつかは、その前で説明するモデル例で、既に、出てきているもの

です。スケジュール作成についての章では、スケジュール作成問題の定式化を易しくする「task」と

「resource」というモデル作成のためのオブジェクトを導入します。 

最初の方で出てくるいくつかの例は別として、すべての例が、以下のように示されます。 

 

· 例についての説明 

· CP モデルとしての定式化 

· Implementation with Xpress-Kalis: code listing and explanations インプリメンテーションの

ための Xpress-Kalis コードのリストと説明 

· 結果 

このドキュメントに示されているモデル例は、すべて、Xpress-Kalis の配布時に、配布物の一部として

提供される「例題集」の一部として含まれています。 

第 2 章 モデル作成の基礎 

本章では、以下のことを説明します。 

 

· Mosel を使う 

· Xpress-Kalis で、簡単な CP モデルを作成し、解く 

· このソフトウェアが出力するアウトプットを理解し、分析する 

· データハンドリングで、モデルを拡張する 

·  目的関数を定義する 

· ディフォルトの分岐戦略（branching strategy）を修正する 

2.1 最初のモデル 

次の問題について考えてみてください。1 から 3 という番号を付けられている「3 つの時間帯」に、4 つ

の会議 A、B、C、および、D をスケジュールしたいと考えています。いくつかの会議には、同じ人物の

出席が必要です。このことは、これらの会議を同じ時間に開催できないことを意味しています。例えば､

会議 A は、会議 B、会議 D と同時に開催することができず、会議 B は、会議 C、会議 D と同じ時間帯

に開催できません。 

これを、より厳密に表現すれば、以下のようになります。ここで、(mÎMEETINGS ={ }DCBA ,,, ) は、

会議 m の時間帯を示し、これらは、この問題の決定変数です。 

 

{ }3.2,1: ÎÎ" mplanMEETINGSm  

BA planplan ¹  

DA planplan ¹  

CB planplan ¹  

DB planplan ¹  

 

2.1.1 Mosel を使って行うインプリメンテーション 

以下のコードリストにより、上で説明した問題が、実行され、解かれます。 

model "Meeting" 

 uses "kalis" 

  

 declarations 

  MEETINGS = {'A','B','C','D'}        ! Set of meetings 

  TIME = 1..3                         ! Set of time slots    

  plan: array(MEETINGS) of cpvar      ! Time slot per meeting 

 end-declarations 

 forall(m in MEETINGS) setdomain(plan(m), TIME) 

 

! Respect incompatibilities 

 plan('A') <> plan('B') 

 plan('A') <> plan('D') 

 plan('B') <> plan('C') 

 plan('B') <> plan('D')   

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

 

! Solution printing 

 forall(m in MEETINGS) 

  writeln("Meeting ", m, ": ", getsol(plan(m))) 

 

end-model 

 

このMoselモデルは、meeting.mosという名前で、テキストファイルとしてセーブされます。 以下で、上

に書いたモデルを詳しく見てみましょう。 

 

2.1.1.1 一般的な構造 

Mosel プログラムは、すべて、model というキーワードから始まります。続いて、ユーザが選んだ model

名が続きます。そして、Mosel プログラムは、end-model というキーワードで終わります。 

Mosel は、それ自体はソルバーではないので、プログラムの冒頭で、下記のステートメントで、Kalis 

constraint solver を使うことを指定します。 

 uses "kalis" 

すべてのオブジェクトは、assignment で明確に定義されている場合を除き、declarations セクション

でで宣言されなければなりません。例えば、i: =1 は i を整数と定義し、それに値 1 を割り当てます。

モデルには、異なる場所に、このような declarations セクションがいくつかあるかもしれません。 

現在、見ているこのケースでは、以下の二つの集合（set）と配列（array ）を定義しています。 

· MEETINGS は、strings の集合. 

· TIME は、いわゆる、range set で、連続する整数の集合で、この例でえば､言えば、1 から 3

までの整数です。 

· plan は、集合 MEETINGS というインデックスを付されたタイプ cpvar という決定変数の配列

です。ここに、cpvarは有限ドメイン CP 変数です（Xpress-Kalis の二つ目の決定変数のタイ

プは cpfloatvar で、これは、連続変数のためのものです）。 

次いで、モデルは、Xpress-Kalis の手順（procedure）setdomain を使って変数のドメインを定義します。

決定変数は、実際、それらの declaration により、デフォルトの大きなドメインを与えられて生成され、

setdomainは、これらのドメインを、表示値に従って、デフォルトドメインの intersection まで減少させま

す。数学モデルと同じように、forall loop を使い、集合 MEETINGS の中で、すべてのインデックスを列

挙（enumerate）します。 

この後に、制約式のステートメントが続きます。今、ここで見ているモデルでは、4 つの（ ¹ を持つ）

disequality 制約式が制約式のステートメントです。 

 

2.1.1.2 モデルを解き、アウトプットを出力する 

cp_find_next_sol ファンクションで、Kalis を呼び、問題を解きます(決定変数のすべてにたいして、実

行可能な値の割当てを見つける)。ここで、このファンクションの return value をテストします。もし、ソリ

ューションが全く見つけられないと、このファンクションは、false を返してきますので、この時点で、

Mosel 手順 exit を呼び、モデルの実行を止め、そうでない場合は、ソリューションを印刷します。 

問題を解くために、Xpress-Kalis は、内蔵するデフォルト検索戦略を使います。後で、これらの戦略を

変更する方法を説明します。 

ファンクション getsol を使い、CP 変数のソリューションを得ます。一行にいくつかの項目を書くには、

出力を印刷するための writeln ではなく、write を使います。 

 

2.1.1.3 フォーマッティング 

モデルを書くときに、インデント、スペース、空の行を使うと、モデルが読み易くなります。しかし、Mosel

は、これらをスキップします。 

Line breaks：一行に、いくつかのステートメントを書くことは可能です。それを行うには、下記のように、

セミコロンでステートメントを切り離してください。 

plan('A') <> plan('B'); plan('A') <> plan('D')  

しかし、特別な'line end'や continuation character がないので、数行にまたがるステートメント行は、

operator ( +, >=, etc.)や’ , ‘のようなキャラクタで終えなければなりません。それにより、ステートメン

トが終わっていないことが示されます。 

例で示されているように. Mosel のコメント行は、! で始まります。複数行のコメント行は、る倍数が単線

コメントに先行する、、! で始まり、! で終わります。 

 

2.1.2 モデルをランする 

Mosel モデルをランする方法には、三つのモードがあり、その中から一つを選ぶことができます。 

· Mosel command line から行う方法：この方法は、Mosel が利用可能であるすべてのプラット

ホームで使えます。 このモードは、例えば、異なるパラメタ設定を持つ複数のモデルの一

連のランを実行したいような場合、特に便利です。このランモードでは、デバッガを含む、完

全な Mosel の機能性にアクセスできます。 

· Xpress-IVE のグラフィカル環境から行う方法：この方法は、Windows ユーザが使えます。

IVE は、モデル作成と最適化のための開発環境で、Mosel モデルを扱うための内蔵のテキ

ストエディタを持ち、また、多くの、ソリューション、および、サーチツリーのディスプレーも持

っており、これは、開発段階で、モデルとソリューションアルゴリズムを分析するのにとても

便利です。 モデルは、インタラクティブに変更しながらリランできます。 

· アプリケーション・プログラムの中から行う方法： Mosel モデルは、アプリケーション・プログ

ラム(C/C++、Java、VB、.NET)からアクセスでき、実行できます。この機能性は、通常、

Mosel モデルを企業の情報システムと統合して使う場合に使用されます。 

このマニュアルでは、このマニュアルで開発するモデルをランするために、最初の 2 つの方法を使い

ます。モデルをアプリケーション・プログラムに埋め込むことについての詳細は、Mosel のユーザガイド

を参照してください。 

 

2.1.2.1 Mosel command line から行う方法 

完全なモデルを、ファイル meeting.mos に入れれば、問題のソリューションに進むことができます。そ

れには、下記の三つのステージが必要です。 

· meeting.mos をコンパイルして、meeting.bim に変換する。 

· コンパイルされた meeting.bim をロードする。 

· ロードしたモデルをランする。 

Mosel をコマンド・プロンプトから開始し、次の順序でコマンドを入力します。 

mosel 

compile meeting 

load meeting 

run 

quit 

これらのコマンドで、モデルはコンパイルされ、ロードされ、ランされます。そうすると、次のようなアウト



 16 

プットが出てきます。ここで、Mosel の出力はハイライトされ、太字で強調されます。 

mosel 

** Xpress-Mosel ** 

(c) Copyright Dash Associates 1998-2005 

> compile meeting 

Compiling 'meeting'... 

> load meeting 

> run 

Meeting A: 1 

Meeting B: 2 

Meeting C: 1 

Meeting D: 3 

Returned value: 0 

> quit 

Exiting. 

compile/load/run というシークエンスはよく使われるので、下記のように短く表記することもできます。 

exec meeting 

下記のコマンドラインから、同じステップが、即座に、行えます。 

mosel -c "compile meeting; load meeting; run" 

もしくは、 

mosel -c "exec meeting" 

-c オプションには、ダブルコーテーションで挟まれたコマンド・リストが続きます。 

2.1.2.2 IVE を使う 

 

Figure 2.1: IVE after opening a model 

 

モデルファイル meeting.mos を IVE で使うには、下記のステップを実行する必要があります。 

· IVE の立ち上げ： Xpress-IVE の標準インストール手順に従い、インストールを行った後、

デスクトップのアイコンをダブルクリックしてプログラムを始動するか、または、Start >> 

Programs >> Xpress-MP >> Xpress-IV と進んで、プログラムを始動する。また、DOS コマン

ド・プロンプトで ive とタイプインするか、または、拡張子.mos を持つモデルファイルをダブ

ルクリックすることで、IVE を立ち上げることもできます。 

· File >> Open と進んで、モデルファイルを開いてください。そうすると、モデルのソースが、

中央のウインドウ(IVE Editor)にディスプレーされます。 

· Run ボタン をクリックするか、もしくは、Build >> Run と進んでください。  

ワークスペースの下部の Build 枠は、Mosel のモデル実行ステータスメッセージを表示します。モデル

にシンタックスエラーが見つかると、それらは、ここに示され、エラーが検出された行とキャラクターの

位置の詳細と、もしあるなら、問題の説明が表示されます。エラーをクリックすると、エラーを起こして

いる行にユーザを連れて行きます。 

モデルのランが行われると、ワークスペース右側の Output/Input 枠には、プログラムによって生成

されるアウトプットが表示されます。IVE は、また、CP のサーチツリーのグラフ表示(CP search 枠)と、

問題のサマリー統計を表示(CP 統計枠)します。さらに、IVE では、Mosel モデルにサブルーチンを埋め

込むことによって、ユーザはグラフを描くことができます(詳細は、モジュール mmive のドキュメンテー

ションを見てください)。 

 

IVE は、左手の窓の Entities 枠に、ソリューションのすべての情報を表示します。この枠で決定変数の

リストを大きし、マウス・ポインタを近づけると、ソリューション値が表示されます。 

 

 

Figure 2.2: IVE after model execution 

 

2.1.2.3 モデルをディバッグする 

モデルをデバッグするための第一歩は、確かに、追加的な情報を加えることです。ここのモデルで言

えば、例えば、変数のドメインの定義の後に、writeln(plan)というラインを加えることによって、決定

変数の定義を印刷したり、また、手順 cp_show_prob により、問題定義を、全部、印刷したりすること

もきます。デバッグを容易にするために、もっと読み易い出力を出させるには、ユーザは、決定変数に

名前を与えることも出来ます。例えば､ 

 forall(m in MEETINGS) setname(plan(m), "Meeting "+m)  

ここで、ストリングを連結する '+' サインを使用したことに注意してください。 

手順 cp_show_stats を呼ぶと、CP solving のサマリー統計が表示されます。 

コマンドラインバージョンで、run-time errors の詳細情報を得るためには、モデルを、 flag -g でコン

パイルする必要があります。例えば、 

mosel -c "exec -g meeting" 

Mosel デバッガ(詳細は、Mosel 言語リファレンスマニュアルを参照)を使用するには、 compilation 

flag -G を使う必要があります。 

IVE は、デフォルトで、デバッグモードでモデルをコンパイルし、そして、そして、( ボタンをクリックす

るか、Debug メニューから)デバッガをスターとすると、それに対応して、モデルを対応する再コンパイ

ルします。 

2.2 ファイルからのデータ入力 

ここで、前の例を、別のデータセットと共に使用するために拡張します。 別の、異なるデータセットで

モデルをランしたい場合、モデルファイルのデータセットの変更のすべてを編集するのは実用的では

なく、間違いの元となりがちです。したがって、そうする代わりに、テキストファイルからデータを読みこ

みますが、その方法について説明しましょう。 

この問題は、Applications of optimization with Xpress-MPという文書のセクション14.4から

取ってきた問題で、いま、これを解きたいと思っているとします。 

ある技術大学では、いくつかのオプション・モジュールで、期末に、コース試験を計画する必要があり

ます。試験は、すべて、2時間続きます。試験を行うために、2日間が用意され、両日とも、時間帯は、

8:00－10:00、10:15－12:15、14:00－16:00、および、16:15－18:15で、合計8つの時間帯があります。

Table2.1に見られるように、同じ学生が試験を受けなければならない科目は、同じ時間帯には行えず、

したがって、そのような試験科目は、別の時間帯に行われなければなりません（set of incompatible 

exams）。 

 

Table 2.1: Incompatibilities between different exams         

  DA NA C++ SE PM J GMA LP MP S DSE 

DA - X - - X - X - - X X 

NA X - - - X - X - - X X 

C++ - - - X X X X - X X X 

SE - - X - X X X - - X X 

PM X X X X - X X X X X X 

J - - X X X - X - X X X 

GMA X X X X X X - X X X X 

LP - - - - X - X - - X X 

MP - - X - X X X - - X X 

S X X X X X X X X X - X 

DSE X X X X X X X X X X - 

 

2.2.1 Model formulation  

この CP モデルは、前のものと同じ構造をもっています。しかし、前のものとの違いは、データ配列

INCOMP で、これは、同時並行的に行えない試験科目のペアを示しており、ここでは、それらをひとつ

ずつ書き上げることの代わりに、disequality constraints のループで定義します。 

 

{ }8,...,1: ÎÎ" eplanEXAMe  

edde planplanINCOMEXAMed ¹=Î" :1,,  

 

2.2.2 Implementation インプリメンテーション 

この Mosel モデルは、いまや、下記にようになります。 

model "I-4 Scheduling exams (CP)" 

 uses "kalis" 

  

 declarations 

  EXAM = 1..11                        ! Set of exams 

  TIME = 1..8                         ! Set of time slots   

  INCOMP: array(EXAM,EXAM) of integer ! Incompatibility between exams 

  EXAMNAME: array(EXAM) of string 

   

  plan: array(EXAM) of cpvar          ! Time slot for exam 

 end-declarations 

  

 EXAMNAME:: (1..11)["DA","NA","C++","SE","PM","J","GMA","LP","MP","S","DSE"] 

  

 initializations from 'Data/i4exam.dat' 

  INCOMP 

 end-initializations 

 

 forall(e in EXAM) setdomain(plan(e), TIME) 

  

! Respect incompatibilities 

 forall(d,e in EXAM | d<e and INCOMP(d,e)=1)  plan(d) <> plan(e) 

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

  

! Solution printing 

 forall(t in TIME) do 



 21 

  write("Slot ", t, ": ") 

  forall(e in EXAM)  

   if (getsol(plan(e))=t) then write(EXAMNAME(e)," "); end-if 

  writeln 

 end-do 

 

end-model 

 

配列 INCOMP の値は、initializations block のファイル i4exam.dat から読み込まれます。

disequality constraints の定義で、対応する array entry-conditions の値がチェックされ、loop、sum、そ

の他の aggregate operators のインデックスは縦棒でマークされます。 

このデータファイルは、下記の内容を持ちます。 

INCOMP: [0 1 0 0 1 0 1 0 0 1 1 

         1 0 0 0 1 0 1 0 0 1 1 

         0 0 0 1 1 1 1 0 1 1 1 

         0 0 1 0 1 1 1 0 0 1 1 

         1 1 1 1 0 1 1 1 1 1 1 

         0 0 1 1 1 0 1 0 1 1 1 

         1 1 1 1 1 1 0 1 1 1 1 

         0 0 0 0 1 0 1 0 0 1 1 

         0 0 1 0 1 1 1 0 0 1 1 

         1 1 1 1 1 1 1 1 1 0 1 

         1 1 1 1 1 1 1 1 1 1 0] 

 

2.2.3 結果 

モデルは、以下の結果をプリントします。 答は、最初の7つの時間帯だけが試験のスケジュールに使

用されることを示しています。 

 

Slot 1: DA C++ LP  

Slot 2: NA SE MP  

Slot 3: PM  

Slot 4: GMA  

Slot 5: S  

Slot 6: DSE  

Slot 7: J  

Slot 8:  

 

2.2.4 データ・ドリブン・モデル 

上のモデルでは、incompatibility データを、ファイルから読み込みましたが、データの一部(すなわち、

index set EXAM と使える時間帯の数)は、モデルで、いまだ、ハードコードされています。 



 22 

任意のデータセットでランできる、完全にフレキシブルなモデルを作成するには、すべてのデータ定義

を、モデルからデータファイルに移動する必要があります。 

新しいデータファイル i4exam2.dat は、データエントリーを定義するだけではなく、配列 INCOMP の

index tuple も定義します。データエントリーは、すべて、その前に index tuple が付きます。set 

EXAM の内容を明確に書く必要はありません。なぜなら、Mosel は、この set に、配列 INCOMP のイ

ンデックスの値を自動的に読み込むからです。さらに、このデータファイルには、時間帯 NT の数の値

もあります。 

 

INCOMP: [("DA" "NA") 1 ("DA" "PM") 1 ("DA" "GMA") 1 ("DA" "S") 1 ("DA" "DSE") 1 

         ("NA" "DA") 1 ("NA" "PM") 1 ("NA" "GMA") 1 ("NA" "S") 1 ("NA" "DSE") 1 

         ("C++" "SE") 1 ("C++" "PM") 1 ("C++" "J") 1 ("C++" "GMA") 1  

 ("C++" "MP") 1 ("C++" "S") 1 ("C++" "DSE") 1 

         ("SE" "C++") 1 ("SE" "PM") 1 ("SE" "J") 1 ("SE" "GMA") 1 ("SE" "S") 1 

 ("SE" "DSE") 1 

         ("PM" "DA") 1 ("PM" "NA") 1 ("PM" "C++") 1 ("PM" "SE") 1 ("PM" "J") 1  

 ("PM" "GMA") 1 ("PM" "LP") 1 ("PM" "MP") 1 ("PM" "S") 1 ("PM" "DSE") 1 

         ("J" "C++") 1 ("J" "SE") 1 ("J" "PM") 1 ("J" "GMA") 1 ("J" "MP") 1  

 ("J" "S") 1 ("J" "DSE") 1 

         ("GMA" "DA") 1 ("GMA" "NA") 1 ("GMA" "C++") 1 ("GMA" "SE") 1   

 ("GMA" "PM") 1 ("GMA" "J") 1 ("GMA" "LP") 1 ("GMA" "MP") 1  

 ("GMA" "S") 1 ("GMA" "DSE") 1 

         ("LP" "PM") 1 ("LP" "GMA") 1 ("LP" "S") 1 ("LP" "DSE") 1 

         ("MP" "C++") 1 ("MP" "PM") 1 ("MP" "J") 1 ("MP" "GMA") 1 ("MP" "S") 1  

 ("MP" "DSE") 1 

         ("S" "DA") 1 ("S" "NA") 1 ("S" "C++") 1 ("S" "SE") 1 ("S" "PM") 1  

 ("S" "J") 1 ("S" "GMA") 1 ("S" "LP") 1 ("S" "MP") 1 ("S" "DSE") 1 

         ("DSE" "DA") 1 ("DSE" "NA") 1 ("DSE" "C++") 1 ("DSE" "SE") 1   

 ("DSE" "PM") 1 ("DSE" "J") 1 ("DSE" "GMA") 1 ("DSE" "LP") 1  

 ("DSE" "MP") 1 ("DSE" "S") 1 ] 

 

NT: 8 

上のことを反映させるため、モデルを、いくつか変える必要があります。set EXAM と set TIME は、いま

や、タイプをはっきり示すことによって、declareされています。これにより、これらのset は、それまでは、

それらの値を提示することによって、constant definition であったのとは対照的に、 いまや、これらの

set は dynamic set になります。その結果、決定変数 plan は、インデックスセット EXAM が知られる前

に、そして、Mosel がこの配列を dynamic array として作成する前に、declare されます。これは、この

declarationが空の配列を作り、インデックスリストがひとたび知られると、(Mosel手順 createを使い)、

その要素が作られる必要があることを意味しています。変数を作る前に、Xpress-Kalis のデフォルト

bound を set TIME に対応する値に変更し、setdomain への call を置き換えます。 

配列 INCOMP の declaration は、また、dynamic array を生成します。initializations block は、正

確に、データファイルにリストされているエントリーに、値を割り当てます(前に行われている constant 

declaration で、すべてのエントリーが定義されています)。これにより、disequality constraints を定義す

るループの condition を、定義し直すことが可能になります。こうして、いまや、すべてのデータの値を

確認することに代わり、エントリーの存在をテストすればよいことになりました。大きいデータセットの

場合、keyword existsを使うことで、「疎な配列（sparse array）」、すなわち、多次元のデータ配列で、0

以外のエントリーが極めて少ない配列でのループの実行時間を大きく減少できるでしょう。 

 

model "I-4 Scheduling exams (CP) - 2" 

 uses "kalis" 

  

 declarations 

  NT: integer                         ! Number of time slots 

  EXAM: set of string                 ! Set of exams 

  TIME: set of integer                ! Set of time slots   

  INCOMP: array(EXAM,EXAM) of integer ! Incompatibility between exams 

   

  plan: array(EXAM) of cpvar          ! Time slot for exam 

 end-declarations 

 

 initializations from 'Data/i4exam2.dat' 

  INCOMP NT 

 end-initializations 

 

 TIME:= 1..NT 

  

 setparam("default_lb", 1); setparam("default_ub", NT) 

 forall(e in EXAM) create(plan(e)) 

  

! Respect incompatibilities 

 forall(d,e in EXAM | exists(INCOMP(d,e)) and d<e)  plan(d) <> plan(e)   

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

 

! Solution printing 

 forall(t in TIME) do 

  write("Slot ", t, ": ") 

  forall(e in EXAM)  

   if (getsol(plan(e))=t) then write(e," "); end-if 

  writeln 

 end-do 

 

end-model 

この、完全に「データ・ドリブン形式になったモデル」も、勿論、前のモデルと同一の解を与えます。 

決定変数 planを明示的に作成することの代替的な方法は、下に示すように、データのinitializationの

後に declaration を動かすことです。この場合、index set EXAM を完成させ、まとめておくこと

（finalize）が重要です。そうしておくことにより、index set EXAM は、現在のコンテンツを持つ

constant set になり、Mosel は、その後に declare されるすべての配列を、この set によりインデックス

を付し、static array として生成します。 

 

 declarations 

  NT: integer                         ! Number of time slots 

  EXAM: set of string                 ! Set of exams 

  TIME: set of integer                ! Set of time slots   

  INCOMP: array(EXAM,EXAM) of integer ! Incompatibility between exams 

 end-declarations 

 

 initializations from 'Data/i4exam2.dat' 

  INCOMP NT 

 end-initializations 

 

 finalize(EXAM) 

 TIME:= 1..NT 

  

 setparam("default_lb", 1); setparam("default_ub", NT) 

 declarations   

  plan: array(EXAM) of cpvar          ! Time slot for exam 

 end-declarations  

 

2.3 最適化と列挙（enumeration） 

2.3.1 最適化 

セクション 2.2.2 のモデル i4exam_ka.mos をランした結果、すべての時間帯を使用してはいない解が得

られましたが、これからの連想で、この問題が必要とする最小の時間帯の数はいくつだろうか、と考え

るでしょう。この質問は、一つの最適化の問題で、それを定式化しましょう。 

ここで、plani 変数と同じ値の範囲（the same value range）を対象に、新しい決定変数 numslot を導入し、

この変数は、すべての plani 変数に等しいか、大きいということを示す制約式を加えます。 単純化した、

一つの定式化は、変数 numslot は、すべての plani 変数の最大値と等しい、とすることです。 

そうすると、目的関数は、numslot の値を最小化することとなり、そうすると、下記のモデルが得られま

す。 

 

model "I-4 Scheduling exams (CP) - 3" 

 uses "kalis" 

  

 declarations 

  NT: integer                         ! Number of time slots 

  EXAM: set of string                 ! Set of exams 

  TIME: set of integer                ! Set of time slots   

  INCOMP: array(EXAM,EXAM) of integer ! Incompatibility between exams 

   

  plan: array(EXAM) of cpvar          ! Time slot for exam 

  numslot: cpvar                      ! Number of time slots used 

 end-declarations 

 

 initializations from 'Data/i4exam2.dat' 

  INCOMP NT 

 end-initializations 

 

 finalize(EXAM) 

 TIME:= 1..NT 

  

 setparam("default_lb", 1); setparam("default_ub", NT) 

 forall(e in EXAM) create(plan(e)) 

  

! Respect incompatibilities 

 forall(d,e in EXAM | exists(INCOMP(d,e)) and d<e)  plan(d) <> plan(e) 

 

! Calculate number of time slots used 

 numslot = maximum(plan) 

 

! Solve the problem 

 if not(cp_minimize(numslot)) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

 

! Solution printing 

 forall(t in TIME) do 

  write("Slot ", t, ": ") 

  forall(e in EXAM)  

   if (getsol(plan(e))=t) then write(e," "); end-if 

  writeln 

 end-do 

 

end-model 

cp_find_next_sol に代わり、ここでは、目的関数変数 numslot を持つ cp_minimize を function 

argument として使います。 

このプログラムは、利用可能な 7 つの時間帯を前提にソリューションを与え、その答は、実行可能なス

ケジュールを作り出すのに必要な最少時間帯数を示します。 

2.3.2 列挙（Enumeration） 

IVE の CP stats pane や、モデルのバージョンの終わりに追加する cp_show_stats をコールすること

で得られる problem statistics を比較すると分かりますが、「実行可能なソリューションを見つける」とい

うことから、「最適化」に切り替えると、CP ソルバーがサーチするノードの数が、かなり、増加します。 

これまでのところ、Xpress-Kalis のデフォルトの列挙戦略でサーチを行ってきました。ここで、サーチす

るノードの数を減少させることができるかどうか、したがって、最適性を確認するのに費やされる時間

を短くさせることができるかどうかを見てみましょう。 

有限ドメイン変数を列挙するための Kalis のデフォルト戦略は、サーチの始まりの前に、下記のステー

トメントを追加することに相当します。 

cp_set_branching(assign_var(KALIS_SMALLEST_DOMAIN, KALIS_MIN_TO_MAX)) 

assign_var は、branching scheme (`a branch is formed by assigning the next chosen value to the 

branching variable')を意味し、KALIS_SMALLEST_DOMAIN は、variable selection strategy (`choose the 

variable with the smallest number of values remaining in its domain')であり、KALIS_MIN_TO_MAX は、

value selection strategy (`from smallest to largest value')です。 

ここでは、時間帯の数を最小にしようとているので、最も小さい値から列挙を開始するのはよい考え

のように思えます。したがって、デフォルト値を選択基準としたままでおきましょう。しかし、ここで、

KALIS_SMALLEST_DOMAIN を KALIS_MAX_DEGREE に変えて、変数選択のヒューリスティクスを変えて見

ましょう。そうすると、ツリーのサイズとサーチ時間が、デフォルトの半分以下になりました。 

以下は、こうした改訂を行った後のモデルの全体です。 

model "I-4 Scheduling exams (CP) - 4" 

 uses "kalis" 

  

 declarations 

  NT: integer                         ! Number of time slots 

  EXAM: set of string                 ! Set of exams 

  TIME: set of integer                ! Set of time slots   

  INCOMP: array(EXAM,EXAM) of integer ! Incompatibility between exams 

   

  plan: array(EXAM) of cpvar          ! Time slot for exam 

  numslot: cpvar                      ! Number of time slots used 

 end-declarations 

 

 initializations from 'Data/i4exam2.dat' 

  INCOMP NT 

 end-initializations 

 finalize(EXAM) 

 TIME:= 1..NT 

  

 setparam("default_lb", 1); setparam("default_ub", NT) 

 forall(e in EXAM) create(plan(e)) 

  

! Respect incompatibilities 

 forall(d,e in EXAM | exists(INCOMP(d,e)) and d<e)  plan(d) <> plan(e) 

 

! Calculate number of time slots used 

 numslot = maximum(plan) 

 

! Setting parameters of the enumeration 

 cp_set_branching(assign_var(KALIS_MAX_DEGREE, KALIS_MIN_TO_MAX))   

 

! Solve the problem 

 if not(cp_minimize(numslot)) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

 

! Solution printing 

 forall(t in TIME) do 

  write("Slot ", t, ": ") 

  forall(e in EXAM)  

   if (getsol(plan(e))=t) then write(e," "); end-if 

  writeln 

 end-do 

  

 cp_show_stats 

  

end-model 

注意： 最適化のないモデルバージョンでは、ランダムに値を選ぶので、より均等に分布するスケジュ

ー ル を 得 よ う と し て いま す 。 す な わ ち 、 Kalis_MIN_TO_MAX で は な く 、 値 の 選 択 基 準 Kalis 

RANDOM_VALUE を使用している、ということです。 

分岐戦略の定義の詳細は、第 4 章を見てください。 

Figure 2.3: Search tree displays in IVE (left: default strategy, right: KALIS_MIN_DEGREE strategy for variable choice) 

2.3.2.1 IVE search tree display 

IVE では、この二つのサーチツリーの違いは、CP search pane に示されるツリーのグラフィカルな表示

により、容易に目に見えるようになります(Figure2.3 を参照)。実現可能な解は緑色の正方形によって

表され、最適化問題の最適解は、わずかに大きい正方形でマークされます。Figure2.3 で見るように、

両方の戦略で、一つのソリューションが見つけられます。サーチツリー・ディスプレーのノードの上で、

マウスのポインターを動かすと、ノード番号、分岐変数の名前を含む情報の詳細がポップアップボック

スの中に現れます。また、ツリー・ディスプレーの上の selection box に変数名を示してやると、所与の

変数で分岐するすべてのノードをハイライトすることもできます。IVE で、もっと意味のあるディスプレー

を得たい場合は、手順 setname を使い、決定変数に名前を割り当てます(セクション 2.1.2.3 を参照)。 

2.4 連続変数 

本章を通して、これまで、決定変数 type cpvar(離散変数)を扱ってきました。Xpress-Kalis の 2 番目

の変数 type は、連続変数(type cpfloatvar)です。このような変数は、上で、離散変数について見

てきたのと同様の方法で使います。例えば、下記の例を見てください。 

 

 setparam("DEFAULT_CONTINUOUS_LB", 0) 

 setparam("DEFAULT_CONTINUOUS_UB", 10)  

 

 declarations 

  x,y: cpfloatvar 

 end-declarations 

 

 x >= y                   ! Define a constraint 

                          ! Retrieve information about continuous variables 

 writeln(getname(x), ":", getsol(x))    

 writeln(getlb(y), " ", getub(y))  

離散変数タイプと連続変数タイプの使い方には、下記のような、いくつかの違いがあります。 

· cpfloatvar を含む制約式は、厳密な不等式（strict inequalities）にはなれない(すなわち、演

算子は、<=、>=、および、= だけが使用できる。 

· ほとんどの global constraints（第 3 章を参照）は、cpvar のみに適用できる。 

· 変数のドメインで値を列挙するサーチ戦略は、cpvar のみとしか使えない(第 4 章を参照)。 

· getnext のような、ドメインの値を列挙する access functions は、cpfloatvar に適用できな

い。 

第 3 章 制約式 

本章では、異なるタイプの(最適化)問題例を使い、これらを解くために、どのようにXpress-Kalisを使う

方法を示します。最初のセクションは、簡単な線形制約式の例で、問題をどのように定義し、ポストす

るためのいくつかの別の方法を示します。以下のいくつかのセクションでは、それぞれ、新しい制約式

タイプを導入します。 ほとんどの例が、異なる制約式の組み合わせを使用するので、以下のリストは、

ある、一つの制約式タイプの使用例を見つけるの役立つ TOC 思います。 

 

· arithmetic： 2.1 (meeting.mos), 2.2 (exam*.mos), 3.7 (persplan.mos)； 

linear equality/inequality： 3.5 (b4seq*_ka.mos), 3.6 (i1assign_ka.mos),  

3.8 (b5paint*_ka.mos), 3.9 (j5tax_ka.mos), 4.4 (i1assign_ka.mos),  

5.2 (b1stadium_ka.mos)；  

Non-linear}nonlinear： 3.2 

· all_different： 3.3 (sudoku_ka.mos), 3.5 (b4seq_ka.mos), 3.8 (b5paint*_ka.mos),  

3.4 (freqasgn.mos),.3.7 (persplan.mos), 3.11 (eulerkn*.mos), 4.4 (i1assign_ka.mos) 

· abs / distance： 3.4 (freqasgn.mos) 

· cumulative：5.4 (d4backup2_ka.mos) 

· cycle：3.10 (b5paint3_ka.mos) 

· disjunctive： 3.5 (b4seq2_ka.mos) 

· element：3.5 (b4seq_ka.mos), 3.6 (a4sugar_ka.mos), 3.9 (j5tax_ka.mos),  

3.8 (b5paint_ka.mos), 4.4 (i1assign_ka.mos), ； 2D： 3.8 (b5paint2_ka.mos)  

· distribute / occurrence： 3.6 (a4sugar_ka.mos), 3.7 (persplan.mos), 3.9 (j5tax_ka.mos) 

· maximum / minimum： 2.3 (exam3.mos, exam4.mos), 3.5 (b4seq2_ka.mos),  

3.4 (freqasgn.mos), 4.4 (i1assign_ka.mos)  

· implies/equiv：3.8 (b5paint_ka.mos), 3.7 (persplan.mos), 3.9 (j5tax_ka.mos),  

3.11 (eulerkn2.mos) 

· generic binary：3.1 (eulerkn.mos) 

 

3.1 制約式のハンドリング 

このセクションでは、セクション 2.1 の会議のスケジュールを作成するという問題を例にとります。この

モデルには、離散変数についての簡単な算数制約式からなっています。しかし、ここで言えることは、

すべて、他の、連続決定変数(すなわち、タイプ cpfloatvar の変数)の制約式関係を含む、すべての

タイプの Xpress-Kalis の制約式についても言えることです。 

下に、この問題の制約式として、いくつか、付け加えたいと思います。 

· 会議 B は、第 3 日よりも前に開催されなければならない。 

· 会議 D は、第 2 日においては開催できない。 

· 会議 A は、第 1 日に開催されなければならない。 

 

3.1.1 モデルの定式化 

これらの 3 つの追加制約式は、簡単な(単項)リニア制約式で表現できます。こうして、以下のように、

モデルを完成します。 

 

{ }3,2,1: ÎÎ" mplanMEETINGSm  

2£Bplan  

2¹Dplan  

1=Aplan  

BA Planplan ¹  

DA Planplan ¹  

CB Planplan ¹  

DB Planplan ¹  

3.1.2 Implementation  

この新しい制約式の Mosel による実行は、極めて、簡単です。 

model "Meeting (2)" 

 uses "kalis" 

  

 declarations 

  MEETINGS = {'A','B','C','D'}        ! Set of meetings 

  TIME = 1..3                         ! Set of time slots    

  plan: array(MEETINGS) of cpvar      ! Time slot per meeting 

 end-declarations 

 

 forall(m in MEETINGS) do 

  setdomain(plan(m), TIME) 

  setname(plan(m), "plan"+m) 

 end-do 

 writeln("Original domains: ", plan) 

  

 plan('B') <= 2                       ! Meeting B before day 3 

 plan('D') <> 2                       ! Meeting D not on day 2 

 plan('A') = 1                        ! Meeting A on day 1 

 writeln("With constraints: ", plan) 

  

! Respect incompatibilities 

 plan('A') <> plan('B') 

 plan('A') <> plan('D') 

 plan('B') <> plan('C') 

 plan('B') <> plan('D')   

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

  

! Solution printing 

 forall(m in MEETINGS) 

  writeln("Meeting ", m, ": ", getsol(plan(m))) 

 

end-model 

 

3.1.3 結果 

読者の皆さんも、多分、既に、お気付きのように、モデルの中の数箇所に、変数 planm のプリントアウト

を加えました。したがって、このモデルの実行の出力は以下のようになります。 

 

Original domains: [planA[1..3],planB[1..3],planC[1..3],planD[1..3]] 

With constraints: [planA[1],planB[1..2],planC[1..3],planD[1,3]] 

Meeting A: 1 

Meeting B: 2 

Meeting C: 1 

Meeting D: 3  

この出力からわかるように、制約式を記述した直後に、関係する変数のドメインは減少しています。制

約式は、「即座に、かつ、自動的に」ソルバーに渡され、そして、それらの結果は問題全体に伝播され

ます。 

 

3.1.4 制約式に名前を付ける 

定義を行った後に、特に、制約式がロジック関係の一部なったり、（disjunctive constraintsのように）ブ

ランチされたりするような場合、後で制約式にアクセスすることが必要になることがあります。この目的

に対応するため、Xpress-Kalis は、新しいタイプ、cpctr を定義します。 cpctr は、制約式を declare し、

制約式が定義された後に使う「一つの名前」を、それらの制約式に与えることに使います。制約式は、

制約式関係（constraint relation）を割り当てることによって、定義されます。 

 

制約式に名前をつけることにより、制約式が自動的にソルバーに渡されないという副次的な効果があ

ります。これは、以下のモデルで示されているように、(定義の後に)それ自身へのステートメントとして

制約式の名前を書くことで行う必要があります。 

model "Meeting (3)" 

 uses "kalis" 

  

 declarations 

  MEETINGS = {'A','B','C','D'}        ! Set of meetings 

  TIME = 1..3                         ! Set of time slots    

  plan: array(MEETINGS) of cpvar      ! Time slot per meeting 

  Ctr: array(range) of cpctr 

 end-declarations 

 

 forall(m in MEETINGS) do 

  setdomain(plan(m), TIME) 

  setname(plan(m), "plan"+m) 

 end-do 

 writeln("Original domains: ", plan) 

   

 Ctr(1):= plan('B') <= 2              ! Meeting B before day 3 

 Ctr(2):= plan('D') <> 2              ! Meeting D not on day 2 

 Ctr(3):= plan('A') = 1               ! Meeting A on day 1 

 writeln("After definition of constraints:\n ", plan) 

  

 forall(i in 1..3) Ctr(i)  

 writeln("After posting of constraints:\n ", plan) 

  

! Respect incompatibilities 

 plan('A') <> plan('B') 

 plan('A') <> plan('D') 

 plan('B') <> plan('C') 

 plan('B') <> plan('D')   

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if    

  

! Solution printing 

 forall(m in MEETINGS) writeln("Meeting ", m, ": ", getsol(plan(m))) 

 

end-model 

このモデルの出力から、名付けられた制約式の単なる定義によっては、変数のドメインに、なんら、影

響しないことがわかります。すなわち、制約式は、Mosel で定義されますが、いまだ、Kalis ソルバーに

送られていない、ということです。制約式の名前をステートした後にだけ、すなわち、それらをソルバー

に送った後においてのみ、前のモデルバージョンと同じ、ドメインの減少が得られます。 

 

Original domains: [planA[1..3],planB[1..3],planC[1..3],planD[1..3]] 

After definition of constraints: 

 [planA[1..3],planB[1..3],planC[1..3],planD[1..3]] 

After posting of constraints: 

 [planA[1],planB[1..2],planC[1..3],planD[1,3]] 

Meeting A: 1 

Meeting B: 2 

Meeting C: 1 

Meeting D: 3  

 

3.1.5 制約式を明示的に渡す 

これまでのすべての例で、制約式をソルバーに渡しても、失敗しない、すなわち、ソルバーがインフィ

ージビリティを見つけないと、暗黙に仮定していました。 

実際には、これは、必ずしも、常時、適切な仮定あるわけではありません。したがって、Xpress-Kalis

は、ソルバーに明示的に制約式を渡し、それを追加した後に、制約式システムの状態を返すファンク

ション cp_post を定義します。このファンクションは、フィージブルには true を、そしてインフィージブ

ルには false を返してきます。この機能は、制約を厳しくし過ぎた問題に対応するのに便利です。つ

まり、インフィージビリティが見つかると、それ以上の制約式の追加を止め、そして、どの制約式がイン

フィージビリティを惹き起しているかをユーザに知らせてくれるからです。 

cp_post を使うには、前のモデルで、forall(i in 1..3) Ctr(i)という行を、下記のコードに置き

換えます。 

 

 forall(i in 1..3)  

  if not cp_post(Ctr(i)) then  

   writeln("Constraint ", i, " makes problem infeasible") 

   exit(1) 

  end-if 

制約式をソルバーに渡すと、そのすべてについて、返されてくる値がチェックされ、ある一つの制約式

の追加がインフィージビリティを惹き起こすと、プログラムは止められます。 

このモデルバージョンの出力は、前のセクションで見たものと、まったく、同一です。 

3.1.6 明示的な制約式のプロパゲーション 

別の方法で、制約式の行動に影響を及ぼすこともできます。自動的な制約式のプロパゲーションをオ

フにする（setparam("AUTO_PROPAGATE", false)と、ソルバーに渡された制約式のプロパゲーション

は行われません。この場合、制約式のプロパゲーションは、cp_propagate へのコールか、列挙

(subroutines cp_find_next_sol, cp_minimize, etc.)の開始によってのみ始められます。 

3.2 算数制約式（Arithmetic constraints） 

前のセクションで、有限ドメイン変数の線形制約式の例をいくつか見ました。線形制約式は、算数制

約式の特殊なケース、すなわち、演算子 +, -, /, *, ^, sum, prod や、abs、ln のような関数で形

成されている方程式や不等式であると見なされます。ソルバーによってサポートされている算数ファン

クションの全リストについては、Xpress-Kalis reference manual を参照してください。 

Xpress-Kalis の算数制約式は、有限ドメイン変数(type cpvar)、連続型変数(type cpfloatvar)、もしく

は、両方が混ざったもので定義されます。しかし、連続変数を含む算数制約式は、strict inequalities と

しては定義できないのに注意してください。これは、関係を示す演算子として、>=、<=、 および､ = だ

けが使用できることを意味します。 

以下に、Xpress-Kalis で定義できる非線形算数制約式のいくつかの例を示します。 

 

model "Nonlinear constraints" 

 uses "kalis" 

  

 setparam("DEFAULT_LB", 0) 

 setparam("DEFAULT_UB", 5) 

 setparam("DEFAULT_CONTINUOUS_LB", -10) 

 setparam("DEFAULT_CONTINUOUS_UB", 10)  

  

 declarations 

  a,b,c: cpvar 

  x,y,z: cpfloatvar 

 end-declarations  

  

 x = ln(y) 

 y = abs(z) 

 x*y <= z^2 

 z = -a/b 

 a*b*c^3 >= 150 

  

 while (cp_find_next_sol)  

  writeln("a:", getsol(a), ", b:", getsol(b), ", c:", getsol(c), 

         ", x:", getsol(x), ", y:", getsol(y), ", z:", getsol(z)) 

  

end-model  

3.3 all_different: 数独 

「数独パズル」は、日本発のパズルで、最近、多くの欧米の新聞でも見られるようになりました。 この

パズルは、与えられた 9x9 のボードに 1 から 9 の数字を埋めてゆくのですが、このボードの一部の枡

には数字が入っています。そして、問題は、空白の升目に埋め込む数字は、その升目の所属している

行、列、および、3x3 のサブボードのすべてが、1 から 9 の数字を一つだけ含むように数字を埋めて、

9x9 のボードのすべての空の枡に数字を入れ込むことです。Table 3.1 と Table 3.2 は、このパズルの 2

つの例題です。 人が解くには、扱いにくい問題ですが、これらのパズルは、CP を使って解くことが出

来ます。 

（訳注：「数独パズル」については、http://www.oct.zaq.ne.jp/woodside/jsudok/ を参照。 

 

Table 3.1: Sudoku (`The Times', 26 January, 2005)     

  A B C D E F G H I 

1         4 3   6   

2   6 5       7     

3 8     7         3 

4   5       1 3   7 

5 1 2           8 4 

6 9   7 5       2   

7 4         5     9 

8     9       4 5   

9   3   4 6         

 

Table 3.2: Sudoku (`The Guardian', 29 July, 2005)     

  A B C D E F G H I 

1 8         3       

2   5         4     

3 2       7     6   

4       1         5 

5     3       9     

6 6         4       

7   7     2       3 

8     4         1   

9       9         8 

 

3.3.1 モデルの定式化 

これらの例のように、ボードの列を集合 XS = {A,B,...,I}で、そして、行を集合 YS = {1,2,...,9}

で示します。XS のすべての x と、YS のすべての y に関して、決定変数 vxyを定義し、その値として、

位置(x, y)の数を取るとします。 

この問題の制約式は、下記のものだけです。 

 (1) ある行の数字は、すべて、異なる数字でなければならない。 

 (2) ある列の数字は、すべて、異なる数字でなければならない。 

 (3) 3x3 のサブボードの数字は、すべて、異なる数字でなければならない。 

 

これらの制約式は、Xpress-Kalis の all_different relation で容易に表現できます。この制約式

は、この関係にあるすべての変数が、異なる値を取るようにします。 

 

{ }9,...,1:, ÎÎ" xyvyXSx  

allXSx :Î" - ( )91 ,..., xx vvdifferent  

allYSy :Î" - ( )IyAy vvdifferent ,...,  

all - ( )31 ,..., CA vvdifferent  

all - ( )64 ,..., CA vvdifferent  

all - ( )97 ,..., CA vvdifferent  

all - ( )31 ,..., FD vvdifferent  

all - ( )64 ,..., FD vvdifferent  

all - ( )97 ,..., FD vvdifferent  

all - ( )31 ,..., IG vvdifferent  

all - ( )64 ,..., IG vvdifferent  

all - ( )97 ,..., IG vvdifferent  

そして、いくつかの変数 vxyを、与えられている数に固定します。 

 

3.3.2 インプリメンテーション 

Table3.2 の Sudoku パズルの Mosel インプリメンテーションは、以下のようになります。 

 

model "sudoku (CP)" 

 uses "kalis" 

 

 forward procedure print_solution(numsol: integer) 

  

 setparam("default_lb", 1) 

 setparam("default_ub", 9)               ! Default variable bounds 

 

 declarations 

  XS = {'A','B','C','D','E','F','G','H','I'}   ! Columns 

  YS = 1..9                              ! Rows 

  v: array(XS,YS) of cpvar               ! Number assigned to cell (x,y) 

 end-declarations 

! Data from "The Guardian", 29 July, 2005. http://www.guardian.co.uk/sudoku 

 v('A',1)=8; v('F',1)=3 

 v('B',2)=5; v('G',2)=4 

 v('A',3)=2; v('E',3)=7; v('H',3)=6 

 v('D',4)=1; v('I',4)=5 

 v('C',5)=3; v('G',5)=9 

 v('A',6)=6; v('F',6)=4 

 v('B',7)=7; v('E',7)=2; v('I',7)=3 

 v('C',8)=4; v('H',8)=1  

 v('D',9)=9; v('I',9)=8   

 

 

! All-different values in rows  

 forall(y in YS) all_different(union(x in XS) {v(x,y)}) 

 

! All-different values in columns 

 forall(x in XS) all_different(union(y in YS) {v(x,y)}) 

 

! All-different values in 3x3 squares 

 forall(s in {{'A','B','C'},{'D','E','F'},{'G','H','I'}}, i in 0..2) 

  all_different(union(x in s, y in {1+3*i,2+3*i,3+3*i}) {v(x,y)}) 

 

! Solve the problem 

 solct:= 0 

 while (cp_find_next_sol) do 

  solct+=1 

  print_solution(solct) 

 end-do 

  

 writeln("Number of solutions: ", solct)   

 writeln("Time spent in enumeration: ", getparam("COMPUTATION_TIME"), "sec") 

 writeln("Number of nodes: ", getparam("NODES")) 

 

!**************************************************************** 

! Solution printing 

 procedure print_solution(numsol: integer) 

  writeln(getparam("COMPUTATION_TIME"), "sec: Solution ", numsol) 

  writeln("   A B C   D E F   G H I") 

  forall(y in YS) do 

   write(y, ": ") 

   forall(x in XS)  

    write(getsol(v(x,y)), if(x in {'C','F'}, " | ", " ")) 

   writeln 

   if y mod 3 = 0 then 

    writeln("   ---------------------") 

   end-if  

  end-do 

 end-procedure 

end-model  

 

 

!**************************************************************** 

 

このモデルでは、cp_find_next_sol が、while ループに埋め込まれ、すべての実現可能な解をサー

チします。すべてのループの実行で、見つかったソリューションを綺麗にフォーマットして印刷するため

に手順 print_solution がコールされます。Mosel のサブルーチンには、ローカルの宣言のために

declarations blocks があるかもしれません。そして、それらは、いろいろなアーギュメントを取ります。

このモデルでは、このままでは、この手順へのコールが定義の前に起こるので、キーワード forward

を使い、最初のコールの前に宣言する必要があります。 

サブルーチンによって印刷される情報を選択するために、Mosel の 2 つの異なるバージョンの if ステ

ートメントを使います。すなわち、inline if と、一連のステートメントを含む if-then です。 

モデルランの終わりに、ソルバーからランタイム統計(parameter COMPUTATION_TIME)と、サーチに

よって探索したノードの数(parameter NODES)をリトリーブします。 

Xpress-Kalis タイプによって定義されたパラメタの全リストを得るには、 

mosel -c "exam -p kalis" 

というコマンドをタイプします。Xpress-Kalis の機能の完全なリストを得るには、flag -p を取ってくださ

い。IVE では、Modules >> List available modules を選択するか、あるいは、 ボタンをクリック

してください。 

3.3.3 結果 

上のモデルは、以下を出力します。通常、数独パズルの解は一つしかありませんが、この問題の答も

一つです。 

 

0.16sec: Solution 1 

   A B C   D E F   G H I 

1: 8 6 9 | 2 4 3 | 1 5 7 

2: 3 5 7 | 6 1 9 | 4 8 2 

3: 2 4 1 | 8 7 5 | 3 6 9 

   --------------------- 

4: 4 9 8 | 1 3 2 | 6 7 5 

5: 7 1 3 | 5 8 6 | 9 2 4 

6: 6 2 5 | 7 9 4 | 8 3 1 

   --------------------- 

7: 1 7 6 | 4 2 8 | 5 9 3 

8: 9 8 4 | 3 5 7 | 2 1 6 

9: 5 3 2 | 9 6 1 | 7 4 8 

   --------------------- 

Number of solutions: 1 

Time spent in enumeration: 0.41sec 

Number of nodes: 2712 

all_different 関係は、オプショﾅﾙな 2 番目のアーギュメントを取り、これにより、ユーザは、制約条件を

評価するのに使われるプロパゲーションアルゴリズムを指定できます。ここで、例えば､2 番目のアー

ギ ュ メ ン ト と し て 、 デ ィ フ ォ ル ト 設 定 (_Kalis FORWARD_CHECKING)を 、 も っ と 攻 撃 的 な 戦 略

Kalis_GEN_ARC_CONSISTENCY に変更したとすると、 

  forall(y in YS) 

   all_different(union(x in XS) {v(x,y)}, KALIS_GEN_ARC_CONSISTENCY) 

ノードの数が、たったの一つのノードにまで減少するのが見られるでしょう。すなわち、問題が、単に

制約条件をポストすることで解かれた、ということです。サーチに費やされた時間は、ゼロにまで減少

しますが、制約条件をポストすることに費やさる時間は 4-5 倍になりますが、それでも、1 秒の何分の

一です。この時間が永くなるのは、generalized arc consistency algorithm の計算オーバーヘ

ッドが大きくなるからです。しかし、全体としてみると、問題の定義と、問題を解くのための時間は、前

に較べ、10 分の 1 未満に短縮されます。 

一般的に、generalized arc consistency algorithm は、より強い刈り込み（stronger pruning）

を実現します。すなわち、このアルゴリズムは、変数のドメインから、より多くの値を取り除いてくれま

す。しかし、計算時間の増加のため、このアルゴリズムの使用は、いつも正当化されるというわけでは

ありません。したがって、読者は、皆さんのモデルで、両方のアルゴリズム設定を試みるようにしてくだ

さい。 

· 3.4 abs と distance: 周波数割当 

テレコミュニケーションの領域、特に、移動体通信は、周波数割当問題に関係する、興味深いさまざま

な問題をもたらしました。 

ここに、離散的な周波数帯要件を持つセル(ノード)のネットワークがあります。各セルには、多数の周

波数(バンド)の既知の需要があります。Figure3.1 は、ネットワークの構造を示しています。「線 

(edge)」で結ばれたノードは、隣接する「隣人」であるとみなされます。それらのセルは、干渉を避ける

ために、同じ周波数を割り当てられません。その上、あるセルがいくつかの周波数を使用するなら、そ

れらは少なくとも 2 だけ、異なっていなければなりません。ここでの目的は、ネットワークに使用される

周波数の総数を最小にすることです。 

Figure 3.1: Telecommunications network 

 

Table 3.3 は、個々のセルの周波数の数の需要を示しています。 

 

Table 3.3: Frequency demands at nodes 

Cell 1 2 3 4 5 6 7 8 9 10 

Demand 4 5 2 3 2 4 3 4 3 2 

 

 

3.4.1 Model の定式化 

NODES がネットワークのすべてのノードの集合を、そして、DEMn が n Î NODES の周波数の需要を表

すものとします。また、 ネットワークは、線 LINKS の集合として与えられます。さらに、DEMANDS = 

{1,2,...,NUMDEM}が、周波数の集合を表すものとし、この集合では、すべてのノードは連続して付番

されており、上限 NUMDEM が需要の総数によって与えられているとします。補助的な array INDEXn は、

ノード n の DEMANDS の starting index を示すものとします。 

あらゆる需要 dÎDEMANDSに割り当てられる周波数を表すために、usedを導入します。ここで、used

は、set {1, 2,...,NUMDEM}.からその値を取ります。 

こうして、2 つの集合の制約条件(隣接しているノードに割り当てられる異なる周波数、および、ノード

の中の周波数間の最小の距離)は、以下のようにモデル化できます。 

目的関数は、使用される周波数の数を最小にすることです。ここでは、used 変数の最大の周波数の

数を最小にすることで定式化します。 

 

minimize maximum )( dDEMANDSd useÎ  

 

3.4.2 インプリメンテーション 

テレコミュニケーション・ネットワークを形成する線は、リスト LINK としてモデル化されます。そこでは、

「線 l（edge l）」は、(LINK(l,1),LINK(l,2))として与えられます。 

同じノードに割り当てられる周波数の値に関しての制約条件のインプリメンテーションには、Kalis では、

2 つの同等な選択があります。すなわち、abs 制約条件を使うか、もしくは、distance 制約条件を使う

かです。 

model "Frequency assignment" 

 uses "kalis" 

  

 forward procedure print_solution 

   

 declarations 

  NODES = 1..10                               ! Range of nodes 

  LINKS = 1..18                               ! Range of links between nodes 

  DEM: array(NODES) of integer                ! Demand of nodes 

  LINK: array(LINKS,1..2) of integer          ! Neighboring nodes  

  INDEX: array(NODES) of integer              ! Start index in 'use'  

  NUMDEM: integer                             ! Upper bound on no. of freq. 

 end-declarations 

 

 DEM :: (1..10)[4, 5, 2, 3, 2, 4, 3, 4, 3, 2] 

 LINK:: (1..18,1..2)[1, 3, 1, 4, 1, 6,  

                     2, 4, 2, 7,  

                     3, 4, 3, 6, 3, 8, 3, 9, 

                     4, 7, 4, 9, 4,10,  

                     5, 7, 5, 8, 5, 9,  

                     6, 9, 7, 8, 8,10] 

 NUMDEM:= sum(n in NODES) DEM(n) 

 

! Correspondence of nodes and demand indices: 

! use(d) d = 1, ..., DEM(1) correspond to the demands of node 1 

!            d = DEM(1)+1, ..., DEM(1)+DEM(2))     - " -     node 2  etc. 

 INDEX(1):= 1 

 forall(n in NODES | n > 1) INDEX(n) := INDEX(n-1) + DEM(n-1) 

 

 declarations 

  DEMANDS = 1..NUMDEM                         ! Range of frequency demands  

  use: array(DEMANDS) of cpvar                ! Frequency used for a demand 

  numfreq: cpvar                              ! Number of frequencies used 

  Strategy: array(range) of cpbranching 

 end-declarations 

 

! Setting the domain of the decision variables 

 forall(d in DEMANDS) setdomain(use(d), 1, NUMDEM) 

 

! All frequencies attached to a node must be different by at least 2 

 forall(n in NODES, c,d in INDEX(n)..INDEX(n)+DEM(n)-1 | c<d) 

  distance(use(c), use(d)) >= 2 

!  abs(use(c) - use(d)) >= 2 

   

! Neighboring nodes take all-different frequencies 

 forall(l in LINKS) 

  all_different( 

   union(d in INDEX(LINK(l,1))..INDEX(LINK(l,1))+DEM(LINK(l,1))-1) {use(d)} +  

   union(d in INDEX(LINK(l,2))..INDEX(LINK(l,2))+DEM(LINK(l,2))-1) {use(d)}, 

   KALIS_GEN_ARC_CONSISTENCY) 

 

! Objective function: minimize the number of frequencies used, that is,  

! minimize the largest value assigned to 'use' 

 setname(numfreq, "NumFreq") 

 numfreq = maximum(use) 

 

! Search strategy 

 Strategy(1):=assign_var(KALIS_SMALLEST_DOMAIN, KALIS_MIN_TO_MAX, use) 

 Strategy(2):=assign_var(KALIS_MAX_DEGREE, KALIS_MIN_TO_MAX, use) 

 cp_set_branching(Strategy(1)) 

 setparam("MAX_COMPUTATION_TIME", 1) 

 cp_set_solution_callback("print_solution") 

 

! Try to find solution(s) with strategy 1 

 if (cp_minimize(numfreq)) then 

  cp_show_stats 

  sol:=getsol(numfreq) 

 end-if 

 

! Restart search with strategy 2 

 cp_reset_search 

 if sol>0 then                             ! If a solution was found: 

  numfreq <= sol-1                         ! Add upper bound on objective 

 end-if 

 cp_set_branching(Strategy(2)) 

 setparam("MAX_COMPUTATION_TIME", 1000) 

 

 if (cp_minimize(numfreq)) then 

  cp_show_stats 

 elif sol>0 then 

  writeln("Optimality proven") 

 else 

  writeln("Problem has no solution") 

 end-if 

!******************************************************************** 

! **** Solution printout **** 

 procedure print_solution 

  writeln("Number of frequencies: ", getsol(numfreq)) 

  writeln("Frequency assignment: ") 

  forall(n in NODES) do 

   write("Node ", n, ": ") 

   forall(d in INDEX(n)..INDEX(n)+DEM(n)-1) write(getsol(use(d)), " ") 

   writeln 

  end-do 

 end-procedure 

 

end-model 

異なるサーチ戦略で実験してみると、variable selection criterion を KALIS_MAX_DEGREE に変え

ることで得られる戦略は、ひとたび、「良いソリューション」が見つけられると、最適性を容易に立証

できるのがわかりました。したがって、この問題は以下の 2 ステップで解くことができます。まず、「良い

ソリューション」を見つけるために、ディフォルト戦略を使います。このサーチは、time limit を設定す

ることで、1 秒後に止められます。そして、2 番目の戦略と、前のランで得られた目的関数の値に置

いた bound を使い、再度、サーチを開始します。サーチを再開する前に、cp_reset_search を使って、

ソルバーの中のサーチツリーをリセットする必要があります。 

異なるサーチ戦略の実験を容易にするため、タイプ cpbranching の array Strategy を定義しました。

この array は、異なるサーチ戦略定義をストアするためのものです。 

このインプリメンテーションによって実証された新しいフィーチャーは、callback の使用、より正確に

は、Xpress-Kalis の solution callback です。solution callback は、サーチがソリューションを見つけ

たときは、いつでも、ソルバーがコールできるユーザサブルーチンで定義されます。その典型的な用

途は、中間的なソリューションの logging や storing、もしくは、統計を取ることなどです。手順

print_solution は、単に、それまでに見つかったソリューションを印刷するに過ぎません。 

問題の定式化を改善する：上の問題定式化では、あるノードのすべての demand 変数、および、

これらの変数の制約条件が完全に左右対称であることに気づかれるかも知れません。他の制約

条件がないので、同じセルに属す需要 d と d +1 に関して、下記を、変数 use に課すことで、

これらの左右対称性を利用できるでしょう。 

used + 1 £  used+1 

こうすることで、ディフォルト戦略を使って、問題は、40以下のノードで最適解が得られました。これを、

下記のように書くことで、さらに、もう一歩進めることも出来ます。 

used + 2 £  used+1 

これらの制約条件の追加により、もう少し、ノード・サーチが短なります。これらの制約条件は、 abs 

constraint や distance constraint の代わりに使うことさえできます。 

 

3.4.3 結果 

この問題の最適解は、11 の異なる周波数を使用します。 上記のプログラム・リストのモデルは、ノー

ドに以下の周波数割当を印刷します。 

 

Node 1: 1 3 5 7  

Node 2: 1 3 5 7 10  

Node 3: 2 8  

Node 4: 4 6 9  

Node 5: 4 6  

Node 6: 4 6 9 11  

Node 7: 2 8 11  

Node 8: 1 3 5 7  

Node 9: 1 3 5  

Node 10: 2 8  

 

3.5 element: 一つの機械でのジョブを順序付ける  

このセクションで説明される問題は、「Applications of optimization with Xpress-MP」という本

の Section 7.4、「Sequencing jobs on a bottleneck machine」から取られたものです。 

この問題の目的は、単一の(ボトルネック)マシンの上で、異なる目的関数を与え、作成されるスケジュ

ールがどのようになるかをみるためのモデルを作成することです。ここで、合計処理時間、平均処理

時間、および、遅延の合計を最小にするにはどうしたらよいか、について考えましょう。 

一組のタスク(または、ジョブ)が、単一の機械で処理されるとします。ここでの作業は、タスクの実行は、

ひとたび開始されたら、中断されることなく、完成されるまで、継続的におこなわれるとします。

Table3.4 に、すべてのタスク i に関して、そのリリース日付、タスク期間、納期が与えられています。 

 

Table 3.4: Task time windows and durations 

Cell 1 2 3 4 5 6 7   

Release date 2 5 4 0 0 8 9   

Duration 5 6 8 4 2 4 2   

Due date 10 21 15 10 5 15 27   

 

以下では、スケジュールの総期間(makespan)、平均処理時間、遅延時間合計を最少にすることを考

え、それぞれの目的のための最適値はどのようなものになるかをみましょう。ここで、遅延時間合計と

は、個々のジョブの完成が、それぞれの納期を越えてしまう時間の合計を意味します。 

3.5.1 モデルの定式化 1 

以下で、二つの代替的なモデルの定式化を説明します。最初の定式化は、「Applications of 

optimization with Xpress-MP」の中の数理計画法の定式化に近いものです。二番目の定式化は、

disjunctive constraints と branching を使います。これら二つのモデル定式化では、順次、異

なる目的関数を使いますが、モデル本体は同一です。 

仕事のシークエンスを表すために、ここで、変数 rankk (KÎJOBS{ }NJ,...,1 )を導入します。 あらゆ

るジョブ j は、ジョブのシークエンスのどかの、ただ一箇所だけに位置します。この制約条件は、変数

rankk の all-different で表すことができます。 

all-different(rank1, . . . , rankNJ) 

位置 k のジョブの処理時間 Durk は krankDUR で与えられます。ここで、Durj は、前のセクションのテーブ

ルで与えられたプロジェクト期間を表します。同様に、リリース時間 relk は、
krankREL で与えられます。

ここで、RELj は、リリース日付を表します。 

 

krankk DURdurJOBSk =Î" :  

krankk RELrelJOBSk =Î" :  

startk を位置 k のジョブの開始時刻であるとするなら、この値は、この位置に割り当てられたジョブのリ

リース日と少なくとも同じか、大きくなくてはなりません。このジョブの完成時刻 compk は、その開始時

刻と、そのジョブの処理時間の合計です。 

 

kk relstartJOBSk ³Î" :  

kkk durstartcompJOBSk +=Î" :  

もう一つ、別の制約条件が必要です。それは、同時に 2 つの仕事を処理できない、という条件です。 

したがって、時間軸上の位置 k+1 のジョブは、ジョブ k が完了した位置以降に始まらなければなりませ

ん。この条件は、下のようになります。 

{ } kkk durstartstartNJk +³-Î" +1:1,...,1  

目的関数 1：目的関数 1 は、makespan(スケジュール全体の完成時間)を最小にすることです。これは、

また、最後のジョブ(NJ 番目のジョブ)の完成時間を最小にすることと同等です。こうして、モデル全体

は、下記のようになります。ここで、MAXTIME は、例えば､すべてのリリース日とすべての処理時間の

合計のような、十分に大きい値です。 

 

minimize  compNJ 

JOBSrankJOBSk k ÎÎ" :  

{ }MAXTIMEcompstartJOBSk kk ,...,0,: ÎÎ"  

{ }jJOBSjjJOBSjk DURDURdurJOBSk ÎÎÎÎ" max,...,min:  

{ }jJOBSjjJOBSjk RElRELrelJOBSk ÎÎÎÎ" max,...,min:  

( )NJrankrankdifferentall ,...,1-  

krankk DURdurJOBSk =Î" :  

krankk RELrelJOBSk =Î" :
 

kk relstartJOBSk ³Î" :  

kkk durstartcompJOBSk +=Î" :
 

{ } kkk durstartstartNJk +³-Î" +1:1,...,1  

目的関数 2：平均処理時間を最小にするために、すべてのジョブの完成時間の合計を表す変数

totComp を追加します。そして、totComp を計算するために、問題に、以下の制約条件を追加します。 

å
Î

=
JOBSk

kcomptotComp  

新しい目的関数は、平均処理時間を最小にすることです。これは、ジョブの処理時間を最小にするこ

とと同等です。 

 

minimize  totComp 

目的関数 3：  

今度は、遅延の合計を最小にする、としましょう。ここでも、新しい変数を導入します。導入する変数は、

納期の後にしか完了しない、ジョブの納期遅れの時間を計算する変数です。変数 latek は、k 番目のジ

ョブの納期遅れの時間です。その値は、納期と完成時刻の差です。 ジョブが納期以前に終わる場合

は、その値はゼロです。 こうして、以下の制約条件を得ます。 

krankk DURdurJOBSk =Î" :  

kkk duecomplateJOBSk -³Î" :  

新しい目的関数の定式化のために、すべてのジョブの総遅延を表す変数 totLate を導入します。 目

的は、この変数の値を最小にすることです。 

 

minimize  totLate 

å
Î

=
JOBSk

klatetotLate  

 

3.5.2 モデル 1のインプリメンテーション 

以下の Mosel インプリメンテーションは、それぞれ、異なる目的関数を使い、同じ問題を、3 回、解き、

結果として得られるソリューションを、手順 print_solおよび､print_solをコールして、プリントしま

す。 

model "B-4 Sequencing (CP)" 

 uses "kalis" 

 

 forward procedure print_sol 

 forward procedure print_sol3 

  

 declarations    

  NJ = 7                          ! Number of jobs 

  JOBS=1..NJ 

 

  REL: array(JOBS) of integer     ! Release dates of jobs 

  DUR: array(JOBS) of integer     ! Durations of jobs 

  DUE: array(JOBS) of integer     ! Due dates of jobs 

 

  rank: array(JOBS) of cpvar      ! Number of job at position k 

  start: array(JOBS) of cpvar     ! Start time of job at position k 

  dur: array(JOBS) of cpvar       ! Duration of job at position k 

  comp: array(JOBS) of cpvar      ! Completion time of job at position k 

  rel: array(JOBS) of cpvar       ! Release date of job at position k 

 end-declarations 

  

 initializations from 'Data/b4seq.dat' 

  DUR REL DUE 

 end-initializations 

 

 MAXTIME:= max(j in JOBS) REL(j) + sum(j in JOBS) DUR(j) 

 MINDUR:= min(j in JOBS) DUR(j); MAXDUR:= max(j in JOBS) DUR(j) 

 MINREL:= min(j in JOBS) REL(j); MAXREL:= max(j in JOBS) REL(j) 

 

 forall(j in JOBS) do 

  1 <= rank(j); rank(j) <= NJ 

  0 <= start(j); start(j) <= MAXTIME  

  MINDUR <= dur(j); dur(j) <= MAXDUR 

  0 <= comp(j); comp(j) <= MAXTIME 

  MINREL <= rel(j); rel(j) <= MAXREL 

 end-do  

 

! One posistion per job 

 all_different(rank) 

 

! Duration of job at position k 

 forall(k in JOBS) dur(k) = element(DUR, rank(k)) 

 

! Release date of job at position k 

 forall(k in JOBS) rel(k) = element(REL, rank(k)) 

  

! Sequence of jobs 

 forall(k in 1..NJ-1) start(k+1) >= start(k) + dur(k) 

! Start times 

 forall(k in JOBS) start(k) >= rel(k) 

 

! Completion times 

 forall(k in JOBS) comp(k) = start(k) + dur(k) 

 

! Set the branching strategy 

 cp_set_branching(split_domain(KALIS_SMALLEST_DOMAIN, KALIS_MIN_TO_MAX)) 

 

!**** Objective function 1: minimize latest completion time **** 

 if cp_minimize(comp(NJ)) then 

  print_sol 

 end-if 

 

 

!**** Objective function 2: minimize average completion time **** 

 declarations    

  totComp: cpvar 

 end-declarations 

  

 totComp = sum(k in JOBS) comp(k) 

 

 if cp_minimize(totComp) then 

  print_sol 

 end-if 

 

 

!**** Objective function 3: minimize total tardiness **** 

 declarations    

  late: array(JOBS) of cpvar      ! Lateness of job at position k 

  due: array(JOBS) of cpvar       ! Due date of job at position k 

  totLate: cpvar 

 end-declarations 

 

 MINDUE:= min(k in JOBS) DUE(k); MAXDUE:= max(k in JOBS) DUE(k) 

 

 forall(k in JOBS) do 

  MINDUE <= due(k); due(k) <= MAXDUE 

  0 <= late(k); late(k) <= MAXTIME 

 end-do  

 

! Due date of job at position k 

 forall(k in JOBS) due(k) = element(DUE, rank(k)) 

  

! Late jobs: completion time exceeds the due date 

 forall(k in JOBS) late(k) >= comp(k) - due(k)  

 

 totLate = sum(k in JOBS) late(k) 

 

 if cp_minimize(totLate) then 

  writeln("Tardiness: ", getsol(totLate)) 

  print_sol 

  print_sol3 

 end-if 

 

!----------------------------------------------------------------- 

 

! Solution printing 

 procedure print_sol    

  writeln("Completion time: ", getsol(comp(NJ)) , 

          "  average: ", getsol(sum(k in JOBS) comp(k))) 

  write("\t") 

  forall(k in JOBS) write(strfmt(getsol(rank(k)),4)) 

  write("\nRel\t") 

  forall(k in JOBS) write(strfmt(getsol(rel(k)),4)) 

  write("\nDur\t") 

  forall(k in JOBS) write(strfmt(getsol(dur(k)),4)) 

  write("\nStart\t") 

  forall(k in JOBS) write(strfmt(getsol(start(k)),4)) 

  write("\nEnd\t") 

  forall(k in JOBS) write(strfmt(getsol(comp(k)),4)) 

  writeln 

 end-procedure 

 

 procedure print_sol3 

  write("Due\t") 

  forall(k in JOBS) write(strfmt(getsol(due(k)),4)) 

  write("\nLate\t") 

  forall(k in JOBS) write(strfmt(getsol(late(k)),4)) 

  writeln 

 end-procedure  

  

end-model 

 

注意： 読者は、なぜ、この例で、変数に名前を付けるとき、もっと、start-end が明白な名前を使わな

いのか、と不思議に思われるかももしれません。既にお分かりの読者の皆さんもいらっしゃるかも知

れませんが、Mosel 言語では、end は、キーワードだからです。（Mosel language reference manual

の reserved word のリストを参照）。これにより、end や END は、Mosel プログラムでキーワードともな

されてしまう可能性があります。もっとも、End のように、大文字、小文字を組み合わせて使うバージョ

ンも可能ですが、起こりえる混乱を回避するために、これらの変数名を使うことを推薦しません。 

 

3.5.3 結果 

スケジュールの最小 makespanは 31 で、最小の完成時間合計は 103 です(これは、平均 103/7=14.71

を与えます)。 この目的関数の値に対応するスケジュールは、ジョブ 5→ ジョブ 4→ジョブ 1→ ジョブ

7→ジョブ 6→ ジョブ 2→ ジョブ 3 です。また、完成時刻と納期を比較すると、ジョブ 1、ジョブ 2、ジョ

ブ 3、および、ジョブ 6 が納期に遅れ、遅れの合計は 21 です。 

最小の遅延は 18 です。 この遅延を持つスケジュールはジョブ 5→ ジョブ 1→ ジョブ 4→ ジョブ 6→ 

ジョブ 2→ ジョブ 7→ ジョブ 3 で、ジョブ 4 とジョブ 7 が、1 タイム・ユニット遅くれ、そして、ジョブ 3 は、

16 タイム・ユニット遅れます。このジョブ 3 は、納期である時刻 15 ではなく、時刻 31 に終わります。こ

のスケジュールの平均完成時間は 15.71 です。 

 

3.5.4 disjunction を使う代替的な定式化 

2 番目のモデル定式化は、多分、問題を、より直接的に表現しているものでしょう。この定式化は、

disjunctive constraint、および、これらの branching を使う方法での定式化です。 

今度は、すべてのジョブを、開始時刻、すなわち、変数 { }( )NJJOBSjstartj ,...,1=Î によって表し、これ

らは、{ＲＥＬｊ,…,ＭＡＸＴＩＭＥ}の中の値を取ります。ここで、MAXTIME は、例えば､すべてのリリース日と

すべての処理時間の合計のような、十分に大きい値です。ここでは、下記のように、disjunction を、

すべてのジョブの開始時刻と処理時間に関する単一の disjunctive relation として表現します。 

[ ] [ ]( )NJNJ DURDURstartstartedisjunctiv ,...,,,..., 11  

この制約条件は、すべてのジョブの組合せ JOBSji Î< に関して、下記の pair-wise disjunction

を置換します。 

ijjjii startDURstartstartDURstart £+Ú£+  

位置 k のジョブの処理時間 durk は、
krankDUR によって与えられます。ここで、DURj は、前のセクション

のテーブルで与えられた処理時間を意味します。同様に、リリース日 relk は、RELrankk(RELj によっ

て与えられます。ここで、
krankREL は、所与のリリース日を意味します。 

 

krankk DURdurJOBSk =Î" :  

krankk RELrelJOBSk =Î" :  

ジョブ j の完成時刻 compj は、開始時刻と処理時間 DURjの合計です。 

kjjj DURstartcompJOBSk +=Î" :  

目的関数 1： 一番目の目的関数は、makespan(スケジュール全体の完成時刻)を最小にすること、言

い換えると、最後のジョブの完成時刻を最も早くすることです。下は、このモデルの全体です。ここで、

MAXTIME は、例えば､すべてのリリース日とすべての処理時間の合計のような、十分に大きい値で

す。 

minimize  finish 

)(max jJOBSj compimumfinish Î=  

{ }MAXTIMEcompJOBSk j ,...,0: ÎÎ"  

{ }MAXTIMERELstartJOBSk jj ,...,: ÎÎ"  

[ ] [ ]( )NJNJ DURDURstartstartedisjunctiv ,...,, 1,...,1  

jjj DURstartcompJOBSj +=Î" :  

目的関数 2： 二番目の目的関数は、最初のモデルの目的関数と同じで、平均処理時間を最小にす

るか、または、それと同等ですが、ジョブの処理時間を最小にすることです。ここで、すべてのジョブの

完成時間の合計を表す変数 totComp を導入します。 

 

minimize  totComp 

å
Î

=
JOBSk

kcomptotComp  

 

目的関数 3：遅延の合計を最小にする目的関数を定式化するために、新しい変数 latej を導入します。

これは、ジョブが期日までに終わらないで遅延するとき、その時間に測定するためのものです。これら

の変数の値は、ジョブ j の完成日と納期 DUEj の差を示すものです。ジョブが、納期以前終わる場合、

その値はゼロです。目的は、これらの遅延変数の合計を最小にすることです。 

Minimize totLate 

å
Î

=
JOBSj

jlatetotLate  

{ }MAXTIMElateJOBSj j ,...,0: ÎÎ"  

jjj DUEcomplateJOBSj -³Î" :  

3.5.5 モデル 2のインプリメンテーション 

モデル 1 と同じように、Mosel インプリメンテーションは、それぞれ、異なる目的関数を使い、同じ問題

を、3回、解き、結果として得られるソリューションを、手順 print_solおよび､print_solをコールし

て、プリントします。 

 

model "B-4 Sequencing (CP)" 

 uses "kalis" 

 

 forward procedure print_sol 

 forward procedure print_sol3 

  

 declarations    

  NJ = 7                          ! Number of jobs 

  JOBS=1..NJ 

 



 53 

  REL: array(JOBS) of integer     ! Release dates of jobs 

  DUR: array(JOBS) of integer     ! Durations of jobs 

  DUE: array(JOBS) of integer     ! Due dates of jobs 

  DURS: array(set of cpvar) of integer  ! Dur.s indexed by start variables 

 

  start: array(JOBS) of cpvar     ! Start time of jobs 

  comp: array(JOBS) of cpvar      ! Completion time of jobs 

  finish: cpvar                   ! Completion time of the entire schedule 

  Disj: set of cpctr              ! Disjunction constraints  

  Strategy: array(range) of cpbranching  ! Branching strategy 

 end-declarations 

  

 initializations from 'Data/b4seq.dat' 

  DUR REL DUE 

 end-initializations 

 

 MAXTIME:= max(j in JOBS) REL(j) + sum(j in JOBS) DUR(j) 

 

 forall(j in JOBS) do 

  0 <= start(j); start(j) <= MAXTIME  

  0 <= comp(j); comp(j) <= MAXTIME 

 end-do  

 

! Disjunctions between jobs 

 forall(j in JOBS) DURS(start(j)):= DUR(j) 

 disjunctive(union(j in JOBS) {start(j)}, DURS, Disj, 1) 

 

! Start times 

 forall(j in JOBS) start(j) >= REL(j) 

 

! Completion times 

 forall(j in JOBS) comp(j) = start(j) + DUR(j) 

 

!**** Objective function 1: minimize latest completion time **** 

 finish = maximum(comp) 

 

 Strategy(1):= settle_disjunction(Disj) 

 Strategy(2):= split_domain(KALIS_LARGEST_MAX, KALIS_MIN_TO_MAX) 

 cp_set_branching(Strategy) 

  

 if cp_minimize(finish) then 

  print_sol 

 end-if 

 

 

!**** Objective function 2: minimize average completion time **** 

 declarations    

  totComp: cpvar 

 end-declarations 

 totComp = sum(k in JOBS) comp(k) 

 

 if cp_minimize(totComp) then 

  print_sol 

 end-if 

 

 

!**** Objective function 3: minimize total tardiness **** 

 declarations    

  late: array(JOBS) of cpvar      ! Lateness of jobs 

  totLate: cpvar 

 end-declarations 

 

 forall(k in JOBS) do 

  0 <= late(k); late(k) <= MAXTIME 

 end-do  

  

! Late jobs: completion time exceeds the due date 

 forall(j in JOBS) late(j) >= comp(j) - DUE(j)  

 

 totLate = sum(k in JOBS) late(k) 

 if cp_minimize(totLate) then 

  writeln("Tardiness: ", getsol(totLate)) 

  print_sol 

  print_sol3 

 end-if 

 

!----------------------------------------------------------------- 

 

! Solution printing 

 procedure print_sol 

  writeln("Completion time: ", getsol(finish) , 

          "  average: ", getsol(sum(j in JOBS) comp(j))) 

  write("Rel\t") 

  forall(j in JOBS) write(strfmt(REL(j),4)) 

  write("\nDur\t") 

  forall(j in JOBS) write(strfmt(DUR(j),4)) 

  write("\nStart\t") 

  forall(j in JOBS) write(strfmt(getsol(start(j)),4)) 

  write("\nEnd\t") 

  forall(j in JOBS) write(strfmt(getsol(comp(j)),4)) 

  writeln 

 end-procedure 

 

 procedure print_sol3 

  write("Due\t") 

  forall(j in JOBS) write(strfmt(DUE(j),4)) 

  write("\nLate\t") 

  forall(j in JOBS) write(strfmt(getsol(late(j)),4)) 

  writeln 

 end-procedure  

  

end-model 

このインプリメンテーションでは、settle_disjunction という新しい branching scheme を導入しまし

た。これまで見てきた branching 戦略とは異なり、この branching scheme は、変数ではなく、制約

式についてのbranching戦略を定義します。このschemeにより、与えられた集合から制約条件を選ら

んでノードを作成し、そのノードからの branch は、この disjunctive constraint を形成する、互い

に排他的な制約条件の一つを、この constraint system に加えることで得られます。 

注意： Xpress-Kalis の disjunctive constraint により、タスクの処理時間の間の pair-wise 

inequalities が確立されます。しかし、disjunctive constraint の定義は、このケースにとどまりま

せん。disjunction は、2 つ以上のコンポーネントも持つことができ、どんなタイプの制約条件でも構

いません。これには、'and'や'or'によって制約条件を結合することで得られる他のロジック関係も含

くまれています。 

3.6 occurrence: 砂糖の生産 gar production  

このセクションの問題は、「Applications of optimization with Xpress-MP」という本の Section 6.4

「Cane sugar production」から取ったものです。 

オーストラリアでの甘蔗糖の収穫は、非常に機械化されています。 サトウキビはすぐに、狭軌道のワ

ゴンで、製糖所に輸送されます。 

ワゴン 1 台分の積荷のサトウキビから取れる砂糖は、収穫された畠とサトウキビの成熟度合いに依存

しています。ひとたび、収穫されると、糖分は、発酵により急速に減少してゆきます。したがって、そし

て、ワゴン 1 台分の積み荷は、ある時間が経つと、その価値が完全になくなります。いま、同じ量を積

んだ11台のワゴンが製糖所に到着しました。これらのワゴンは、ワゴンごとに、1時間経過ごとの糖分

の損失と、時間数で計られる「残っている寿命」を見定めるために調べられました。以下のテーブルは、

これらのデータをまとめたものです。 

 

Table 3.5: Properties of the lots of cane sugar 

Lot 1 2 3 4 5 6 7 8 9 10 11 

Loss (kg/h)  43 26 37 28 13 54 62 49 19 28 30 

Life span (h)  8 8 2 8 4 8 8 8 8 8 8 

これらの 11 のロットは、すべて、三本ある、まったく同等の製造ラインによって処理されます。 一つの

ロット処理するには 2 時間かかります。そして、処理は、遅くとも、ワゴンの積み荷の寿命が終わるま

でに終了していなければなりません。 

製糖所のマネジャーは、現在、手許にある、処理できるロットを、どのようなスケジュールで処理したら、

砂糖のロスを最小にすることができるかを知りたいと思っています。 

 

3.6.1 モデルの定式化 

WAGONS={1,…,NW}でワゴンの集合を、NL で生産ラインの数を、DUR で、すべてのロットの処理時間を

表すものとします。そして、各ワゴン w の、1 時間ごとのロスは LOSSw、そして、寿命は LIFEwだとします。

最適解では、製造ラインは中断なしで動くことが必要です。なぜなら、中断しなければ、ロットの始まり

を遅らせなくてすむので、それだけ、砂糖のロスを抑えることができるからです。これは、あらゆるロッ

トの処理の完了時間が s*DUR という形式のものであることを意味します。ここで、s > 0 であり、整数で

す。s の取る最大の値は、製糖所が稼動する、長さ DUR の時間帯の数、すなわち、NS = ceil(NW/NL)

です。ここで、ceil は、「rounded to the next largest integer」を意味します。もし、NW/NL が

整数であるなら、ラインは、すべて、NS ロットだけ処理します。そうでない場合は、いくつかのラインは、

NS-1 ロットを処理することになりますが、少なくとも 1 つのラインは NS ロットを処理します。 すべての

場合で、最適スケジュールの長さは NS*DUR 時間です。ここでは、SLOTS={1,...,NS}を「時間帯の集合

（set of time slots）」と呼びましょう。 

ロットは、すべて、どこかの時間帯に割り当てられなければなりません。ここで、ワゴン w に割り当てら

れた時間帯を表す変数 processw を、このワゴンで発生するロスを表す変数 lossw を定義します。他

方、NL 本の並列ラインがあるので、製糖所が稼動中の各時間帯には、すべて、最大 NL 個のロットを

割り当てられます。したがって、processw 変数で、時間帯での出現回数を制限します(this 

constraint relation is often called cardinality constraint)。 

NLsprocessSLOTSs WAGONSww £=Î Î|:|  

時間帯 s で処理されるロットのワゴン一台分の砂糖のロス w は、COSTws = s * DUR * LOSSw で

す。ここで、変数 losswでワゴン wに積まれているロットで発生するロスを示すものとします。 

wprocessww tlossWAGONSw ,cos: =Î"  

目的関数（砂糖のロス合計）は、すべてのロスの合計となります。 

minimize å
ÎWAGONSw

wloss  

 

3.6.2 インプリメンテーション 

以下のモデルは、この問題のMoselインプリメンテーションです。このモデルは、時間帯の最大数の計

算に function ceil を使っています。 

The constraints on the processing variables are expressed by occurrence relations and the losses are 

obtained via element constraints. The branching strategy uses the variable selection criterion 

KALIS_SMALLEST_MAX, that is, choosing the variable with the smallest upper bound. 

処理の変数についての制約条件は、occurrence relations で表現されています。そして、ロスは

element constraint を 使 っ て 表 現 さ れ て い ま す 。 branching strategy は 、 変 数 選 択 基 準

KALIS_SMALLEST_MAX を使っています。KALIS_SMALLEST_MAX は、最も小さい upper bound を選択しま

す。 

 

model "A-4 Cane sugar production (CP)" 

 uses "kalis" 

  

 declarations 

  NW = 11                            ! Number of wagon loads of sugar 

  NL = 3                             ! Number of production lines 

  WAGONS = 1..NW 

  NS = ceil(NW/NL) 

  SLOTS = 1..NS                      ! Time slots for production 

   

  LOSS: array(WAGONS) of integer     ! Loss in kg/hour 

  LIFE: array(WAGONS) of integer     ! Remaining time per lot (in hours) 

  DUR: integer                       ! Duration of the production (in hours) 

  COST: array(SLOTS) of integer      ! Cost per wagon 

   

  loss: array(WAGONS) of cpvar       ! Loss per wagon 

  process: array(WAGONS) of cpvar    ! Time slots for wagon loads 

 

  totalLoss: cpvar                   ! Objective variable 

 end-declarations 

  

 initializations from 'Data/a4sugar.dat' 

  LOSS LIFE DUR 

 end-initializations 

 

 forall(w in WAGONS) setdomain(process(w), 1, NS) 

 

! Wagon loads per time slot 

 forall(s in SLOTS) occurrence(s, process) <= NL 

 

! Limit on raw product life 

 forall(w in WAGONS) process(w) <= floor(LIFE(w)/DUR) 

  

! Objective function: total loss   

 forall(w in WAGONS) do 

  forall(s in SLOTS) COST(s):= s*DUR*LOSS(w) 

  loss(w) = element(COST, process(w)) 

 end-do 

 totalLoss = sum(w in WAGONS) loss(w) 

 

 cp_set_branching(assign_var(KALIS_SMALLEST_MAX, KALIS_MIN_TO_MAX, process)) 

 

! Solve the problem 

 if not (cp_minimize(totalLoss)) then 

  writeln("No solution found") 

  exit(0) 

 end-if 

 

! Solution printing 

 writeln("Total loss: ", getsol(totalLoss)) 

 forall(s in SLOTS) do  

  write("Slot ", s, ": ") 

  forall(w in WAGONS)    

   if(getsol(process(w))=s) then 

    write("wagon ", strfmt(w,2), strfmt(" (" + s*DUR*LOSS(w) + ")  ", 8)) 

   end-if  

  writeln 

 end-do 

 

end-model 

処理変数についての制約条件の代替的な定式化は、それらを、単一の distribute relation で置

き換えることです。ここでの distribute relationは、すべての時間帯で、ラインが使える最少、

最大の数(MINUSEs=0 and MAXUSEs=NL)を知らせるのに使われます。 

forall(s in SLOTS) MAXUSE(s):= NL 

distribute(process, SLOTS, MINUSE, MAXUSE) 

Xpress-Kalis では、この問題を、さらにもう一つ、別の形で定式化することもできます。すなわち、この

問題を cumulative scheduling problem(see Section 5.4)と解釈することです。そこでは、ワ

ゴンに積まれたロットは、キャパシティが生産ラインに対応する discrete resource 上にスケ

ジュールされる tasks of unit duration で表されると考えます。 

 

3.6.3 結果 

ロスの合計は、1620 kg の砂糖です。それに対応するスケジュールは、下の Table 3.6 に示さ

れています。 

 

Table 3.6: Optimal schedule for the cane sugar lots 

Slot 1 Slot 2 Slot 3 Slot 4 

lot 3 (74 kg)  lot 1 (172 kg) lot 4 (168 kg) lot 2 (208 kg) 

lot 6 (108 kg) lot 5 (52 kg) lot 9 (114 kg)   lot 10 (224 kg) 

lot 7 (124 kg) lot 8 (196 kg) lot 11 (180 kg)   

 

3.7 distribute: 人員配置 

映画館の経営者は、従業員の働く場所のプランを立てたいと思っています。劇場には、デヴィッド、ア

ンドリュー、レスリー、ジェイソン、オリバー、マイケル、ジェーン、および、マリリンという 8 人の従業員

がいます。切符売場は、3 人の配置が、劇場の入口 1 と入口 2 は、それぞれ、2 人の配置がが必要で

す。さらに、1 人をクロークにも配置しなければなりません。従業員のスキルや個々人の好みも考えな

ければならないので、以下の条件を考慮する必要があります。 

 

1. レスリーは、劇場の 2 番目の入り口に配置しなければならない。 

2. マイケルは、劇場の 1 番目の入り口に配置しなければならない。 

3. デヴィッド、マイケル、および、ジェイソンは、お互いに、一緒に働けない。 

4. オリバーをチケット売場に配置すると、マリリンも、そこに配置しなければならない。 

3.7.1 Model formulation 

PERS を人員の集合とし、LOC = {1,...,4}を働く位置の集合とします。ここで、1 は切符売場、2 は

入口 1、3 はは入口 3、4 はクロークを表すとします。 

ここで、従業員 p に割り当てられる場所を示す決定変数 placep を導入します。こうして、以下のように、

働く位置についての 4 つの制約条件を定式化できます。 

3=Leslieplace  

2=Michaelplace  

( )JasonMichaelDavid placeplaceplacedifferentall ,,-  

11 =Þ= MarylinOliver placeplace  

また、働く位置、すべての配置人員の条件も満たさなければなりません。 

lPERSpp REQlplacel ==Î"
Î

:  

 

3.7.2 インプリメンテーション 

この配置問題のインプリメンテーションには、Xpress-Kalis の持つ、2 つの異なる constraint から一

つを選べます。すなわち、(配置場所ごとに一つの)occurrence constraint、もしくは、(global 

cardinality constraint として知られている)distribute constraint です。後者は、すべての

職場を対象とするものです。以下のモデルは、これらの両方のオプションを示します。セクション

3.6.2 の例とは対照的に、このモデルには、equality constraint があります。したがって、ここでは、

4 つのアーギュメントを持つ distribute ではなく、3 つのアーギュメントの distribute バージョンを

使います。4つのアーギュメントを持つdistributeでは、最後の2のアーギュメントは、lower bound、

upper bound です。 

このモデルは、attractive display のために、集合 LOC の中の配置場所の名前、および、それに

対応するインデックス値についての補助的なデータ構造も導入してあります。 

 

model "Personnel Planning (CP)" 

 uses "kalis" 

  

 forward procedure print_solution  

 

 declarations 

  PERS = {"David","Andrew","Leslie","Jason","Oliver","Michael", 

          "Jane","Marilyn"}                ! Set of personnel 

  LOC = 1..4                               ! Set of locations 

  LOCNAMES = {"Ticketoffice", "Theater1", "Theater2",  

              "Cloakroom"}                 ! Names of locations 

  LOCNUM: array(LOCNAMES) of integer       ! Numbers assoc. with loc.s 

  REQ: array(LOC) of integer               ! No. of pers. req. per loc. 

 

  place: array(PERS) of cpvar              ! Workplace assigned to each peson 

 end-declarations 

 

! Initialize data 

 LOCNUM("Ticketoffice"):= 1; LOCNUM("Theater1"):= 2 

 LOCNUM("Theater2"):= 3; LOCNUM("Cloakroom"):= 4 

 REQ:: (1..4)[3, 2, 2, 1] 

 

! Each variable has a lower bound of 1 (Ticketoffice) and an upper bound  

! of 4 (Cloakroom) 

 forall(p in PERS) do 

  setname(place(p),"workplace["+p+"]") 

  setdomain(place(p), LOC) 

 end-do 

 

! "Leslie must be at the second entrance of the theater" 

 place("Leslie") = LOCNUM("Theater2") 

 

! "Michael must be at the first entrance of the theater" 

 place("Michael") = LOCNUM("Theater1") 

 

! "David, Michael and Jason cannot work with each other" 

 all_different({place("David"), place("Michael"), place("Jason")}) 

 

! "If Oliver is selling tickets, Marylin must be with him" 

 implies(place("Oliver")=LOCNUM("Ticketoffice"), 

         place("Marilyn")=LOCNUM("Ticketoffice")) 

   

! Creation of a resource constraint of for every location 

! forall(d in LOC) occurrence(LOCNUM(d), place) = REQ(d) 

 

! Formulation of resource constraints using global cardinality constraint 

 distribute(place, LOC, REQ) 

 

! Setting parameters of the enumeration 

 cp_set_branching(assign_var(KALIS_SMALLEST_MIN, KALIS_MAX_TO_MIN, place)) 

 

! Solve the problem 

 if not(cp_find_next_sol) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

  

! Solution output 

 nbSolutions:= 1  

 print_solution 

 

! Search for other solutions 

 while (cp_find_next_sol) do 

  nbSolutions += 1 

  print_solution 

 end-do 

  

! **** Solution printout ****  

 procedure print_solution  

  declarations 

   LOCIDX: array(LOC) of string 

  end-declarations 

  forall(l in LOCNAMES) LOCIDX(LOCNUM(l)):=l 

   

  writeln("\nSolution number ", nbSolutions) 

  forall(p in PERS) 

   writeln(" Working place of ", p, ": ", LOCIDX(getsol(place(p)))) 

 end-procedure 

 

end-model 

 

ここでは、単に、一つの実行可能なソリューションを見つけたいので、このモデルは、最初に、実行可

能なソリューションの有無をテストします。実行可能なソリューションがある場合は、モデルは、見つか

ったソリューション以外の、他のすべての実行可能なソリューションを列挙します。 ソリューションが見

つかるたびに、procedure print_solution をコールして、それをプリントします。 

3.7.3 結果 

この問題には、38 の実現可能なソリューションがあります。 実行可能なソリューションを、IVE を使っ

てグラフ表示させると、Figure3.2 のようになります。 

 

 

Figure 3.2: Personnel schedule representation in IVE 

 

このグラフィック表示は、以下の Mosel コードを使って表示させたものです。詳しくは、Mosel language 

reference manual のモジュール mmive についての説明を参照してください。 

 

 procedure draw_solution  

  IVEerase 

  PersGraph:= IVEaddplot("Personnel", IVE_BLUE) 

  LocGraph:= IVEaddplot("Locations", IVE_CYAN) 

  AsgnGraph:= IVEaddplot("Assignments", IVE_BLACK) 

  

  forall (d in LOCNAMES) 

   IVEdrawlabel(LocGraph, 1.2, LOCNUM(d)-getsize(LOC)/2, d) 

 

  idx:= 1 

  forall(p in PERS) do 

   IVEdrawline(AsgnGraph, 0, idx-getsize(PERS)/2, 

               1, getsol(place(p))-getsize(LOC)/2) 

   IVEdrawlabel(PersGraph, -0.1, idx-getsize(PERS)/2, p) 

   idx:= idx + 1 

  end-do 

  IVEzoom(-1, getsize(PERS)/2+1, 2, -getsize(PERS)/2) 

  IVEpause("press a key to continue") 

 end-procedure 

 

3.8 implies : ペイントの生産 

このセクションの問題は、「Applications of optimization with Xpress-MP」という本の Section 

7.5 にある「Paint production」から取ったものです。 

毎週の生産の一部として、あるペイント会社は、安定した需要を持っている大きいクライアントに納入

するため、このペイントの 5 つのバッチを生産しています。この 5 つのバッチは、いつも同じです。ペイ

ントのバッチは、すべて、単一の、同じ生産プロセスで生産されます。生産に使う混合槽は、あるバッ

チから次のバッチに移るとき、必ず、洗浄されなければなりません。ペイントの 5 つの混合バッチ 1～5

の所要時間は、それぞれ 40 分、35 分、45 分、32 分、および 50 分です。 洗浄時間は、色とペイント

のタイプに依存します。例えば、油ベースのペイントを水性ペイントの後に生産したり、暗い色のペイ

ントの生産の後に、白いペイントを生産したりするような場合、洗浄時間は長くかかります。Table 3.7: 

Matrix of cleaning times の数字は、分単位で、CLEANij は、バッチ i とバッチ j との間の洗浄時

間です。 

 

Table 3.7: Matrix of cleaning times   

  1 2 3 4 5 

1 0 11 7 13 11 

2 5 0 13 15 15 

3 13 15 0 23 11 

4 9 13 5 0 3 

5 3 7 7 7 0 

 

この会社は、他の仕事も持っているので、毎週行う、この生産活動（戦場 TOC 調合）を、できるだけ短

い時間で済ませたいと考えています。そうするためには、ペイントバッチを、どのような順序で行ったら

よいでしょうか。この順序は、毎週、行われるので、総処理時間を最短にするには、ある週の最後の

バッチと次の週の最初のバッチの洗浄時間も考慮する必要があります。 

 

 

3.8.1 モデル 1の定式化 

ここでは、Section 3.5 の問題を、2 つの代替的なモデルとして定式化しましょう。最初の定式化は、

「Applications of optimization with Xpress-MP」の数理計画法による定式化に近いもので、二



 64 

番目の定式化は、two-dimensional element constraint を使用する定式化です。 

JOBS = {1,...,NJ}で生産するバッチの集合を、DURjでバッチ j の処理時間を、CLEANijで二つ

の連続したバッチ iとバッチ jの間の洗浄時間を示すものとします。また、各ジョブの後続ジ

ョブを示すため、JOBS の中の値を取る決定変数 succjを導入し、各ジョブの後の洗浄に必要な

時間を示す cleanj を導入します。各ジョブの後の洗浄に必要な時間は、succj の値でインデッ

クスして得ます。 こうして、以下のように、問題の定式化を行います。 

Minimize ( )å
Î

+
JOBSj

jj cleanDUR  

{ }j\: JOBSsuccJOBSj jÎ"  

jsuccjj CLEANcleanJOBSj ,: =Î"  

( )jJOBSj succdifferentall Î- U  

目的関数は、すべてのバッチの処理時間と洗浄時間をまとめます。最後のall-different制約条

件は、あらゆるバッチが、生産順序の中で一回しか起こらないことを保証します。 

残念ながら、このモデルは、ソリューションが一つのサイクルを形成することを保証しません。

実際、モデルを解くと、総作業時間239が得られますが、その中には、2つのサブサイクル1→ 3 

→ 2 → 1とサブサイクル4 → 5 → 4があり、ソリューションとして、適切ではありません。

これを解決するための一つの方法は、モデルにdisjunctionを加え、サブサイクルをもたないソ

リューションを得るまで、繰り返し、モデルを解くことです。 

45123 51231 ¹Ú¹Ú¹Ú¹Ú¹" succsuccsuccsuccsucc  

しかし、この手順は、潜在的に非常に多くの disjunction を導入する可能性があるので、より

大きいデータセットでは非実用的でしょう。したがって、サブサイクル除去のために、下記の

ように、変数 yjと implication constraint を使い、先験的な定式化の方式を選びます。 

{ }NJrankJOBSj j ,...,1: ÎÎ"  

1:,,...,2, +=Þ=¹="Î" iji yyjsuccjiNJjJOBSi  

変数 yjは、生産サイクルの中でのジョブ j の位置に対応します。これらの制約条件で、ジョブ

1 は、いつも第 1ポジションを取ります。 

 

3.8.2 モデル 1のインプリメンテーション 

前のセクションで定式化されたモデルのMoselインプリメンテーションは、かなり簡単です。サブサイク

ル除去の制約条件は、impliesを持つロジック関係として実行されます。(equivを使い、implication

をequivalenceに置き換えることで、これらの制約条件の、より強い定式化が得られます。） 

model "B-5 Paint production (CP)" 

 uses "kalis" 

 

 declarations 

  NJ = 5                             ! Number of paint batches (=jobs) 

  JOBS=1..NJ 

 

  DUR: array(JOBS) of integer        ! Durations of jobs 

  CLEAN: array(JOBS,JOBS) of integer ! Cleaning times between jobs 

  CB: array(JOBS) of integer         ! Cleaning times after a batch 

 

  succ: array(JOBS) of cpvar         ! Successor of a batch 

  clean: array(JOBS) of cpvar        ! Cleaning time after batches 

  y: array(JOBS) of cpvar            ! Variables for excluding subtours 

  cycleTime: cpvar                   ! Objective variable 

 end-declarations 

 

 initializations from 'Data/b5paint.dat' 

  DUR CLEAN  

 end-initializations 

  

 forall(j in JOBS) do 

  1 <= succ(j); succ(j) <= NJ; succ(j) <> j 

  1 <= y(j); y(j) <= NJ 

 end-do 

 

! Cleaning time after every batch 

 forall(j in JOBS) do 

  forall(i in JOBS) CB(i):= CLEAN(j,i) 

  clean(j) = element(CB, succ(j)) 

 end-do  

  

! Objective: minimize the duration of a production cycle 

 cycleTime = sum(j in JOBS) (DUR(j)+clean(j)) 

 

! One successor and one predecessor per batch 

 all_different(succ) 

 

! Exclude subtours 

 forall(i in JOBS, j in 2..NJ | i<>j)  

  implies(succ(i) = j, y(j) = y(i) + 1) 

 

! Solve the problem 

 if not cp_minimize(cycleTime) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

 cp_show_stats 

 

! Solution printing 

 writeln("Minimum cycle time: ", getsol(cycleTime)) 

 writeln("Sequence of batches:\nBatch Duration Cleaning") 

 first:=1  

 repeat 

  writeln("  ", first, strfmt(DUR(first),8), strfmt(getsol(clean(first)),9)) 

  first:=getsol(succ(first)) 

 until (first=1) 

 

end-model 

 

3.8.3 モデル 2の定式化 

Section 3.5.1 の順序付けモデルに似た rank variable を使って、このペイント生産の問題を実行するこ

ともできます。 

前と同様、JOBS = {1,...,NJ }が生産バッチの集合、DURj がバッチ j の処理時間、CLEANij が連続したバ

ッチ i とバッチ j の間の洗浄時間を示すとします。また、位置 k にある job の数値として、JOBS の中の

値を取る決定変数 rankk を導入します。変数 cleank (ここで、kÎJOBS) は、k 番目の洗浄時間を示し

ます。この洗浄時間は、2 の連続した変数 rankk の値で CLEANij をインデクシングして得られます。 

こうして、以下の問題の定式化を得ます。 

 

Minimize åå
ÎÎ

+
JOBSk

k
JOBSj

j cleanDUR  

JOBSrankJOBSk k ÎÎ" :  

{ }
1,:1,...,1

+
=-Î"

kk rankrankk CLEANcleanNJk  

1,rankrankNJ NJ
CLEANclean =  

( )kJOBSk rankdifferentall Î- U  

モデル 1 と同様、目的関数は、すべてのバッチの処理と洗浄時間の合計の最小化です。数学的観点

からは厳密には必要ではありませんが、処理時間と洗浄時間をそれぞれ別個に、異なる集計インデ

ックで集計しています。これにより、ジョブに関する集計と、ジョブの位置に関する集計との違いを示し

ます。 

このモデルでは、rank variable に関する all-different constraint を組み込み、すべてのバッチが、生産

順序の中に、厳密に、ただ一度だけしか入らないようにしてあります。 

 

3.8.4 2 のインプリメンテーション 

二番目のモデルのインプリメンテーションは、Xpress-Kalis の 2-dimensional version of the element 

constraint を使用しています。 

 

model "B-5 Paint production (CP)" 

 uses "kalis" 

 

 declarations    

  NJ = 5                             ! Number of paint batches (=jobs) 

  JOBS=1..NJ 

 

  DUR: array(JOBS) of integer        ! Durations of jobs 

  CLEAN: array(JOBS,JOBS) of integer ! Cleaning times between jobs 

 

  rank: array(JOBS) of cpvar         ! Number of job in position k 

  clean: array(JOBS) of cpvar        ! Cleaning time after batches 

  cycleTime: cpvar                   ! Objective variable 

 end-declarations 

 

 initializations from 'Data/b5paint.dat' 

  DUR CLEAN  

 end-initializations 

  

 forall(k in JOBS) setdomain(rank(k), JOBS)  

 

! Cleaning time after every batch 

 forall(k in JOBS) 

  element(CLEAN, rank(k), if(k<NJ, rank(k+1), rank(1))) = clean(k) 

  

! Objective: minimize the duration of a production cycle 

 cycleTime = sum(j in JOBS) DUR(j) + sum(k in JOBS) clean(k) 

 

! One position for every job 

 all_different(rank) 

 

! Solve the problem 

 if not cp_minimize(cycleTime) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

 cp_show_stats 

 

! Solution printing 

 writeln("Minimum cycle time: ", getsol(cycleTime)) 

 writeln("Sequence of batches:\nBatch Duration Cleaning") 

 forall(k in JOBS) 

  writeln("  ", getsol(rank(k)), strfmt(DUR(getsol(rank(k))),8),  

          strfmt(getsol(clean(k)),9)) 

 

end-model 

 

3.8.5 結果 

この問題の最小のサイクルタイムは 243 分です。これは、1 → 4 → 3 → 5 → 2 → 1 の順序でバ

ッチを処理することで達成できる値です。また、243 分は、(圧縮できない)処理時間 202 分と 41 分の洗

浄時間からなっています。 

この問題の Xpress-Kalis によって作成された問題についての統計を見ると、この二番目のモデルは、

(ディフォルト戦略を使用して)かなり長い列挙を行っているので、「弱い定式化」であることがわかりま

す。 

3.9 equiv:所得税の税務署の配置場所 

この例は、「Applications of optimization with Xpress-MP」という本の Section 15.5 にある「Location of 

income tax offices」から取ったものです。 

所得税担当の役所は、管轄区域の所得税徴収オフィスのネットワークを再構築を計画しようとしてま

す。ここで、各都市の住民の数と、すべての都市間の距離がわかっています(Table Table 3.8)。所得

税担当事務所は、十分な範囲をカバーするためには、3 つの都市にオフィスを設置すべきである、と

いう決定を行っています。これらのオフィスを、住民一人あたりの、最も近い所得税オフィスまでの平

均距離を最小にするためには、オフィスをどこに設置すべきでしょうか。 

 

Table 3.8: Distance matrix and population of cities   

  1 2 3 4 5 6 7 8 9 10 11 12 

1 0 15 37 55 24 60 18 33 48 40 58 67 

2 15 0 22 40 38 52 33 48 42 55 61 61 

3 37 22 0 18 16 30 43 28 20 58 39 39 

4 55 40 18 0 34 12 61 46 24 62 43 34 

5 24 38 16 34 0 36 27 12 24 49 37 43 

6 60 52 30 12 36 0 57 42 12 50 31 22 

7 18 33 43 61 27 57 0 15 45 22 40 61 

8 33 48 28 46 12 42 15 0 30 37 25 46 

9 48 42 20 24 24 12 45 30 0 38 19 19 

10 40 55 58 62 49 50 22 37 38 0 19 40 

11 58 61 39 43 37 31 40 25 19 19 0 21 

12 67 61 39 34 43 22 61 46 19 40 21 0 

Pop. (in 1000) 15 10 12 18 5 24 11 16 13 22 19 20 

 

3.9.1 モデルの定式化 

CITIES を都市の集合であるとしましょう。問題の定式化には、2 つのグループの決定変数が必要です。

すなわち、変数 buildc と変数 dependc です。変数 buildc は、 都市 c に所得税徴収オフィスが設置さ

れた場合、そして、その場合だけ 1 という値をとります。変数 dependc は、都市 c を管轄とするオフィス

の番号を取ります。制約条件の定式化のために、さらに、2 つの補助変数の集合、すなわち、

depdistc、および、numdepcを導入します。depdistcは、都市 cから dependcで示されるオフィスまでの

距離を示し、numdepc は、徴収オフィスが置かれる位置によって決まってくる管轄都市の数です。 

以下の関係が、変数 dependc に変数 buildc をリンクするのに必要です。 

(1) numdepcは、変数dependcの中でのoffice location cの発生回数をカウントします。 

(2)、都市 c にオフィスが設置される場合、そして、その場合にだけ、numdepc ³ 1 となります(結果とし

て、都市 c にオフィスが設置されない場合は、numdepc は=0)。 

CITIESddc cdependnumdepCITIESc
Î

==Î" :  

11: =Û³Î" cc buildnumdepCITIESc  

設置されるオフィスの数が所与の NUMLOC で制限されているので、下記の式を得ます。 

å
Î

£
CITIESc

c NUMLOCbuild  

変数 buildc と変数 dependc の間の二番目の関係の定式化は、実際には、implication「もし、

numdepc³ 1 であるならば、そうすれば、都市 c のオフィスを設置しなければならない。そして、逆に、都

市 c のオフィスが設置されないならば、numdepc=0 でなければならない」とすれば十分でしょう。 

最小にされるべき目的関数は、都市の住民の数によって重みづけされた総距離です。住民一人あた

りの、最も近い所得税オフィスへの平均距離を得るには、結果として得られる値を、区域の総人口で

割る必要があります。都市 c から最も近い所得税徴収オフィスの位置までの距離 depdistc は、

discrete function、すなわち、dependc の値でインデックスされた距離マトリクス DISTcd の行 c で得られ

ます。 

cdependcc DISTdepdist ,=  

以上より、以下の CP モデルを得ます。 

minimize å
Î

×
CITIESc

cc distPOP  

{ } ,,1,0: CITIESdependbuildCITIESc cc ÎÎÎ"  

{ } { }dcCITIESddcCITIESdcc DISTDISTdepdistCITIESnumdep ,,...,, maxmin,0 ÎÎÎÎ U  

cdependcc DISTdepdistCITIESc ,: =Î"  

NUMLOCbuild
CITIESc

c £å
Î

 

CITIESddc cdependnumdepCITIESc
Î

==Î" :  

11: =Û³Î" cc buildnumdepCITIESc  

3.9.2 インプリメンテーション 

この問題を解くために、2 つの部分からなる branching strategy を定義します。その一つは、変数

buildc のためのも、そして、もう一つは、変数 depdistc のためのものです。後者は、split_domain 

branching scheme を使って列挙されます。この branching scheme は、(値を変数に割り当てるのでは

なく）branching variable のドメインを数個の disjoint subsets に分割します。そして、アーギュメントと

して、procedure cp_set_branching に、type cpbranching の array を渡します。このアレイの中のオ

ーダーで、異なる戦略が適用されるます。ここでの列挙の戦略は、問題のすべての決定変数を明示

的に含んではいないので、Xpress-Kalis は、サーチ戦略を適用した後に、アサインされていない変数

が残っている場合は、ディフォルト戦略を使ってこれらを列挙すします。 

 

model "J-5 Tax office location (CP)" 

 uses "kalis" 

 

 forward procedure calculate_dist 

 

 setparam("DEFAULT_LB", 0) 

 

 declarations 

  NC = 12 

  CITIES = 1..NC                        ! Set of cities 

 

  DIST: array(CITIES,CITIES) of integer ! Distance matrix 

  POP: array(CITIES) of integer         ! Population of cities 

  LEN: dynamic array(CITIES,CITIES) of integer ! Road lengths 

  NUMLOC: integer                       ! Desired number of tax offices 

  D: array(CITIES) of integer           ! Auxiliary array used in constr. def. 

   

  build: array(CITIES) of cpvar         ! 1 if office in city, 0 otherwise 

  depend: array(CITIES) of cpvar        ! Office on which city depends 

  depdist: array(CITIES) of cpvar       ! Distance to tax office 

  numdep: array(CITIES) of cpvar        ! Number of depending cities per off. 

  totDist: cpvar                        ! Objective function variable 

  Strategy: array(1..2) of cpbranching  ! Branching strategy 

 end-declarations 

 

 initializations from 'Data/j5tax.dat' 

  LEN POP NUMLOC 

 end-initializations 

! Calculate the distance matrix 

 calculate_dist     

 

 forall(c in CITIES) do 

  build(c) <= 1 

  1 <= depend(c); depend(c) <= NC 

  min(d in CITIES) DIST(c,d) <= depdist(c)  

  depdist(c) <= max(d in CITIES) DIST(c,d) 

  numdep(c) <= NC 

 end-do 

 

! Distance from cities to tax offices 

 forall(c in CITIES) do 

  forall(d in CITIES) D(d):=DIST(c,d) 

  element(D, depend(c)) = depdist(c) 

 end-do 

 

! Number of cities depending on every office 

 forall(c in CITIES) occurrence(c, depend) = numdep(c)  

 

! Relations between dependencies and offices built 

 forall(c in CITIES) equiv( build(c) = 1, numdep(c) >= 1 ) 

  

! Limit total number of offices 

 sum(c in CITIES) build(c) <= NUMLOC 

  

! Branching strategy 

 Strategy(1):= assign_and_forbid(KALIS_MAX_DEGREE, KALIS_MAX_TO_MIN, build) 

 Strategy(2):= split_domain(KALIS_SMALLEST_DOMAIN, KALIS_MIN_TO_MAX,  

                            depdist, true, 5) 

 cp_set_branching(Strategy) 

  

! Objective: weighted total distance 

 totDist = sum(c in CITIES) POP(c)*depdist(c) 

 

! Solve the problem 

 if not cp_minimize(totDist) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

  

! Solution printing 

 writeln("Total weighted distance: ", getsol(totDist),  

        " (average per inhabitant: ",  

    getsol(totDist)/sum(c in CITIES) POP(c), ")") 

 forall(c in CITIES) if(getsol(build(c))>0) then 

  write("Office in ", c, ": ") 

  forall(d in CITIES) write(if(getsol(depend(d))=c, " "+d, "")) 

  writeln 

 end-if 

 

!----------------------------------------------------------------- 

 

! Calculate the distance matrix using Floyd-Warshall algorithm 

 procedure calculate_dist 

 ! Initialize all distance labels with a sufficiently large value 

  BIGM:=sum(c,d in CITIES | exists(LEN(c,d))) LEN(c,d) 

  forall(c,d in CITIES) DIST(c,d):=BIGM 

 

  forall(c in CITIES) DIST(c,c):=0    ! Set values on the diagonal to 0 

  

! Length of existing road connections 

  forall(c,d in CITIES | exists(LEN(c,d))) do 

   DIST(c,d):=LEN(c,d) 

   DIST(d,c):=LEN(c,d) 

  end-do 

  

! Update shortest distance for every node triple  

  forall(b,c,d in CITIES | c<d ) 

   if DIST(c,d) > DIST(c,b)+DIST(b,d) then 

    DIST(c,d):= DIST(c,b)+DIST(b,d) 

    DIST(d,c):= DIST(c,b)+DIST(b,d) 

   end-if  

 end-procedure  

  

end-model 

このモデルのインプリメンテーションには、Mosel でのサブルーチンの別の使用例が含まれています。

すなわち、距離データの計算は、手順 calculate_dist で行われています。その結果、モデルの構造化

にサブルーチンを使用し、主モデルの定式化から、補助的なタスクを取り除くことができます。 

 

3.9.3 結果 

この問題の最適解として、人口をウエートとして、総距離 2438という値が得られました。管轄区域には、

合計 18 万 5000 人の住民がいるので、住民一人あたりの平均距離は、2438/185、すなわち、約

13.178km となります。3 つのオフィスは、ノード 1、ノード 6、ノード 11 に設立されることになります。ノー

ド 1 のオフィスは都市 1、2、5、7 をサービス区域とし、ノード 6 のオフィスは都市 3、4、6、9 をサービス

区域とし、オノード 11 のフィスは都市 8、10、11、12 をサービス区域とします。 

3.10 cycle : ペイントの生産 

このセクションでは、もう一度、Section 3.8 のペイント生産の問題について考えて見ましょう。この問題

の目的は、すべての組の仕事の間には、ジョブの組合せで、洗浄時間が順序に依存する状態で、ジョ

ブの集合を与えられたとき、すべてのジョブをどのような順番で行ったら、生産サイクルが最小になる

かを決定することです。 

3.10.1 モデルの定式化 

セクション 3.8 の問題定式化は、ジョブのすべてが、一度だけ行われることを保証するするために、

'all-different' constraint を使い、'element' constraint で洗浄時間を計算し、そして、生産順序でのサブ

サイクルを防ぐために、補助変数と制約条件を導入しました。実は、これらの制約条件が示す関係の

すべてが、ただ一つの制約関係、'cycle' constraint'で表現することができます。 

JOBS = {1,...,NJ}で生産するバッチの集合を、DURj でバッチ j の処理時間を、そして、CLEANij で

二つの連続するバッチ i と j の間の洗浄時間を表すものとします。前と同じように、各ジョブに続くジョ

ブを示すために、JOBS の中の値を取る決定変数 succj を定義します。以下は、その完全なモデル定

式化です。 

 

minimize å
Î

+
JOBSj

j cleantimeDUR  

{j}\: JOBSsuccJOBSj j ÎÎ"  

( ) ( )( )
JOBSjiijJOBSjj CLEANsucccyclecleantime

ÎÎ
=

,
,  

where 'cycle' stands for the relation 'sequence into a single cycle without subcycles 

or repetitions'. The variable cleantime equals the total duration of the cycle. 

ここで、'cycle'は、「subcycle、および、repetitionなしに、単一のcycleに順序付けする」という関係を意

味します。また、変数 cleantime は、 ( ) ( )( )
JOBSjiijJOBSjj CLEANsucccycle

ÎÎ ,
, に等しくなります。 

 

3.10.2 インプリメンテーション 

cycle constraint を使った Mosel モデルは、下記のようになります。 

 

model "B-5 Paint production (CP)" 

 uses "kalis" 

 

 setparam("DEFAULT_LB", 0) 

 

 declarations 

  NJ = 5                             ! Number of paint batches (=jobs) 

  JOBS=0..NJ-1 

 

  DUR: array(JOBS) of integer        ! Durations of jobs 

  CLEAN: array(JOBS,JOBS) of integer ! Cleaning times between jobs 

  CB: array(JOBS) of integer         ! Cleaning times after a batch 

 

  succ: array(JOBS) of cpvar         ! Successor of a batch 

  cleanTime,cycleTime: cpvar         ! Durations of cleaning / complete cycle 

 end-declarations 

 

 initializations from 'Data/b5paint.dat' 

  DUR CLEAN  

 end-initializations 

  

 forall(j in JOBS) do 

  0 <= succ(j); succ(j) <= NJ-1; succ(j) <> j 

 end-do 

 

! Assign values to 'succ' variables as to obtain a single cycle 

! 'cleanTime' is the sum of the cleaning times 

 cycle(succ, cleanTime, CLEAN) 

  

! Objective: minimize the duration of a production cycle 

 cycleTime = cleanTime +sum(j in JOBS) DUR(j) 

 

! Solve the problem 

 if not cp_minimize(cycleTime) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

 cp_show_stats 

 

! Solution printing 

 writeln("Minimum cycle time: ", getsol(cycleTime)) 

 writeln("Sequence of batches:\nBatch Duration Cleaning") 

 first:=1  

 repeat 

  writeln("  ", first, strfmt(DUR(first),8), 

                strfmt(CLEAN(first,getsol(succ(first))),9) ) 

  first:=getsol(succ(first)) 

 until (first=1) 

 

end-model 

cycle constraint で求められる入力フォーマットに従うために、タスクの番号が、0 から始まるインデッ

クスで、付け替えられているのに注意してください。 

 

3.10.3 結果 

この問題の最適解では、ミニマム・サイクルタイムは243分となり、その内訳は、(圧縮できない)処理時

間202分と洗浄時間41分です。 

Xpress-Kalis によって作成されたモデル・ランについての統計によると、このモデルの'cycle'バージョ

ンは、この問題を表現し、解くための最も効率的な方法であることが明らかになりました。これは、セク

ション 3.8 の 2 つの定式化よりも、より少ないノード、より短い時間で解が得られました。 

3.11 Generic binary constraints: オイラーのナイトツアー 

ここでの課題は、チェス盤の上で、ナイトを動かし、すべてのセルを、一回だけ、通過し、最後に、出発

点へ戻ってくるツアー（経路）を見つけることです。ナイトの経路は、チェスの標準ルールに従わなけれ

ばなりません。ナイトは、一回に、垂直方向に1つのセルだけ動き、併せて、水平方向には2つのセル

だけ動く、もしくは、垂直方向に 2 つのセルだけ動き、併せて、水平方向には 1 つのセルだけ動かせま

す。つまり、Figure3.3 で示されているように、この図の中央に位置しているナイトは、茶色のセルにの

み、動ける、ということです。 

 

 

Figure 3.3: Permissible moves for a knight 

3.11.1 モデルの定式化 

チェス盤を表すために、ここでは、セルに、0 から N-1 までの番号を付けます。ここで、N はチェス盤の

セルの数です。ナイトの経路は、ナイトが、ツアーの途中の i 番目の位置を示す N 個の変数 posiによ

って定義されます。 

最初の変数を、「ツアーの開始地点」としてゼロに固定します。もう一つ、明らかに必要な制約条件は、

すべての変数 posi は異なる値を取るということで、換言すれば、チェス盤のセルは、すべてが、まさに、

一回だけ、訪問されるということです。 

( )nposposdifferentall ,...,1-  

ここで、「連続した位置が有効なナイトの移動を定義するかどうか」をチェックできる制約関係を確立す

る必要があります。この目的のために、新しい binary constraint 'valid_knight_move'を定義し

ます。'valid_knight_move'は、与えられた二つの値の組が、ナイトの移動についてのチェスのル

ールに従う、許された移動を定義するかどうかチェックします。二つの値は、垂直方向、および、水平

方向に、2 つ以下のセルで離れていること、そして、垂直方向、水平方向の場所の違いの合計が 3 で

なければなりません。こうして、モデル全体は以下のようになります。 

{ } { }1,...,0:,...,1 -Î=Î" NposNPATHi i  

01 =pos  

( )Nposposdifferentall ,...,1-  

{ } ( )1,__:1,...,1 +-=Î" ii posposmoveknightvalidNPOSi  

( )1,__ posposmoveknightvalid N  

3.11.2 インプリメンテーション 

ナ イ ト の 位 置 a か ら 位 置 b へ の 動 き が 正 し い か ど う か を テ ス ト す る こ と は 、 function 

valid_knight_move で行えます。ここで、'div'は、余りを持たない整数の割り算（integer division 

without rest ）で、'mod'は、整数の割り算の余り（rest of the integer division）を意味します。 

 

-------------------------------------------------------------------- 

function valid_knight_move(a,b) 

 xa := a div E 

 ya := a mod E 

 xb := b div E 

 yb := b mod E 

 deltax} := |xa-xb| 

 deltay} := |ya-yb| 

 return ((deltax} <= 2) and (deltay} <= 2) and (deltax} + deltay} = 3)) 

end-function 

  

-------------------------------------------------------------------- 

   

以下の Mosel モデルは、movep という二つの値のペア変数についての「新しい binary constraint の実

行」としての、user constraint function 'valid_knight_move' の定義です。(この constraint は、 

generic_binary_constraint により設定されます)。 

 

model "Euler Knight Moves" 

 uses "kalis" 

 

 parameters 

  S = 8                                   ! Number of rows/columns 

  NBSOL = 1                               ! Number of solutions sought  

 end-parameters 

  

 forward procedure print_solution(sol: integer) 

 

 N:= S * S                                ! Total number of cells 

 setparam("DEFAULT_LB", 0) 

 setparam("DEFAULT_UB", N-1) 

 

 declarations 

  PATH = 1..N                            ! Cells on the chessboard 

  pos: array(PATH) of cpvar              ! Cell at position p in the tour 

 end-declarations 

 

! Fix the start position 

 pos(1) = 0 

 

! Each cell is visited once 

 all_different(pos, KALIS_GEN_ARC_CONSISTENCY) 

 

! The path of the knight obeys the chess rules for valid knight moves 

 forall(i in 1..N-1) 

  generic_binary_constraint(pos(i), pos(i+1), "valid_knight_move") 

 generic_binary_constraint(pos(N), pos(1), "valid_knight_move") 

 

! Setting enumeration parameters 

 cp_set_branching(probe_assign_var(KALIS_SMALLEST_MIN, KALIS_MAX_TO_MIN, 

                                   pos, 14)) 

 

! Search for up to NBSOL solutions 

 solct:= 0 

 while (solct<NBSOL and cp_find_next_sol) do 

  solct+=1 

  cp_show_stats 

  print_solution(solct) 

 end-do 

 

 

! **** Test whether the move from position a to b is admissible **** 

 function valid_knight_move(a:integer, b:integer): boolean 

  declarations 

   xa,ya,xb,yb,delta_x,delta_y: integer 

  end-declarations 

 

  xa := a div S 

  ya := a mod S 

  xb := b div S 

  yb := b mod S 

  delta_x := abs(xa-xb) 

  delta_y := abs(ya-yb) 

  returned := (delta_x<=2) and (delta_y<=2) and (delta_x+delta_y=3) 

 end-function 

 

!**************************************************************** 

! Solution printing 

 procedure print_solution(sol: integer) 

  writeln("Solution ", sol, ":") 

  forall(i in PATH) 

   write(getval(pos(i)), if(i mod 10 = 0, "\n ", ""), " -> ") 

  writeln("0") 

 end-procedure 

end-model 

このモデルで使用される branching scheme は、probe_assign_var heuristic で、変数選択のための

KALIS_SMALLEST_MIN（smallest lower bound を持つ変数を選択する）、および、値の選択基準である

KALIS_MAX_TO_MIN（最も大きいドメイン値から最も小さいドメイン値へ）と組み合わせて使います。(上

のコードリストの戦略より遅いですが)よく機能することが判明している、もう一つのサーチ戦略は、以

下のものです。 

cp_set_branching(assign_var(KALIS_SMALLEST_MIN, KALIS_MAX_TO_MIN, pos)) 

このモデルは、2 つのパラメタを定義します。 これにより、モデルを実行するとき、モデルソースを変

更することなく、チェス盤(S)のサイズや、実行するときに求めるソリューションの数(NBSOL)を変えるこ

とが可能になります。 

3.11.3 結果 

このモデルによって印刷された最初のソリューションは、以下のツアーです。 

 

0 -> 17 -> 34 -> 51 -> 36 -> 30 -> 47 -> 62 -> 45 -> 39 

  -> 54 -> 60 -> 43 -> 33 -> 48 -> 58 -> 52 -> 35 -> 41 -> 56 

  -> 50 -> 44 -> 38 -> 55 -> 61 -> 46 -> 63 -> 53 -> 59 -> 49 

  -> 32 -> 42 -> 57 -> 40 -> 25 -> 8 -> 2 -> 19 -> 4 -> 14 

  -> 31 -> 37 -> 22 -> 7 -> 13 -> 28 -> 18 -> 24 -> 9 -> 3 

  -> 20 -> 26 -> 16 -> 1 -> 11 -> 5 -> 15 -> 21 -> 6 -> 23 

  -> 29 -> 12 -> 27 -> 10 -> 0  

3.11.4 代替的なインプリメンテーション 

上のモデルの目的は、user constraint の定義の一例をあげることでしたが、Section 3.10 の cyclic 

scheduling problem 用に作成されたモデルを使って、より効率的な方法で、この問題を実行することも

可能です。 set of successors (domains of variables succp)は、上で、user-defined binary constraint

の実行で使用したのと同じアルゴリズムを使用することで計算できます。 

Without repeating the complete model definition at this place, we simply display the 

model formulation, including the calculation of the sets of possible successors 

(procedure calculate_successors, and the modified procedure print_sol for solution 

printout. We use a simpler version of the 'cycle' constraints than the one we have 

seen in Section 3.1-0, its only argument is the set of successor variables---there 

are no weights to the arcs in this problem. 

こ こ で は 、 モ デ ル 全 体 の 定 義 を 繰 り 返 え さ ず 、 sets of possible successors (procedure 

calculate_successors)、および､ソリューションのプリントのための修正された procedure print_sol

を含む、モデルの定式化のみを表示します。Section 3.10 で見たものよりも、より簡単な'cycle' 

constraint'のバージョンを使いますが、その唯一の argument は、set of successor variables です。こ

の問題の arcs へのウエイトはまったくありません。 

 

model "Euler Knight Moves" 

 uses "kalis" 

 

 parameters 

  S = 8                                   ! Number of rows/columns 

  NBSOL = 1                               ! Number of solutions sought  

 end-parameters 

  

 forward procedure calculate_successors(p: integer) 

 forward procedure print_solution(sol: integer) 

 

 N:= S * S                                ! Total number of cells 

 setparam("DEFAULT_LB", 0) 

 setparam("DEFAULT_UB", N) 

 

 declarations 

  PATH = 0..N-1                           ! Cells on the chessboard 

  succ: array(PATH) of cpvar              ! Successor of cell p 

 end-declarations 

 

! Calculate set of possible successors 

 forall(p in PATH) calculate_successors(p) 

 

! Each cell is visited once, no subtours 

 cycle(succ) 

 

! Search for up to NBSOL solutions 

 solct:= 0 

 while (solct<NBSOL and cp_find_next_sol) do 

  solct+=1 

  cp_show_stats 

  print_solution(solct) 

 end-do 

 

 

! **** Calculate possible successors **** 

 procedure calculate_successors(p: integer) 

  declarations 

   SuccSet: set of integer                ! Set of successors 

  end-declarations 

 

  xp := p div S 

  yp := p mod S 

  forall(q in PATH) do 

   xq := q div S 

   yq := q mod S 

   delta_x := abs(xp-xq) 

   delta_y := abs(yp-yq)   

   if (delta_x<=2) and (delta_y<=2) and (delta_x+delta_y=3) then 

    SuccSet +={q} 

   end-if 

  end-do 

  

  setdomain(succ(p), SuccSet) 

 end-procedure 

 

!**************************************************************** 

! **** Solution printing **** 

 procedure print_solution(sol: integer) 

  writeln("Solution ", sol, ":") 

  thispos:=0 

  nextpos:=getval(succ(0)) 

  ct:=1 

  while (nextpos<>0) do 

   write(thispos, if(ct mod 10 = 0, "\n ", ""), " -> ")  

   val:=getval(succ(thispos)) 

   thispos:=nextpos 

   nextpos:=getval(succ(thispos)) 

   ct+=1 

  end-do 

  writeln("0") 

 end-procedure 

 

end-model 

変数 succp のドメインの計算により、これらを、2-8 要素まで減少させます。あらゆる変数 posp の N = 

64 と比較してください。これにより、この問題のサーチスペースは、明らかに減少します。 

This second model finds the first solution to the problem after 131 nodes taking just 

a fraction of a second to execute on a standard PC whereas the first model requires 

several thousand nodes and considerably longer running times. It is possible to reduce 

the number of branch-and-bound nodes even further by using yet another version of 

the 'cycle' constraint that works with successor and predecessor variables. This 

version of 'cycle' performs a larger amount of propagation, at the expense of 

(slightly) slower execution times for our problem. The procedure 

calculate_successors now sets the domain of predp} to the same values as succp} for 

all cells p. 

この二番目のモデルは、標準のパソコンで、131 ノードのサーチのあと、この問題の最初のソリューシ

ョンを、まさしく、1秒も掛けずにみつけました。それにたいして、最初のモデルでは、数千のノードのサ

ーチが必要で、したがって、かなり長い実行時間が掛かります。さらに、'cycle' constraint の、

successor 変数、predecessor 変数を考慮するさらに別のバージョンを使うと、branch-and-bound node

の数を、もっと減らすことができます。'cycle'の、このバージョンは、この問題の実行時間を、ごくわず

かですが犠牲にすることと引き換えに、もっと多くのプロパゲーションを実行できます。procedure 

calculate_successors は、すべてのセル p にたいして、domain of predp を succp と同じ値にセットす

るようになります。 

 declarations 

  PATH = 0..N-1                           ! Cells on the chessboard 

  succ: array(PATH) of cpvar              ! Successor of cell p 

  pred: array(PATH) of cpvar              ! Predecessor of cell p 

 end-declarations 

 

! Calculate sets of possible successors and predecessors 

 forall(p in PATH) calculate_successors(p) 

 

! Each cell is visited once, no subtours 

 cycle(succ, pred) 

Figure3.4 は、IVE の'user graph'機能を使って作成されたナイトのツアーのグラフィカルなディスプレー

です。 

 

 

Figure 3.4: Knight’s tour solution graph in IVE 

3.11.5 代替的なインプリメンテーション その二 

しかし、さらに、(同じような数のノードですが、かなり短い実行時間という意味で)もっと効率的なのは、

Section 3.10 で説明されている'cycle' constraint を使用する下記のモデルバージョンです。こ

のモデルでは、1 つの位置から次の位置までの移動時間は、無意味なので、ここでは、単純に、

'cycle' constraint のすべての係数を定数に固定して、cycle length を対応する値に制約しま

す。 

sets of possible successors (procedure calculate_successors)、および､ソリューションのプリ

ントアウトのための procedure print_sol を含む、その他のすべての部分は、前のモデルからのコピ

ーです。 

 

model "Euler Knight Moves" 

 uses "kalis" 

 

 parameters 

  S = 8                                   ! Number of rows/columns 

  NBSOL = 1                               ! Number of solutions sought  

 end-parameters 

  

 forward procedure calculate_successors(p: integer) 

 forward procedure print_solution(sol: integer) 

 

 N:= S * S                                ! Total number of cells 

 setparam("DEFAULT_LB", 0) 

 setparam("DEFAULT_UB", N) 

 

 declarations 

  PATH = 0..N-1                           ! Cells on the chessboard 

  succ: array(PATH) of cpvar              ! Successor of cell p 

 end-declarations 

 

! Calculate set of possible successors 

 forall(p in PATH) calculate_successors(p) 

 

! Each cell is visited once, no subtours 

 cycle(succ) 

 

! Search for up to NBSOL solutions 

 solct:= 0 

 while (solct<NBSOL and cp_find_next_sol) do 

  solct+=1 

  cp_show_stats 

  print_solution(solct) 

 end-do 

! **** Calculate possible successors **** 

 procedure calculate_successors(p: integer) 

  declarations 

   SuccSet: set of integer                ! Set of successors 

  end-declarations 

 

  xp := p div S 

  yp := p mod S 

     

  forall(q in PATH) do 

   xq := q div S 

   yq := q mod S 

   delta_x := abs(xp-xq) 

   delta_y := abs(yp-yq)   

   if (delta_x<=2) and (delta_y<=2) and (delta_x+delta_y=3) then 

    SuccSet +={q} 

   end-if 

  end-do 

  

  setdomain(succ(p), SuccSet) 

 end-procedure 

 

!**************************************************************** 

! **** Solution printing **** 

 procedure print_solution(sol: integer) 

  writeln("Solution ", sol, ":") 

  thispos:=0 

  nextpos:=getval(succ(0)) 

  ct:=1 

  while (nextpos<>0) do 

   write(thispos, if(ct mod 10 = 0, "\n ", ""), " -> ")  

   val:=getval(succ(thispos)) 

   thispos:=nextpos 

   nextpos:=getval(succ(thispos)) 

   ct+=1 

  end-do 

  writeln("0") 

 end-procedure 

 

end-model 

第 4 章 列挙（Enumeration） 

この章は、下記のような、Xpress-Kalis のサーチ戦略の定義に関して、その概観を説明します。 

· 事前に定義されているサーチ戦略 

· 列挙を中断し、再開する方法 

· サーチの callback 

· ユーザサーチ戦略の定義 

4.1 事前に定義されているサーチ戦略 

Xpress-Kalis では、ブランチ戦略は、下記の 3 つのコンポーネントから構成されます。 

· サーチツリーの形を決定するのはブランチスキームです。その中には、assign_var、

split_domain (enumeration of decision variables)、 settle_disjunction (enumeration over 

constraints)、task_serialize (enumeration of tasks in scheduling problems)、もしくは、

probe_assign_var や probe_settle_disjunction のように、潜在的に不完全なサーチな

ども含まれています。 

· 変数選択戦略は、ブランチする変数の選択を決定します。あらかじめ、事前に定義されて

いるクライテリアには、以下の物があります。 

KALIS_INPUT_ORDER: variables in the given order 

KALIS_LARGEST_MAX: variable with largest upper bound 

KALIS_LARGEST_MIN: variable with largest lower bound 

KALIS_MAX_DEGREE: variable occurring in the largest number of  

constraints 

KALIS_MAXREGRET_LB: variable with largest difference between its  

lower bound and second-smallest domain value 

KALIS_MAXREGRET_UB: variable with largest difference between its  

upper bound and second-largest domain value 

KALIS_RANDOM_VARIABLE: random variable choice 

KALIS_SDOMDEG_RATIO: variable with smallest ratio domain size/degree 

KALIS_SMALLEST_DOMAIN variable with smallest number of domain  

values/smallest domainintervall 

KALIS_SMALLEST_MAX: variable with smallest upper bound 

KALIS_SMALLEST_MIN: variable with smallest lower bound) 

KALIS_WIDEST_DOMAIN: variable with largest number of domain values/  

largest domain interval 

また、ユーザも、自分自身の変数選択戦略を定義することもできます(下の Section 4.4.3 を

見てください)。 

· ひとたび、ブランチするべき変数が決まると、value selection 戦略により、branching value

の選択が決定されます。下記に列記した、あらかじめ、事前に定義された選択クライテリア

が利用可能です。 

 KALIS_MAX_TO_MIN: enumeration of values in decreasing order, 

 KALIS_MIDDLE_VALUE: enumerate first values in the middle of the  

variable's domain, 

 KALIS_MIN_TO_MAX: enumeration of values in increasing order, 

 KALIS_NEAREST_VALUE: choose the value closest to a target value  

previously specified with settarget, 

 KALIS_RANDOM_VALUE: choose a random value out of the variable's  

domain. 

あらかじめ、事前に定義された評価クライテリアは、ユーザ自身の value 選択戦略に取り替

えることがもきます(Section 4.4.3 を見てください)。 

branching scheme の選択によっては、列挙を構成するための、追加パラメタを指定することができま

す。詳細については、Xpress-Kalis reference manual を参照してください。constraint branching の場合

は、明らかなことですが、変数や branching value の選択はありません。Section 5.7 で、「task-based 

branching strategies (branching scheme task_serialize)」の特別なケースについて議論します。連

続変数(type cpfloatvar)は、いつも、branching scheme 'split_domain' を、KALIS_WIDEST_DOMAIN 

変数選択と一緒に使い、これは、上でリストした変数選択のサブセットです。 

ブランチイング戦略は、ある指定された変数(もしくは、適用できる場合には、constraint や task）、そし

て、branching scheme function のアーギュメントが省かれている場合は、モデルのすべての決定変数

に適用されます（例えば、Section 3.5 のモデル b4seq_ka.mos を見てください)。 

ユーザのモデルが、ブランチ戦略をなにも指定しないと、Xpress-Kalis は、ディフォルト戦略を適用して、

モデルの中の、すべての変数を列挙します。よしんば、列挙のディフォルト戦略(for discrete variables: 

`smallest domain first' and ̀ smallest value first')を変える必要がないと思う場合でも、決定変数の異な

るグループの列挙の順序を固定するために、ブランチ戦略を定義するのは望ましいことです。前の章

には、これのいくつかの例が含まれています。例えば､Section 3.6 のモデル a4sugar_ka.mosには、モ

デルの変数のいくつかの列挙を定義し、その結果、モデルの変数のいくつかは、残りの変数に対して

優先されます。また、Section 3.9のモデルj5tax_ka.mosでは、決定変数のグループごとに、異なる列

挙で構成されるサーチ戦略の例を見ました。  

モデルが離散変数と連続変数(cpvar と cpfloatvar)の両方を含んでいる場合は、ディフォルト戦略は、

最初に、離散変数を列挙して、次いで、連続変数を列挙します。  

4.2 列挙を中断し、再開する方法 

大きいアプリケーションを解くとき、それなりの妥当な時間で、完全なサーチツリーを列挙するのは、し

ばしば、可能でないことがあります。したがって、Xpress-Kalis では、サーチを中断するための、下に

記したような、いくつかの中断クライテリアが利用可能です。 

MAX_BACKTRACKS: maximum number of backtracks, 

MAX_COMPUTATION_TIME: limit on total time spent in search, 

MAX_DEPTH: maximum search tree depth, 

MAX_NODES: maximum number of nodes to explore, 

MAX_SOLUTIONS: maximum number of solutions, 

OPT_ABS_TOLERANCE: absolute difference between the objective  

function value in a solution and its best 

possible value (= current upper bound on 

objective function in maximization problems and 

lower bound with minimization), 

OPT_REL_TOLERANCE: relative difference between the objective  

function value in a solution and its best 

possible value (= current upper bound on 

objective function in maximization problems and 

lower bound with minimization). 

これらのパラメタには、Mosel ファンクションの setparam と getparam でアクセスできます(例えば、

Section 3.3 のモデル sudoku_ka.mos の問題統計の出力や、Section 3.4 のモデル freqasgn.mos の

search time limit set を見てください)。 

最適化問題では、パラメタ OPTIMIZE_WITH_RESTART の設定が変えられない場合、ソリューションが見

つけられた後に、このポイントからサーチが続きます。サーチが、上でリストしたクライテリア 1 つによ

って中断されて、次に、例えば、異なるサーチ戦略で続けられても同じです。root node からサーチを

再開するには、手順 cp_reset_search をコールする必要があります(例としては、Section 3.4 のモデ

ル freqasgn.mos を見てください)。 

4.3 サーチの callback 

サーチの間に、「callback function」によって、ユーザ・モデルは、あらかじめ定義されたポイントで、ソ

ルバーと対話（interact）できます。この機能は、最適化を行っている最中に、Section 3.4 のモデル

freqasgn.mos で示されているように、中間的なソリューションの情報をリトリーブするのに、特に便利

です。 このsolution callbackを除いて、ユーザは、あらゆるbranch、または、あらゆるノードでコールさ

れるファンクションを設定できます(詳細は、Xpress-Kalis reference manual を見てください)。 

4.4 ユーザサーチ戦略の定義 

以下の問題は、`Applications of optimization with Xpress-MP''の Section 14.1 から取ったものです。 

あるワークショップには 6 つの機械があり、そのそれぞれに、オペレータを割り当てれる必要がありま

す。あらかじめ、6 人の労働者が選択されています。6 人の労働者は、6 つの機械のそれぞれで、生産

性のテストを受けています。Table 4.1 は、1 時間あたりの生産性を、個数単位で記したものです。機

械は、パラレルに動かされます。これは、ワークショップ全体の生産性は、個々の機械に割り当てられ

た人々の生産性の合計であることを意味します。 

 

Table: Productivity in pieces per hour  

--------------------------------------------- 

                Machines 

  Workers 1 2 3 4 5 6 

--------------------------------------------- 

   1 13 24 31 19 40 29 

   2 18 25 30 15 43 22 

   3 20 20 27 25 34 33 

   4 23 26 28 18 37 30 

   5 28 33 34 17 38 20 

   6 19 36 25 27 45 24 

--------------------------------------------- 

目的関数は、全体の生産性を最大にするには、機械への労働者の割り当てをどのように決めたらよ

いか、ということです。この問題を解くのに、以下のかなり自然な方法を使って、(non-optimal な)ヒュ

ーリスティック解を得ることから始めることができます。労働者 p を、その労働者が最も高い生産性を

示した機械 m に割り当てます(p→m)。（こうして、労働者 p が既に機械 m に配置され、機械 m にはオ

ペレータがいるので)行 p、列 m を削除します。このプロセスを、すべての労働者が割り当てられるまで、

繰り返し行います。結果として得られた割り当ては、データテーブルの太字で示されています。 しかし、

ここでの目的は、実は、この並列に稼動される 6 台の機械への労働者の割り当ての最適解を得ること、

および、直列で稼動される 6 台の機械への労働者の割り当ての最適解を得ることです。 

 

4.4.1 モデルの定式化 

このタイプの問題は、割当問題という名前でよく知られています。PERS で労働者の集合を、MACH で機

械の数号を、そして、OUTPpmで労働者 p の、機械 m での生産性を表すものとします。 PERS と MACH は、

両方とも、同じサイズ N の集合です。ここで、N 個の整数変数 assignp を定義します。assignp は、機械

の集合の中の値を取り、そこでは、assignpは、労働者 p が割り当てられる機械の番号を示します。労

働者 p が、単一のマシン m に割り当てられ、機械 m が、単一の労働者 p によって操作されるという事

実は、変数 assignp のすべてが異なる値を取る、という制約条件で表現されます。 

MACHassignPERSp p ÎÎ" :  

( )pMACHm assigndifferentall Î- U  

さらに、outputpで、労働者pによって生産されるアウトプットを示すことにします。これらの変数の値は、

変数 assignp の離散関数として得られます。 

passignpp OUToutputpPERS ,: ="  

4.4.1.1 パラレルに動かされ機械への割当て 

目的関数は、下記のように定式化され、outputp の合計を最大にします。 

Maximize å
ÎPERSp

poutput  

割当てによっては、インフィージブルかもしれません。このような場合には、対応する機械の値を、変

数 assignp のドメインから取り除く必要があります。 

 

4.4.1.2 直列で稼動される機械 

機械が直列で稼動される場合、それぞれの機械に割り当てられている労働者の中の、一番、生産性

の低い労働者が、ライン全体の生産性を決定します。この場合も、割当は、N 個の変数 assignpと、こ

れらの変数についての all-different constraint によって表現できます。前のモデルの assignp変数

の値に、outputp 変数をリンクする制約条件をともなう outputp 変数もあります。これに、最小の生産

性を表す変数 pmin を加えます。こうして、目的関数は、pmin を最大にすることになります。「最小を

「最大にしたい」という、このようなタイプの最適化問題は、maximin 問題とか bottleneck 問題と呼ばれ

ます。 

maximize pmin 

pmin = minimum ( )pPERSp outputÎ  

4.4.2 インプリメンテーション 

以下のMoselプログラムは、最初に、並列マシンのケースのモデルを実行し、解きます。 その後に、

機械が直列に動くケースの問題を解くのに必要な変数pmin を定義します。 

model "I-1 Personnel assignment (CP)" 

 uses "kalis" 

 

 forward procedure parallel_heur 

 forward procedure print_sol(text1,text2:string, objval:integer) 

 

 declarations 

  PERS = 1..6                        ! Personnel 

  MACH = 1..6                        ! Machines 

  OUTP: array(PERS,MACH) of integer  ! Productivity 

 end-declarations 

 

 initializations from 'Data/i1assign.dat' 

  OUTP 

 end-initializations 

  

! **** Exact solution for parallel machines **** 

 

 declarations 

  assign: array(PERS) of cpvar       ! Machine assigned to a person  

  output: array(PERS) of cpvar       ! Productivity of every person 

  totalProd: cpvar                   ! Total productivity 

  O: array(MACH) of integer          ! Auxiliary array for constraint def. 

  Strategy: cpbranching              ! Branching strategy 

 end-declarations 

 

 forall(p in PERS) setdomain(assign(p), MACH) 

 

! Calculate productivity per worker 

 forall(p in PERS) do 

  forall(m in MACH) O(m):= OUTP(p,m) 

  element(O, assign(p)) = output(p) 

 end-do 

  

! Calculate total productivity 

 totalProd = sum(p in PERS) output(p) 

 

! One person per machine 

 all_different(assign) 

 

! Branching strategy 

 Strategy:= assign_var(KALIS_LARGEST_MAX, KALIS_MAX_TO_MIN, output) 

 cp_set_branching(Strategy) 

 

! Solve the problem 

 if cp_maximize(totalProd) then 

  print_sol("Exact solution (parallel assignment)", "Total", getsol(totalProd)) 

 end-if 

 

! **** Exact solution for serial machines **** 

 

 declarations 

  pmin: cpvar                        ! Minimum productivity 

 end-declarations 

 

! Calculate minimum productivity 

 pmin = minimum(output) 

 

! Branching strategy 

 Strategy:= assign_var(KALIS_SMALLEST_MIN, KALIS_MAX_TO_MIN, output) 

 cp_set_branching(Strategy) 

 

! Solve the problem 

 if cp_maximize(pmin) then 

  print_sol("Exact solution (serial machines)", "Minimum", getsol(pmin)) 

 end-if 

パラレルな割当問題へのソリューションが見つかると、そのソリューションをプリントアウトし、新しいブ

ランチ戦略と新しい目的関数でサーチを再開します。(タイムリミットによる中断などがなく)最初のサー

チが完全に終わったので、2 つのランの間で、ソルバーをリセットする必要は全くありません。 

パラレルな割当の問題に選ばれたブランチ戦略は、Section 4.4 のイントロダクションで説明された直

感的な手順にヒントを得たものです。すなわち、機械に労働者割り当てる可能な方法を列挙すること

(変数 assignp に関する列挙）の代わりに、変数 outputp についての列挙を定義し、largest remaining 

value (KALIS_LARGEST_MAX)変数を選び、大きいほうから小さいほう順(KALIS_MAX_TO_MIN)に、その

値にブランチします。二番目の問題に対しては、別の途をとる必要があります。すなわち、労働者 p の

生産性の値が小さい値になるのを避けるために、まず、smallest lower bound (KALIS_SMALLEST_MIN)

を持つ変数 outputp を選びます。そして、ここでも、最も大きい値から、それらを列挙します。 

上記のプログラムに、以下の手順 parallel_heur を追加しましょう。これは、Section 4.4 へのイント

ロダクションで説明された直感的な手順を使い、パラレルな割当問題の(non-optimal な)ソリューショ

ンをヒューリスティックに計算します。 

 procedure parallel_heur 

  declarations 

   ALLP, ALLM: set of integer        ! Copies of sets PERS and MACH 

   HProd: integer                    ! Total productivity value 

   pmax,omax,mmax: integer 

  end-declarations 

 

 ! Copy the sets of workers and machines 

  forall(p in PERS) ALLP+={p} 

  forall(m in MACH) ALLM+={m} 

 

 ! Assign workers to machines as long as there are unassigned persons 

  while (ALLP<>{}) do 

   pmax:=0; mmax:=0; omax:=0 

 

 ! Find the highest productivity among the remaining workers and machines 

   forall(p in ALLP, m in ALLM) 

    if OUTP(p,m) > omax then 

     omax:=OUTP(p,m) 

     pmax:=p; mmax:=m 

    end-if 

 

   assign(pmax) = mmax               ! Assign chosen machine to person pmax 

   ALLP-={pmax}; ALLM-={mmax}        ! Remove person and machine from sets 

  end-do 

 

  writeln("Heuristic solution (parallel assignment):") 

  forall(p in PERS) 

   writeln("  ",p, " operates machine ", getval(assign(p)), 

           " (",getval(output(p)), ")") 

  writeln("  Total productivity: ", getval(totalProd)) 

 end-procedure 

モデルは、適切にフォーマットされた方法でソリューションを印プリントアウトするための手順を置き、

完成します。 

procedure print_sol(text1,text2:string, objval:integer) 

  writeln(text1,":") 

  forall(p in PERS) 

   writeln("  ",p, " operates machine ", getsol(assign(p)), 

           " (",getsol(output(p)), ")") 

  writeln("  ", text2, " productivity: ", objval) 

 end-procedure 

  

end-model 

 

4.4.3 ユーザによるサーチ 

これまでのすべてのモデルのインプリメンテーションに示されている、事前に定義された変数と値の選

択クライテリアを使うことの代わりに、ユーザは、ユーザ自身のサーチヒューリスティックを定義するこ

ともできます。ここでは、前のモデルのサーチ戦略を、どのように「手動」で実行するかを示します。こ

こでのインプリメンテーションでは、毎回、二番目のクライテリオンを加え、メインのクライテリオンが、

一度に、いくつかの変数に適用されるケースで、それを断ち切ります。 

変数の選択： 変数選択ファンクションは、固定フォーマットを持っています。これは、アーギュメントと

して、タイプcpvarlistの変数のリストを受け取ります。そして、整数(すなわち、選ばれたブランチング

変数のリスト・インデックス)を返します。このファンクションに渡される変数のリストは、既に、挙げられ

ている変数を含むかもしれません。したがって、インプリメンテーションの最初のステップとして、いま

だ、挙げられていない変数の集合Vset、もしくは、より正確に言えば、リストの中の、それらのインデッ

クの集合を決定します。変数のリストのエントリーには、ファンクションgetvarでアクセスします。いま

だ、挙げられていない変数を対象に、(ファンクションgetubを使用して)最も大きい上限値を持つ変数

の集合Iset（set of variables Iset with the largest upper bound value）を計算します。最後に、これらの

変数の中から、最も小さい2番目に大きい値（smallest second-largest value）を持つ変数を選びます

(これは、maximum regret strategyに対応している)。決定変数のドメインの中の、predecessor 

(next-smallest) valuesは、ファンクションgetprevで得ます。逆に、昇順に、ドメインの値を列挙するため

のファンクションgetnextも持っています。 選ばれたインデックス値は、ファンクションのリターン値とし

て、returnedに割り当てられます。 

function varchoice(Vars: cpvarlist): integer 

  declarations 

   Vset,Iset: set of integer 

  end-declarations 

 

 ! Set of uninstantiated variables 

  forall(i in 1..getsize(Vars)) 

   if not is_fixed(getvar(Vars,i)) then Vset+= {i}; end-if 

  

  if Vset={} then 

   returned:= 0 

  else  

  ! Get the variable(s) with largest upper bound 

   dmax:= max(i in Vset) getub(getvar(Vars,i))  

   forall(i in Vset) 

    if getub(getvar(Vars,i)) = dmax then Iset+= {i}; end-if 

   dmin:= dmax 

 

  ! Choose variable with smallest next-best value among those indexed by 'Iset' 

   forall(i in Iset) do 

    prev:= getprev(getvar(Vars,i),dmax) 

    if prev < dmin then 

     returned:= i 

     dmin:= prev 

    end-if  

   end-do 

  end-if  

 end-function 

2 番目の最適化のラン（直列で稼動する機械）のための変数の選択戦略 varchoiceminも、同じような

方法で実行されます。最初に、getlbを使って、最も小さい下限値（smallest lower bound value）を持つ

変数の集合 Iset を確立します。 これらの中から、 smallest upper bound (getub)を持つ変数を選び

ます。 

 function varchoicemin(Vars: cpvarlist): integer 

  declarations 

   Vset,Iset: set of integer 

  end-declarations 

 

 ! Set of uninstantiated variables 

  forall(i in 1..getsize(Vars)) 

   if not is_fixed(getvar(Vars,i)) then Vset+= {i}; end-if 

  

  if Vset={} then 

   returned:= 0 

  else  

  ! Get the variable(s) with smallest lower bound 

   dmin:= min(i in Vset) getlb(getvar(Vars,i))  

   forall(i in Vset) 

    if getlb(getvar(Vars,i)) = dmin then Iset+= {i}; end-if 

 

  ! Choose variable with smallest upper bound among those indexed by 'Iset' 

   dmax:= getparam("default_ub") 

   forall(i in Iset) 

    if getub(getvar(Vars,i)) < dmax then 

     returned:= i 

     dmax:= getub(getvar(Vars,i)) 

    end-if  

  end-if  

 end-function 

値の選択： 値の選択ファンクションは、アーギュメントとして、選択されたブランチング変数を受け取っ

て、この変数のブランチング値を返します。選んだ「値の選択クライテリオン」(Kalis_MAX_TO_MIN に対

応する)は、最も大きい残っている値(すなわち、変数の上限)から始め、ブランチング変数のすべての

値を列挙します。 

 function valchoice(x: cpvar): integer 

  returned:= getub(x) 

 end-function 

assign_var、もしくは、assign_and_forbid戦略では、ファンクション containsを使い、ユーザの値の

選択戦略が、必ず、現在、確かに、ブランチング変数のドメインにある値を返すように気をつけてくださ

い。上限、下限の間から選ばれた値は、かならずしも、ドメインにあることを保障することにはなりませ

ん。 

ユーザサーチ戦略の設定: ユーザ自身の変数選択戦略や値の選択戦略を使いことを示すには、

単に、Mosel ファンクションの名前で事前に定義された定数を取り替えるだけです。 

Strategy:= assign_var("varchoice", "valchoice", output) 

上記は、最初の最適化のランのための戦略を定義します。 

Strategy:= assign_var("varchoicemin", "valchoice", output) 

上記は、直列で動く機械のケースに対して、戦略を再定義します。ファンクション valchoice は、

KALIS_MAX_TO_MIN クライテリオンと、まったく、同じことを行うので、それを下記の変数選択ファンク

ションと組み合わせることもできます。 

Strategy:= assign_var("varchoicemin", KALIS_MAX_TO_MIN, output) 

 

4.4.4 結果 

以下のテーブルは、パラレルに機械を動かす問題、および、直列で機械を動かす問題に関して得ら

れた結果の要約です。パラレルに機械を動かす問題へのヒューリスティックな方法と厳密なソリューシ

ョンの間には、著しい違いがあります。 

Table 4.2: Optimal assignments for different model versions   

       Person       

 Alg. 1 2 3 4 5 6 Productivity 

Parallel Machines  Heur.  4 (19) 1 (18) 6 (33) 2 (26) 3 (34) 5 (45)  175 

  Exact 3 (31) 5 (43) 4 (25) 6 (30) 1 (28) 2 (36) 193 

Serial Machines  Exact 5 (40) 3 (30) 6 (33) 2 (26) 1 (28) 4 (27) 26 

モデルにソルバー統計の出力を追加するか(cp_show_stats)、または、IVE で、CP 統計ディスプレー

を使い、以下のようなことがわかります。すなわち、パラレル割当では、ユーザサーチ戦略が、事前に

定義された戦略と同じサーチツリーをもたらしたこと、および、プログラム実行時間も同じであったこと、

また、直列では、わずかに異なったソリューションをもたらしたことがわかりました。 

第 5 章 スケジュール作成 

この章では、下記のことについて説明します。 

· どのように、モデル作成のオブジェクト cptaskと cpresourceを定義して、セットアップしたら

よいか。 

· これらのオブジェクトを使い、どのようにして、スケジュール作成問題を定式化し、解いたら

よいうか。 

· どのようにして、これらのモデル作成オブジェクトの情報にアクセスしたらよいか。 

5.1 タスクと資源 

問題をスケジュールし、計画することは、与えられたタスクの集合の実行のための計画を決定するこ

とに関係があります。目的は、与えられた、タスクの順序とか限られている資源についての制約条件

を満たす、実行可能なスケジュールを生成したり、スケジュールの makespan などの、与えられたクラ

イテリオンを最適化することです。 

Xpress-Kalis は、標準のスケジュール作成問題の定式化を簡素化するために、いくつかの全体的な

モデル作成オブジェクト（aggregate modeling objects）を定義します。タスク(処理/加工の作業、アクテ

ィビティ)は、タイプ cptask によって示され、また、資源(原材料、機械など)は、タイプ cpresource によ

って示されます。 

これらのスケジュール作成オブジェクトを使うには、オブジェクトと、タスク期間や資源の使用のような、

それらの特性を記述するだけで十分です。なぜなら、(implicit constraints と呼ばれる)Xpress-Kalis の

機能により、必要な制約条件の関係は自動的にセットアップされるからです。以下のセクションでは、

下記の例を使い、このメカニズムの使い方を説明します。 

· このスケジュール作成問題(Section 5.2 のプロジェクト・スケジューリング問題)は、非常に簡

単なケースで、タスクとタスク間の順序についての制約条件だけが関っています。 

· Section 5.3のdisjunctive scheduling/sequencing problem 問題では、タスクは、同一の資源

を使用するので、相互に排他的です。 

· 資源は、与えられたキャパシティの範囲内で、同時に使用可能なケースです(cumulative 

resources, see Section 5,4)。 

· 資 源 の 種 類 の 区 分 の 一 つ は 、 再 生 可 能 （ renewable ） な 資 源 か 、 そ う で は な い

（non-renewable）資源かの区分です(Section 5.5)。 

· 例えば､順序依存の段取りのような、標準問題のいろいろな拡張が可能です(Section 5.6)。 

cp_schedule を使って列挙を行うと、ソルバーは、対応する(スケジュール作成)問題のタイプに適した

特別なサーチ戦略を使います。これらの戦略を parameterize することもできますし、また、スケジュー

ル作成オブジェクトのための、ユーザ自身のサーチ戦略を定義することも可能です(see Section 5.7)。

あるいはまた、標準最適化ファンクション cp_minimize / cp_maximize を使用することもできます。こ

の場合、列挙は、スケジュール作成オブジェクトによって提供される構造的な情報を利用せず、決定

変数のみを使って列挙します。 

例えば、制約条件の定義では、スケジュール作成オブジェクトの(タスクの開始時刻や処理時間など

の)特性にアクセスし、使うことができます。こうして、ユーザは、事前に定義された標準の問題を、他

のタイプの制約条件で拡張できる可能性を享受できます。Xpress-Kalisでは、type cpvarの決定変数

についてのdedicated global constraintsを使って、ユーザが、aggregate modeling objectを使わずに、

ユーザのスケジュール作成問題を定式化できるので、さらに大きい柔軟性が得られます。こうした理

由で、本章のほとんどの例では、scheduling objectを使ったインプリメンテーション、scheduling object

を使を使わないインプリメンテーションという、二つの異なるインプリメンテーションが示されています。 

5.2 先行制約条件 (Precedence constraints) 

おそらく、スケジュール作成問題の最も基本的なタイプは、先行制約条件によってリンクされるタスク

の集合を計画することことでしょう。 

このセクションで取り上げる問題は、Applications of optimization with Xpress-MP という本の Section 

7.1 に記載されている「Construction of a stadium」という問題です。 

ある建設会社は、スタジアムを建築する契約を取り、最も短かい時間で完成したいと考えています。

Table 5.1 は、主要なタスクと、それらのタスクに掛かる時間を、週単位でリストしたものです。タスクの

中には、Table 5.1 に示されている他のあるタスクの完成の後に開始できるものもあります。 

Table 5.1: Data for stadium construction 

  Task   Description    Duration  Predecessors 

1 Installing the construction site  2   none   

2 Terracing   16 1 

3 Constructing the foundations  9 2 

4 Access roads and other networks  8 2 

5 Erecting the basement  10 3 

6 Main floor 6 4.5 

7 Dividing up the changing rooms 2 4 

8 Electrifying the terraces 2 6 

9 Constructing the roof 9 4, 6 

10 Lighting of the stadium  5 4 

11 Installing the terraces  3 6 

12 Sealing the roof 2 9 

13 Finishing the changing rooms 1 7 

14 Constructing the ticket office  7 2 

15 Secondary access roads 4 4, 14 

16 Means of signalling 3 8, 11, 14 

17 Lawn and sport accessories 9 12 

18 Handing over the building   1 17 

5.2.1 モデルの定式化 

この問題は、古典的なプロジェクト・スケジューリング問題です。 最後に、プロジェクトの完了に対応

する期間 0 の架空のタスクを加えます。 こうして、ここでは、TASKS = {1,...,N} について考えます

が、N は、この架空のタスクです。 

工事のタスク ( )TASKSjj Î は、すべて、タスクオブジェクト taskjで示され、このタスクオブジェクトは、

可変の開始時刻 taskj.start、および、与えられた値 DURj に固定された工事期間を持っています。タ

スク間の工事の順序は、arcs (i,j) を持った precedence graph with によって表され、タスク i がタ

スク j に優先するということを意味します。 

目的関数は、プロジェクトの完成までの時間、すなわち、最後の、そして、架空のタスク N の開始時刻

を最小にする（できるだけ早くする）ことです。こうして、下記のモデルを得ますが、ここで、開始時間に

ついての upper bound HORIZON は、すべてのタスク期間の合計によって与えられます。 

tasks 
( )tasksjtask j Î

  

minimize  taskN.start 
{ }HORIZONstarttaskjTASKS j ,...,0.: Î"

 

jj DURdurationtaskjTASKS =" .:
 
{ }iARCtsTASKSjj taskrspredecessotaskjTASKS

ij$Î=" ..,.: U
 

5.2.2 インプリメンテーション 

以下のモデ ルは 、Xpress-Kalis によるこの問題の インプリメンテーション を示 しています 。

side-constraints がないので、このスケジュールの最も早い、可能な完成時期は架空のタスク Nを最も

早く始めることになります。 

task-related constraint のプロパゲーションをしどうさせるために、ファンクション cp_propagate をコー

ルします。 

ここで、架空の end task の開始を、その lower bound の制約することで、constraint propagation の効

果により、すべてのタスク開始時期が、その feasible interval に減少し、そして、クリティカルパスの上

にあるタスクの開始時期は、ただ一つの値に固定されます。 

その後の最小化へのコール（call to minimization）は、IVE でソリューションのグラフ表示を可能にする

ために、すべての変数をただ一つの値に例示化する（instantiating all variables with a single value）た

めのみのものです。 

model "B-1 Stadium construction (CP)" 

 uses "kalis" 

  

 declarations 

  N = 19                              ! Number of tasks in the project  

                                      ! (last = fictitious end task) 

  TASKS = 1..N 

  ARC: array(range,range) of integer  ! Matrix of the adjacency graph  

  DUR: array(TASKS) of integer        ! Duration of tasks 

  HORIZON : integer                   ! Time horizon  

 

  task: array(TASKS) of cptask        ! Tasks to be planned 

  bestend: integer 

 end-declarations 

 

 initializations from 'Data/b1stadium.dat' 

  DUR ARC 

 end-initializations 

 

 HORIZON:= sum(j in TASKS) DUR(j) 

 

! Setting up the tasks  

 forall(j in TASKS) do 

  setdomain(getstart(task(j)), 0, HORIZON)   ! Time horizon for start times 

  set_task_attributes(task(j), DUR(j))       ! Duration 

  setsuccessors(task(j), union(i in TASKS | exists(ARC(j,i))) {task(i)}) 

 end-do                                      ! Precedences 

           

 if not cp_propagate then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if  

 

! Since there are no side-constraints, the earliest possible completion 

! time is the earliest start of the fictitiuous task N 

 bestend:= getlb(getstart(task(N))) 

 getstart(task(N)) <= bestend 

 writeln("Earliest possible completion time: ", bestend) 

 

! For tasks on the critical path the start/completion times have been fixed 

! by setting the bound on the last task. For all other tasks the range of 

! possible start/completion times gets displayed. 

 forall(j in TASKS) writeln(j, ": ", getstart(task(j))) 

 

! Complete enumeration: schedule every task at the earliest possible date 

 result:= cp_minimize(getstart(task(N))) 

 forall(j in TASKS) writeln(j, ": ", getstart(task(j))) 

 

end-model 

あるタスクの先行タスクを示すことの代わりに、下記のように、すべてのタスクの後続タスクの集合を

示すことで、同じように、precedence constraint を、はっきり規定できます。 

setsuccessors(task(j), union(i in TASKS | exists(ARC(j,i))) {task(i)}) 

5.2.3 結果 

スタジアム工事の最早な工事期間は 64 週間です。   

IVE のユーザは、このモデルの実行により、特別な window が開かれるのに気付くでしょう。これは、

CP dashboard という window で、ソリューションのグラフィカルなディスプレーを示してくれます(Figure 

5.1 を参照)。表示された図の上にマウスを持ってゆくと、ポップアップボックスが開かれ、タスクの詳細

な情報が示されます。タスクをクリックすると、別の window に、タスク関連のすべての情報が表示され

ます。CP dashboard hide/unhide button をクリックすると、この window を隠すことができます。 

 

Figure 5.1: CP dashboard in IVE displaying the solution 

5.2.4 オブジェクトをスケジュールしない代替的な定式化 

前の定式化では、TASKS={1、…N}というタスクの集合を使いました。ここで、N は、完了を示す架空の

タスクでした。ここで、あらゆるタスク j に関して、決定変数 startj を導入し、タスクの開始時刻を示し

ます。また、DURj、すなわち、タスク jの期間を使い、タスクの順序関係、「タスク iはタスク jに先行す

る」を下記の制約条件で記述します。 

starti + DURi £  startj 

こうして、このプロジェクト・スケジューリング問題のモデルは下記のようになります。 

minimize  startN 
{ }HORIZONstartTASKSj j ,...,0: ÎÎ"  

jiiij startDURstartARCTASKSji £+$Î" :,,  

以下に、対応する Mosel モデルをすべて記します。precedence constraint を明示的に使っているのに

注意してください。実行不可能なデータの場合、これは、インフィージビリティの原因をトレースするの

に役立ちます。 

 

model "B-1 Stadium construction (CP)" 

 uses "kalis" 

  

 declarations 

  N = 19                              ! Number of tasks in the project  

                                      ! (last = fictitious end task) 

  TASKS = 1..N 

  ARC: array(range,range) of integer  ! Matrix of the adjacency graph  

  DUR: array(TASKS) of integer        ! Duration of tasks 

  HORIZON : integer                   ! Time horizon  

 

  start: array(TASKS) of cpvar        ! Start dates of tasks 

  bestend: integer 

 end-declarations 

 

 initializations from 'Data/b1stadium.dat' 

  DUR ARC 

 end-initializations 

 

 HORIZON:= sum(j in TASKS) DUR(j) 

 

 forall(j in TASKS) do 

  0 <= start(j); start(j) <= HORIZON 

 end-do 

 

! Task i precedes task j 

 forall(i, j in TASKS | exists(ARC(i, j))) do 



 101 

  Prec(i,j):= start(i) + DUR(i) <= start(j) 

  if not cp_post(Prec(i,j)) then 

   writeln("Posting precedence ", i, "-", j, " failed") 

   exit(1) 

  end-if 

 end-do 

 

! Since there are no side-constraints, the earliest possible completion 

! time is the earliest start of the fictitiuous task N 

 bestend:= getlb(start(N)) 

 start(N) <= bestend 

 writeln("Earliest possible completion time: ", bestend) 

 

! For tasks on the critical path the start/completion times have been fixed 

! by setting the bound on the last task. For all other tasks the range of 

! possible start/completion times gets displayed. 

 forall(j in TASKS) writeln(j, ": ", start(j)) 

 

end-model 

5.3 離接的なスケジュール作成（Disjunctive scheduling）: unary resources 

Section 3.5 で説明されている単一マシン上にジョブを順序付けるという問題は、モデル作成のため

の「task オブジェクト」と「resource オブジェクト」を使い、離接的なスケジュール作成問題

（disjunctive scheduling problem）として表現できます。 

読者の皆さんは、この問題が、単一マシンにおける非割込み型（non-preemptive）のタスク(ジョブ)の

集合の処理の順序をスケジュールすることであると思い出してください。すべてのタスク j のリリース日、

処理に必要な期間、および、完了日が与えられています。この問題は、全体の完了期間（makespan）、

平均完成時間、または、遅延の合計を最小にするという、 3 つの異なる目的関数で解くことが出来ま

す。 

 

5.3.1 モデルの定式化 

モデル定式化の大半は、スケジュール作成のための「task オブジェクト」と「resource オブジェクト」の

定義から成ります。 

あらゆる jobj { }( )NJJOBSj ,...,1=Î  は、タスクオブジェクトtaskjによって表され、開始日は、taskj. 

start（{RELj, . . . ,MAXTIME}）です。ここで、MAXTIME は、リリース日付とすべての処理時間の合計

のような、十分に大きい値で、durationは、与えられた処理時間 DURjに固定されています。すべての

ジョブは、切り離せないキャパシティ（unitary capacity）を持つ同一の資源 res を使用します。これは、

ある時点で、最大でも、1 つのジョブしか処理できないことを意味しています。こうして、暗黙的に、ジョ

ブ間の離接的な関係を示します。 

task object によって確立されるもう一つの暗黙的な制約条件は、ジョブ j の開始時間、処理時間、完

成時間との間の関係です。 

durationtaskstarttaskendtaskj jjj ...: +=Î"  

目的関数 1： 最初の目的関数は、makespan(スケジュールの完成時間)を最小にすることです。これ

は、最後のジョブの完成時間を最も早くすることと同等です。こうして、モデル全体は以下のようになり

ます。ここで、MAXTIME は、リリース日付とすべての処理時間の合計のような、十分に大きい値です。 

resource res 

tasks ( )JOBSjtask j Î  

minimize finish 

( )endtaskimumfinish jJOBSj .max Î=  

1. =capacityres  

{ }MAXTIMEendtaskJOBSj j ,...,0.: ÎÎ"  

{ }MAXTIMERELstarttaskJOBSj jj ,...,.: ÎÎ"  

jj DURdurationtaskJOBSj =Î" .:  

1.: =Î" resj trequirementaskJOBSj  

目的関数 2： 2 番目の目的関数は、平均処理時間を最小にすることですが、これは、ジョブの完成時

間（job completion times）の合計を最小にすることと同等です。その定式化は、最初のモデルの目的

関数と変わりませんが、すべてのジョブの完了時間の合計を示す変数 totComp を追加します。 

 

minimize  totComp 

å
Î

=
JOBSj

j endtasktotComp .  

目的関数3：遅延の合計を最小にする目的関数を定式化するために、新しい変数latejを導入し、ジョ

ブが納期よりも遅れる時間を測定します。これらの変数の値は、ジョブ j の完成時間と納期 DUEj の差

に対応します。ジョブが期日以前終わる場合は、この変数の値はゼロです。 目的関数は、これらの

遅延変数の合計を最小にすることです。 

minimize  totLate 

å
Î

=
JOBSj

jlatetotLate  

{ }MAXTIMElateJOBSj j ,...,0: ÎÎ"  

jjj DUEendtasklateJOBSj -³Î" .:  

5.3.2 インプリメンテーション 

下記の Xpress-Kalis(ファイル b4seq3_ka.mos)のインプリメンテーションは、どのように必要なタスク、

および、資源のモデル作成オブジェクトをセットアップするかを示しています。資源キャパシティは手順

set_resource_attributes で設定され(資源はタイプ Kalis_UNARY_RESOURCE で、これは、一度に 1

ジョブを処理することを意味する）、また、資源は set_task_attributes で設定します。後者は、アー

ギュメント(タスクのアトリビュート)の異なる組み合わせにたいして、いくつかの overloaded version に

存在しています。詳細については、Xpress-Kalis Reference Manual を参照してください。 

maximum 制約式の定式化のためには、変数の(補助)リストを使用します。Xpress-Kalis は、union(j 

in JOBS) getend(task(j))のような set expressions で、ユーザが access functions を使い、

getstart、getduration などのオブジェクトのモデルを作成することを許していません。 

model "B-4 Sequencing (CP)" 

 uses "kalis" 

 

 forward procedure print_sol 

 forward procedure print_sol3 

  

 declarations    

  NJ = 7                          ! Number of jobs 

  JOBS=1..NJ 

 

  REL: array(JOBS) of integer     ! Release dates of jobs 

  DUR: array(JOBS) of integer     ! Durations of jobs 

  DUE: array(JOBS) of integer     ! Due dates of jobs 

 

  task: array(JOBS) of cptask     ! Tasks (jobs to be scheduled) 

  res: cpresource                 ! Resource (machine) 

 

  finish: cpvar                   ! Completion time of the entire schedule 

 end-declarations 

  

 initializations from 'Data/b4seq.dat' 

  DUR REL DUE 

 end-initializations 

  

! Setting up the resource (formulation of the disjunction of tasks) 

 set_resource_attributes(res, KALIS_UNARY_RESOURCE, 1) 

  

! Setting up the tasks (durations and disjunctions) 

 forall(j in JOBS) set_task_attributes(task(j), DUR(j), res) 

  

 MAXTIME:= max(j in JOBS) REL(j) + sum(j in JOBS) DUR(j) 

 

 forall(j in JOBS) do 

  0 <= getstart(task(j)); getstart(task(j)) <= MAXTIME  

  0 <= getend(task(j)); getend(task(j)) <= MAXTIME 

 end-do  

 

! Start times 

 forall(j in JOBS) getstart(task(j)) >= REL(j) 

!**** Objective function 1: minimize latest completion time **** 

 declarations 

  L: cpvarlist 

 end-declarations 

  

 forall(j in JOBS) L += getend(task(j)) 

 finish = maximum(L) 

 

 if cp_schedule(finish) >0 then 

  print_sol 

 end-if 

 

 

!**** Objective function 2: minimize average completion time **** 

 declarations    

  totComp: cpvar 

 end-declarations 

 

 totComp = sum(j in JOBS) getend(task(j)) 

 

 if cp_schedule(totComp) > 0 then 

  print_sol 

 end-if 

 

 

!**** Objective function 3: minimize total tardiness **** 

 declarations    

  late: array(JOBS) of cpvar      ! Lateness of jobs 

  totLate: cpvar 

 end-declarations 

 

 forall(j in JOBS) do 

  0 <= late(j); late(j) <= MAXTIME 

 end-do  

  

! Late jobs: completion time exceeds the due date 

 forall(j in JOBS) late(j) >= getend(task(j)) - DUE(j)  

 

 totLate = sum(j in JOBS) late(j) 

 if cp_schedule(totLate) > 0 then 

  writeln("Tardiness: ", getsol(totLate)) 

  print_sol 

  print_sol3 

 end-if 

 

 

!----------------------------------------------------------------- 

 

! Solution printing 

 procedure print_sol 

  writeln("Completion time: ", getsol(finish) , 

          "  average: ", getsol(sum(j in JOBS) getend(task(j)))) 

  write("Rel\t") 

  forall(j in JOBS) write(strfmt(REL(j),4)) 

  write("\nDur\t") 

  forall(j in JOBS) write(strfmt(DUR(j),4)) 

  write("\nStart\t") 

  forall(j in JOBS) write(strfmt(getsol(getstart(task(j))),4)) 

  write("\nEnd\t") 

  forall(j in JOBS) write(strfmt(getsol(getend(task(j))),4)) 

  writeln 

 end-procedure 

 

 procedure print_sol3 

  write("Due\t") 

  forall(j in JOBS) write(strfmt(DUE(j),4)) 

  write("\nLate\t") 

  forall(j in JOBS) write(strfmt(getsol(late(j)),4)) 

  writeln 

 end-procedure  

  

end-model 

 

5.3.3 結果 

このモデルは、Section 3.5 のモデルバージョンにたいしてレポートされたのと似ている結果

を生み出します。Figure5.2 は、IVE によって作成されたソリューションのガントチャートディ

スプレーを示しています。 ガントチャートの上に、資源利用ディスプレーがあります。この機

械は、中断なく、タスクによって使用されます。すなわち、リリース時間と納期に与えられた

制約条件を緩和しても、もっと早く終わるスケジュールを生成させるのは可能ではありません。 

 

 
Figure 5.2: Solution display in IVE 

5.4 Cumulative scheduling: discrete resources 

このセクションで説明される問題は、「Applications of optimization with Xpress-MP」という本の

Section 9.4 「Backing up files」からとったものです。 



 106 

ここでの課題は、1.44Mb の容量の空のフロッピーディスクに、46kb、55kb、62kb、87kb、108kb、114kb、

137kb、164kb、253kb、364kb、372kb、388kb、406kb、432kb、461kb、および、851kb のサイズの 16 個

のファイルを保存するとき、ファイルを各ディスクにどのように配分したら、使用されるフロッピーディス

クの数が最小になるか、その配分の方法を見つけることです。 

5.4.1 モデルの定式化 

この問題は、binpacking 問題と呼ばれている問題です。ここでは、この問題を、cumulative scheduling 

problemとして、どのように定式化し、といたらよいかを示します。この問題では、ディスクが資源であり、

そして、ファイルがスケジュールされるタスクです。 

フロッピーディスクは、それぞれ単一の、離散的な資源であると考えます。そこでは、あらゆるタイム・

ユニットは 1 個のディスクを意味します（where every time unit stands for one disk)。資源キャパシティ

は、ディスク容量に対応しています。 

ここでは、すべてのファイル f ( f FILESÎ ) を、タスクオブジェクト filef で表します。ファイルは、1

ユニットの固定期間、所与のサイズ SIZEf に対応する資源リクワイアメントを持っています。こうして、

タスクの`start' フィールドは、このファイルを保存するためのディスクの選択を示します。 

目的関数は、使用されるディスクの数を最小にすることです。これは、タスクの`start' フィールドがと

る最も大きい値(すなわち、ファイルを保存するのに使用されるディスクの数)を最小にすることに対応

します。こうして、以下のモデルを得ます。 

resource disks 

tasks ( )FILESffiles f Î  

minimize diskuse 

=diskuse maximum ( )startfile fFILESf .Î  

disks.capacity CAP=  

1.: ³Î" startfileFILESf f  

1.: =Î" durationfileFILESf f  

fdisksf SIZEtrequiremenfileFILESf =Î" .:  

 

5.4.2 インプリメンテーション 

Xpress-Kalis に よ る イ ン プ リ メ ン テ ー シ ョ ン は 、 か な り 簡 単 で す 。 タ イ プ

KALIS_DISCRETE_RESOURCE の資源を定義します。これは、キャパシティ総量を示します。タスク

の定義は、前の例で見たものと似ています。 

 

model "D-4 Bin packing (CP)" 

 uses "kalis" 

 



 107 

 declarations 

  ND: integer                        ! Number of floppy disks 

  FILES = 1..16                      ! Set of files 

  DISKS: range                       ! Set of disks 

 

  CAP:  integer                      ! Floppy disk size 

  SIZE: array(FILES) of integer      ! Size of files to be saved 

 

  file: array(FILES) of cptask       ! Tasks (= files to be saved) 

  disks: cpresource                  ! Resource representing disks 

  L: cpvarlist 

  diskuse: cpvar                     ! Number of disks used 

 end-declarations 

  

 initializations from 'Data/d4backup.dat' 

  CAP SIZE 

 end-initializations 

 

! Provide a sufficiently large number of disks 

 ND:= ceil((sum(f in FILES) SIZE(f))/CAP) 

 DISKS:= 1..ND 

 

! Setting up the resource (capacity limit of disks) 

 set_resource_attributes(disks, KALIS_DISCRETE_RESOURCE, CAP) 

  

! Setting up the tasks  

 forall(f in FILES) do 

  setdomain(getstart(file(f)), DISKS)           ! Start time (= choice of disk) 

  set_task_attributes(file(f), disks, SIZE(f))  ! Resource (disk space) req.  

  set_task_attributes(file(f), 1)    ! Duration (= number of disks used) 

 end-do 

 

! Limit the number of disks used 

 forall(f in FILES) L += getstart(file(f)) 

 diskuse = maximum(L) 

 

! Minimize the total number of disks used 

 if cp_schedule(diskuse) = 0 then 

  writeln("Problem infeasible") 

 end-if 

  

! Solution printing 

 writeln("Number of disks used: ", getsol(diskuse)) 

 forall(d in 1..getsol(diskuse)) do 

  write(d, ":") 

  forall(f in FILES) write( if(getsol(getstart(file(f)))=d , " "+SIZE(f), "")) 

  writeln("  space used: ", 

          sum(f in FILES | getsol(getstart(file(f)))=d) SIZE(f)) 

 end-do 

 cp_show_stats 

end-model 

 

5.4.3 結果 

このモデルを走らせると、Table5.2 に示されている結果が得られます。すなわち、すべてのファイルを

バックアップするには、3 枚のフロッピーディスクが必要です。 

Table 5.2: Distribution of files to disks    

Disk    File sizes (in kb)    Used space (in Mb)  

1  46 87 137 164 253 364 388     1.439 

2  55 62 108 372 406 432    1.435 

3  114 461 851     1.426 

IVE を使って結果をビジュアライズすると、Figure5.3 に示されているようなスクリーンが得られ、そこに

は、資源利用のプロフィールが上部に、そしてタスクのガントチャートが下部に示されます。 

 

 
Figure 5.3: CP dashboard in IVE displaying the solution 

 

5.4.4 オブジェクトをスケジュールしない代替的な定式化 

この問題を、タスクと資源オブジェクトを定義することをする代わりに、事前に定義されたオブジェクト

に分類することなく、タスクの異なった属性を表す標準の有限ドメイン変数のに関して、`cumulative' 

constraintを使って定式化することもできます。単一の`cumulative' constraintは、アーギュメント間の

以下の関係を確立することによって、一つの離散的な資源に関するタスクの集合のスケジューリング

問題を表現します（タスクの5つの特性、すなわち、開始、処理期間、終了、資源の利用、および、サイ

ズに関する決定変数の5つのarrayで、すべて、同一の集合R、および、コンスタント、もしくは、

time-indexed resource capacityによりインデックスされている）。 

jjj enddurationstartRj =+Î" :  

jjj sizedurationuseRj =×Î" :  

( ) ( )[ ]
t

endLBstartUBtRj
j CAPuseTIMESt

jj

£Î" å
ÎÎ ..|

:  

ここで、UB は決定変数の`upper bound'を、LB は`lower bound'意味しています。 

savef はファイル fの保存に使用されるディスクを、そして、usef はファイル fによって使われるスペー

スを示すものとします。scheduling object と同じように、タスクの`start'の特性は、ファイルを保存する

ために選ばれたディスクに対応し、タスクの resource requirement はファイルサイズです。すべてのフ

ァイルを単一のディスクに保存したいので、‘duration’ durf を 1 に固定します。`cumulative' 

constraint の定式化に必要な残りのタスク特性、すなわち、`end' と`size' (ef 、および、sf)は、この問

題の定式化には、必要ではありません。なぜなら、これらの値は、他の 3 つの特性で決定されるから

です。 

 

5.4.5 インプリメンテーション 

以下のMoselモデルは、cumulative constraintを使用する、二番目のモデルバージョンを実行します。 

 

model "D-4 Bin packing (CP)" 

 uses "kalis", "mmsystem" 

 

 setparam("default_lb", 0) 

 

 declarations 

  ND: integer                        ! Number of floppy disks 

  FILES = 1..16                      ! Set of files 

  DISKS: range                       ! Set of disks 

 

  CAP:  integer                      ! Floppy disk size 

  SIZE: array(FILES) of integer      ! Size of files to be saved 

 

  save: array(FILES) of cpvar       ! Disk a file is saved on 

  use: array(FILES) of cpvar        ! Space used by file on disk 

  dur,e,s: array(FILES) of cpvar    ! Auxiliary arrays for 'cumulative' 

  diskuse: cpvar                    ! Number of disks used 

  Strategy: array(FILES) of cpbranching ! Enumeration 

  FSORT: array(FILES) of integer 

 end-declarations 

  

 initializations from 'Data/d4backup.dat' 

  CAP SIZE 

 end-initializations 

 

! Provide a sufficiently large number of disks 

 ND:= ceil((sum(f in FILES) SIZE(f))/CAP) 

 DISKS:= 1..ND 

 finalize(DISKS) 

 

! Limit the number of disks used 

 diskuse = maximum(save) 

 

 forall(f in FILES) do 

  setdomain(save(f), DISKS)         ! Every file onto a single disk 

  use(f) = SIZE(f) 

  dur(f) = 1 

 end-do  

 

! Capacity limit of disks 

 cumulative(save, dur, e, use, s, CAP) 

 

! Definition of search (place largest files first) 

 qsort(SYS_DOWN, SIZE, FSORT)       ! Sort files in decreasing order of size 

 forall(f in FILES) 

  Strategy(f):= assign_var(KALIS_SMALLEST_MIN, KALIS_MIN_TO_MAX, 

{save(FSORT(f))})  

 cp_set_branching(Strategy) 

 

! Minimize the total number of disks used 

 if not cp_minimize(diskuse) then 

  writeln("Problem infeasible") 

 end-if 

  

! Solution printing 

 writeln("Number of disks used: ", getsol(diskuse)) 

 forall(d in 1..getsol(diskuse)) do 

  write(d, ":") 

  forall(f in FILES) write( if(getsol(save(f))=d , " "+SIZE(f), "")) 

  writeln("  space used: ", sum(f in FILES | getsol(save(f))=d) SIZE(f)) 

 end-do 

  

end-model 

このモデルの実行で得られるソリューションは、同一の目的関数の値を持っていますが、ディスクへの

ファイルの配布は、前のそれと、「すべて、まったく同一」ではありません。この問題は、いくつかの異

なった最適解を持っています。特に、ディスクの番号を変えた場合に得られる解がそうです。このよう

な場合、サーチを短くするために、symmetry breaking constraint を追加することはよいことです。

symmetry breaking constraint は、実行可能解の一部を取り除き、サーチスペースのサイズを小さくし

てくれます。例えば、現在の例では、最も大きいファイルを最初のディスクに割り当て、二番目に大き

いファイルを最初の二つのディスクの片方に割り当て、・・・、これを、必要なディスクの数の lower 

bound に到達するまで行います（保存の lower bound estimate は、ファイル・サイズの合計をディスク

容量で割り、端数を切り上げて、次に大きい整数にすることで得られる）。 

5.5 再生可能（renewable）な資源、再生可能ではない（non-renewable）資源 

これまでのセクションで見た`disjunctive--cumulative'、`unary--discrete'の区分のほかに、資源を表

現したり分類したりする方法があります。もう一つの重要な特性は、再生可能（renewable）な資源、再

生可能ではない（non-renewable）資源という概念です。これまでの例は、(機械や作業員などの)再生

可能な資源の例でした。つまり、あるタスクによって使用されていた資源の量は、そのタスクが完成す

るとリリースされて、他のタスクのために利用可能になります。(例えば、お金、原料、中間製品などの)

再生可能ではない（non-renewable）資源の場合は、資源を使用するタスクは、それを消費してしまい

ます。すなわち、利用可能な量の資源は、タスクを処理することで使われた量だけ、減少してしまいま

す。 

また、タスクは、資源を使用するのではなく、ある量の資源を生産するかもしれません。タスクの中に

は、その実行の間に、ある量の資源を生み出すタスク(再生可能な資源)や、生産の結果を、資源のス

トックに加えるタスク(再生可能ではない資源)もあります。 

ここで、以下の問題を、どのように定式化するか、その方法について考えてみましょう。ここに、P1 か

ら P5 までの 5 つのジョブがあり、これらは、ある生産プロセスの二つの段階を表していますが、いま、

これらの 5 つのジョブをスケジュールしたいと考えているとします。P1 と P2 は、(P5 から P3 までの)第

二段階のジョブで必要な中間製品を生産します。それぞれのジョブの最小、最大の処理時間、費用、

そして、第二段階のジョブに関しては、その利益貢献が与えられています。ケースは 2 つあります。す

なわち、モデル A： 第一段階のジョブは、それらの実行の間のあらゆる時点で、所与の量の(例えば､

電気、熱、蒸気などのような)中間製品を生産しますが、これらの中間製品は、即座に、最終段階のジ

ョブで消費されます。モデル B： 中間製品は、第一段階のジョブのアウトプットとして生れ、第二段階

のジョブを開始するためのインプットとして必要になります。 

モデルAの中間製品は、再生可能な資源です、そして、モデルBでは、私たちは再生可能ではない資

源のケースです。 

5.5.1 モデルの定式化 

FIRST = {P1,P2}で、第一段階のジョブの集合、FINAL = {P3,P4,P5}で、第二段階のジョブの集合を、

集合 JOBS ですべてのジョブを示すものとしましょう。すべてのジョブ j に関して、その最小の処理時間

MINDj、および、最大の処理時間 MAXDj が与えられています。RESAMTj は、ジョブにインプットとして必要

な資源量、もしくは、ジョブからアウトプットとして生れる資源量です。また、第一段階のジョブ j の費用

COSTj、第二段階のジョブ j の利益 PROFITj も分かっています。 

モデル A (再生可能な資源) 

再生可能な資源のケースは、以下のモデルのように定式化されます。資源キャパシティが、0 に設定

されることに注意してください。これは、利用可能な資源の量が、第一生産段階のジョブで生産された

もののみが利用可能であることを示しています。 

resource  res 

tasks ( )JOBSjtask j Î  

maximize ( )å
Î

´-
JOBSj

jjj durationtaskCOSTPROFIT .  

res.capacity = 0 

{ }HORIZONendtaskstarttaskJOBSj jj ,...,0,.: ÎÎ"  

{ }jjj MAXDMONDdurationtaskJOBSj ,...,.: ÎÎ"  

jresj RESAMTprovisiontaskFIRSTj =Î" .:  

jresj RESAMTtrequirementaskFINALj =Î" .:  

 

モデル B (再生可能ではない資源) 

モデル A を参考にして、以下のようにモデルＢのケースを定式化します。 

resource  res 

tasks ( )JOBSjtask j Î  

maximize ( )å
Î

´-
JOBSj

jjj durationtaskCOSTPROFIT .  

res.capacity = 0 

{ }HORIZONendtaskstarttaskJOBSj jj ,...,0,.: ÎÎ"  

{ }jjj MAXDMONDdurationtaskJOBSj ,...,.: ÎÎ"  

jresj RESAMTproductiontaskFIRSTj =Î" .:  

jresj RESAMTnconsumptiotaskFIRSTj =Î" .:  

しかし、このモデルは、中間製品の生産がタスクの開始時点で始まるので、上の問題に完全には対

応していません。この問題を直すために、第一段階でのすべてのジョブjに、補助タスクEndjを導入ま

しょう。この補助のジョブは、処理時間 0、もともとのジョブと同一の完成時間を持っており、もともとの

ジョブのところで中間製品を生産します。 

endtaskendtaskFIRSTj jEnd j
..: =Î"  

0.: =Î" durationtaskFIRSTj
jEnd  

0.: =Î" resj productiontaskFIRSTj  

jresEnd RESAMTproductiontaskFIRSTj
j

=Î" .:  

 

5.5.2 インプリメンテーション 

以下の Mosel モデルでケース A を実行します。ここでは、ディフォルトスケジュール作成ソルバー

(function cp_schedule)を使い、オプションの 2 番目のアーギュメントに値 true を入れ、目的関数を

最大にしたいことを知らせます。 

 

model "Renewable resource" 

 uses "kalis", "mmsystem" 

 

 forward procedure solution_found 

 

 declarations    

  FIRST = {'P1','P2'} 

  FINAL = {'P3','P4','P5'} 

  JOBS = FIRST+FINAL 

 

  MIND,MAXD: array(JOBS) of integer  ! Limits on job durations 

  RESAMT: array(JOBS) of integer  ! Resource use/production 

  HORIZON: integer                ! Time horizon 

  PROFIT: array(FINAL) of real    ! Profit from production 

  COST: array(JOBS) of real       ! Cost of production 

  CAP: integer                    ! Available resource quantity 

 

  totalProfit: cpfloatvar 

  task: array(JOBS) of cptask     ! Task objects for jobs 

  intermProd: cpresource          ! Non-renewable resource (intermediate prod.) 

 end-declarations 

   

 initializations from 'Data/renewa.dat' 

  [MIND,MAXD] as 'DUR' RESAMT HORIZON PROFIT COST CAP 

 end-initializations 

 

! Setting up resources 

 set_resource_attributes(intermProd, KALIS_DISCRETE_RESOURCE, CAP) 

 setname(intermProd, "IntP")  

 

! Setting up the tasks 

 forall(j in JOBS) do 

  setname(task(j), j) 

  setduration(task(j), MIND(j), MAXD(j))  

  setdomain(getend(task(j)), 0, HORIZON)  

 end-do  

! Providing tasks 

 forall(j in FIRST) provides(task(j), RESAMT(j), intermProd) 

 

! Requiring tasks 

 forall(j in FINAL) requires(task(j), RESAMT(j), intermProd) 

 

! Objective function: total profit   

 totalProfit = sum(j in FINAL) PROFIT(j)*getduration(task(j)) - 

               sum(j in JOBS) COST(j)*getduration(task(j))                        

 cp_set_solution_callback("solution_found") 

 

! Solve the problem 

 starttime:= gettime                  

 if cp_schedule(totalProfit,true)=0 then 

  exit(1) 

 end-if 

 

! Solution printing 

 writeln("Total profit: ", getsol(totalProfit)) 

 writeln("Job\tStart\tEnd\tDuration") 

 forall(j in JOBS)  

  writeln(j, "\t ", getsol(getstart(task(j))), "\t ", getsol(getend(task(j))), 

          "\t ", getsol(getduration(task(j)))) 

 

 procedure solution_found 

  writeln(gettime-starttime , " sec. Solution found with total profit = ", 

          getsol(totalProfit)) 

  forall(j in JOBS)  

   write(j, ": ", getsol(getstart(task(j))), "-", getsol(getend(task(j))), 

          "(", getsol(getduration(task(j))), "), ") 

  writeln 

 end-procedure 

  

end-model 

ケース B のモデルでは、二つの補助のタスクを加え、集合 ENDFIRST を作り、第一段階のジョブの完

成をマークします。唯一のもう一つの違いは、資源制約条件を定義するタスク特性 produces、

consumes です。 ここでは、モデルの関連部分のみを繰り返します。 

declarations    

  FIRST = {'P1','P2'} 

  ENDFIRST = {'EndP1', 'EndP2'} 

  FINAL = {'P3','P4','P5'} 

  JOBS = FIRST+ENDFIRST+FINAL 

 

  MIND,MAXD: array(JOBS) of integer  ! Limits on job durations 

  RESAMT: array(JOBS) of integer  ! Resource use/production 

  HORIZON: integer                ! Time horizon 

  PROFIT: array(FINAL) of real    ! Profit from production 

  COST: array(JOBS) of real       ! Cost of production 

  CAP: integer                    ! Available resource quantity 

 

  totalProfit: cpfloatvar 

  task: array(JOBS) of cptask     ! Task objects for jobs 

  intermProd: cpresource          ! Non-renewable resource (intermediate prod.) 

 end-declarations 

  

 initializations from 'Data/renewb.dat' 

  [MIND,MAXD] as 'DUR' RESAMT HORIZON PROFIT COST CAP 

 end-initializations 

 

! Setting up resources 

 set_resource_attributes(intermProd, KALIS_DISCRETE_RESOURCE, CAP) 

 setname(intermProd, "IntP")  

 

! Setting up the tasks 

 forall(j in JOBS) do 

  setname(task(j), j) 

  setduration(task(j), MIND(j), MAXD(j))  

  setdomain(getend(task(j)), 0, HORIZON)  

 end-do  

 

! Production tasks 

 forall(j in ENDFIRST) produces(task(j), RESAMT(j), intermProd) 

 forall(j in FIRST) getend(task(j)) = getend(task("End"+j)) 

 

! Consumer tasks 

 forall(j in FINAL) consumes(task(j), RESAMT(j), intermProd) 

 

5.5.3 結果 

ディフォルト・サーチの行動、2 つのモデルの結果はかなり異なっています。Figure5.5 に示されている、

ケース B の目的関数の値 344.9 を持つ最適解は、1 秒以下で証明されます。ケース A の良いソリュー

ションを得るには、標準の PC で、数秒、掛かります。 最適解(Figure5.4)を見つけて、最適性を証明す

るには、数分が必要です。このような芳しくないサーチの行動の主な理由は、目的関数の選択です。

費用ベースの目的関数は、よく propagate せず、したがって、サーチツリーをカットするのにうまく役立

ちません。一般に、スケジュール作成問題での目的関数のより良い選択は、タスク決定変数(start, 

duration, or completion time, particularly the latter)を含む評価基準です。  

Figure 5.4: Solution for case A (resource provision and requirement) 

 

 

Figure 5.5: Solution for case B (resource production and consumption) 

 

5.5.4 オブジェクトをスケジュールしない代替的な定式化 

この問題の定式化は、タスクオブジェクトと資源オブジェクトを定義することの代わりに、標準の有限ド

メイン変数に関する`producer-consumer' constraint を使っても定式化できます。ここで、標準の有限

ドメイン変数は、事前に定義されたオブジェクトに分類されることなく、タスクの異なる、個々の属性を

表します。たった一つの`producer-consumer' constraint により、再生可能ではない資源を生産、もしく

は、消費するタスクの集合のスケジュール作成問題を、アーギュメント間の以下の関係を確立するこ

とによって表現します(タスクに関する 7 つの特性は、start, duration, end, per unit and cumulated 

resource production, per unit and cumulated resource consumption の 7 つで、すべて、同一の集合Ｒ

でインデックスされている）。 

jjj enddurationstartRj =+Î" :  

jjj psizedurationproduceRj =×Î" :  

jjj csizedurationconsumeRj =×Î" :  

( )
( ) ( )[ ]
å

ÎÎ

³-Î"
jj endLBstartUBtRj

jj consumeproduceTIMESt
..|

0:  

ここで、UB は決定変数の`upper bound'、LB は決定変数の`lower bound'を表します。 

 

5.5.5 インプリメンテーション 

以下の Mosel モデルは、producer_consumer constraint を使って、二番目のモデル・バージョンを

実行します。 

model "Non-renewable resource" 

 uses "kalis" 

 

 setparam("DEFAULT_LB", 0) 

 

 declarations    

  FIRST = {'P1','P2'} 

  ENDFIRST = {'EndP1', 'EndP2'} 

  FINAL = {'P3','P4','P5'} 

  JOBS = FIRST+ENDFIRST+FINAL 

  PCJOBS = ENDFIRST+FINAL 

 

  MIND,MAXD: array(JOBS) of integer  ! Limits on job durations 

  RESAMT: array(JOBS) of integer  ! Resource use/production 

  HORIZON: integer                ! Time horizon 

  PROFIT: array(FINAL) of real    ! Profit from production 

  COST: array(JOBS) of real       ! Cost of production 

  CAP: integer                    ! Available resource quantity 

 

  totalProfit: cpfloatvar 

  fstart,fdur,fcomp: array(FIRST) of cpvar! Start, duration & completion of jobs 

  start,dur,comp: array(PCJOBS) of cpvar  ! Start, duration & completion of jobs 

  produce,consume: array(PCJOBS) of cpvar ! Production/consumption per time unit 

  psize,csize: array(PCJOBS) of cpvar     ! Cumulated production/consumption 

 end-declarations 

  

 initializations from 'Data/renewb.dat' 

  [MIND,MAXD] as 'DUR' RESAMT HORIZON PROFIT COST CAP 

 end-initializations 

 

! Setting up the tasks 

 forall(j in PCJOBS) do 

  setname(start(j), j) 

  setdomain(dur(j), MIND(j), MAXD(j))  

  setdomain(comp(j), 0, HORIZON)  

  start(j) + dur(j) = comp(j) 

 end-do  

 forall(j in FIRST) do 

  setname(fstart(j), j) 

  setdomain(fdur(j), MIND(j), MAXD(j))  

  setdomain(fcomp(j), 0, HORIZON)  

  fstart(j) + fdur(j) = fcomp(j) 

 end-do  

 

! Production tasks 

 forall(j in ENDFIRST) do 

  produce(j) = RESAMT(j) 

  consume(j) = 0 

 end-do  

 forall(j in FIRST) fcomp(j) = comp("End"+j) 

 

! Consumer tasks 

 forall(j in FINAL) do 

  consume(j) = RESAMT(j) 

  produce(j) = 0 

 end-do  

 

! Resource constraint 

 producer_consumer(start, comp, dur, produce, psize, consume, csize) 

 

! Objective function: total profit   

 totalProfit = sum(j in FINAL) PROFIT(j)*dur(j) - 

               sum(j in FIRST) COST(j)*fdur(j) 

 

 if not cp_maximize(totalProfit) then 

  exit(1) 

 end-if 

 

 writeln("Total profit: ", getsol(totalProfit)) 

 writeln("Job\tStart\tEnd\tDuration") 

 forall(j in FIRST)  

  writeln(j, "\t ", getsol(fstart(j)), "\t ", getsol(fcomp(j)), 

          "\t ", getsol(fdur(j))) 

 forall(j in PCJOBS)  

  writeln(j, "\t ", getsol(start(j)), "\t ", getsol(comp(j)), 

          "\t ", getsol(dur(j))) 

  

end-model 

このモデルは、前のモデルバージョンと同じソリューションを生成しますが、時間は、少しだけ長く掛か

ります(もっとも、それでも、標準の PC 上で、1 秒以下ですが・・・)。 

5.6 拡張: 段取り時間 

もう一度、Section 3.8 で使ったペイントのバッチの生産計画の問題について考えましょう。2 つのバッ

チの処理の間で、機械をクリーニングする必要があります。 クリーニング(または、セットアップ)は、順

序に依存していて、非対称（asymetric）です。 目的関数は、最も短い生産サイクルを決定することで

す。 

5.6.1 モデルの定式化 

タスクオブジェクト taskj で表わされたすべてのジョブ j ( { }NJJOBSj ,...,1=Î )にたいして、その処理

時間 DURj が与えられています。また、エントリーCLEANjk を持ったクリーニング時間のマトリクス CLEAN

があり、これにより、タスク k がタスク jに続くときのクリーニング時間が与えられています。ジョブを処

理する機械は、unitary capacity の資源 res としてモデル化されます。こうして、ジョブ間の disjunction

が示されます。 

makespan(最後のバッチの完成時間)を最小にする（最も早く完了する）という目的関数で、以下のモ

デルを得ます。 

resource res 

tasks ( )JOBSjtasks j Î  

minimize finish 

=finish maximum JOBSjÎ ( )endtask j .  

Res.capacity 1=  

jj DURdurationtaskJOBSj =Î" .:  

1.: =Î" resj trequirementaskJOBSj  

( ) jkkj CLEANtasktasksetupJOBSkj =Î" .:,  

元の問題の定式化でやっかいなことは、サイクルタイムを最小にしたいと思うということ自体です。す

なわち、最後のジョブの完了＋最後のジョブとそれ以前のジョブの順序で影響されるセットアップ時間

を最小にすることです。ここでのタスクベースのモデルには、ジョブの順序やジョブのランクについての

情報が含まれていないので、生産サイクルの最初に位置するジョブと最後に位置するのジョブのイン

デックス値のために、補助変数 firstjob と lastjob を導入し、そして、最初のタスクと最後のタスク

の間のセットアップ作業の時間のための、もう一つの変数 cleanlf を導入します。以下の制約条件は、

これらの変数とタスクオブジェクトとの関係を表しています。 

JOBSlastjobfirstjob Î,  

lastjobfirstjob ¹  

jlastjobfinishendtaskJOBSj j =Û=Î" .:  

jfirstjobstarttaskJOBSj j =Û=Î" 1.:  

firstjoblastjobCLEANcleanlf ,=  

サイクルタイムを最小にすることは、finish + cleanlf を最小にすることになります。 

 

5.6.2 インプリメンテーション 

以下の Mosel モデルは、上で定式化されたタスクベースのモデルを実行します。タスク間のセットアッ

プ時間は、手順 setsetuptimes でセットされ、2 つのタスクオブジェクトと対応する処理時間の値を示



 120 

しています。 

model "B-5 Paint production (CP)" 

 uses "kalis" 

  

 declarations    

  NJ = 5                             ! Number of paint batches (=jobs) 

  JOBS=1..NJ 

 

  DUR: array(JOBS) of integer        ! Durations of jobs 

  CLEAN: array(JOBS,JOBS) of integer ! Cleaning times between jobs 

 

  task: array(JOBS) of cptask 

  res: cpresource 

 

  firstjob,lastjob,cleanlf,finish: cpvar 

  L: cpvarlist 

  cycleTime: cpvar                   ! Objective variable 

  Strategy: array(range) of cpbranching 

 end-declarations 

 

 initializations from 'Data/b5paint.dat' 

  DUR CLEAN  

 end-initializations 

  

! Setting up the resource (formulation of the disjunction of tasks) 

 set_resource_attributes(res, KALIS_UNARY_RESOURCE, 1) 

   

! Setting up the tasks  

 forall(j in JOBS) getstart(task(j)) >= 1                     ! Start times 

 forall(j in JOBS) set_task_attributes(task(j), DUR(j), res)  ! Dur.s + disj. 

 forall(j,k in JOBS) setsetuptime(task(j), task(k), CLEAN(j,k), CLEAN(k,j))  

                                     ! Cleaning times between batches 

 

! Cleaning time at end of cycle (between last and first jobs) 

 setdomain(firstjob, JOBS); setdomain(lastjob, JOBS) 

 firstjob <> lastjob 

 forall(j in JOBS) equiv(getend(task(j))=getmakespan, lastjob=j) 

 forall(j in JOBS) equiv(getstart(task(j))=1, firstjob=j) 

 cleanlf = element(CLEAN, lastjob, firstjob) 

 

 forall(j in JOBS) L += getend(task(j)) 

 finish = maximum(L) 

 

! Objective: minimize the duration of a production cycle 

 cycleTime = finish - 1 + cleanlf 

   

! Solve the problem 

 if cp_schedule(cycleTime) = 0 then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

 cp_show_stats 

 

! Solution printing 

 declarations 

  SUCC: array(JOBS) of integer 

 end-declarations 

 

 forall(j in JOBS) 

  forall(k in JOBS) 

   if getsol(getstart(task(k))) = getsol(getend(task(j)))+CLEAN(j,k) then 

    SUCC(j):= k 

    break 

   end-if  

 writeln("Minimum cycle time: ", getsol(cycleTime)) 

 writeln("Sequence of batches:\nBatch Start Duration Cleaning") 

 forall(k in JOBS) 

  writeln("  ", k, strfmt(getsol(getstart(task(k))),7), strfmt(DUR((k)),8),  

          strfmt(if(SUCC(k)>0, CLEAN(k,SUCC(k)), getsol(cleanlf)),9)) 

 

end-model 

 

5.6.3 結果 

結果は、Section 3.8 で報告されたものと似ています。 ここで、注意すべきは、このモデル定式化は、

前のモデルバージョンに比較し、サーチノードと実行時間で、それほど効率的でないということです。

Section 3.10 に提示されている'cycle' constraint version に比べると、特に、そうです。しかし、タスクベ

ースの定式化は、前のモデルバージョンでは、問題の定式化が problem-specific であったのに較べ、

よりジェネリックな定式化なので、フィーチャーの追加が容易だという利点もあります。 

によるグラフ表示は以下の通りにです(Figure5.6)。 

 

 

Figure 5.6: Solution graph in IVE 

 

5.7 列挙（Enumeration） 

スケジュールの完成時間を最小にしたい場合、このソルバーによって使われる内蔵のサーチ戦略は、

特に、目的に適っています。他の目的では、内蔵の戦略は良い選択でないかもしれません。特に、モ

デルがスケジュール作成オブジェクトの一部でない決定変数を含んだり(内蔵の戦略は、いつも、最初

に、スケジュール作成オブジェクトを列挙します)、または、目的を最大にしたい場合などです。

Xpress-Kalisでは、したがって、スケジュール作成オブジェクトを含むモデルでも、標準の最適化ファン

クション cp_minimize、cp_maximize が使えるようになっています。これらの最適化ファンクションが

使うディフォルトサーチ戦略は、スケジュール作成を目的とした戦略とは異なっています。さらに、ユー

ザは、以下の例に示されているように、ユーザ自身の、問題固有の列挙を定義することもできます。 

スケジュール作成問題のためのユーザによって定義された列挙な戦略は、変数ベース、タスクベース

の、2 つの異なる形を取ります。前者は、第 4 章のテーマです。そして、本章では、小さい例(Section 

5.7.1)を示すに止めます。後者は、job-shop scheduling example を使い、もう少し、詳しく説明します。 

 

5.7.1 変数ベースの列挙（Variable-based enumeration） 

Section 5.4 の bin packing problem の統計を調べると、ディフォルトのスケジュール作成サーチ戦略を

使うと、最適性を証明するには、数 100 のノードを必要とするのがわかります。問題固有のサーチ戦

略を使うと、そんなに必要ではありません。確かに、greedy-type strategy で、十分なスペースがある

最初のディスクに大きいファイルを、まず、最初に割り当てると、明らかに、より効率的です。 

 

5.7.1.1 cp_minimize を使う 

この問題で行う唯一の意思決定は、タスクの開始時間変数のための値を選んで、ディスクへファイル

の配分することです。以下の Mosel コードは、ファイルサイズが大きい方から小さい方に、降順に並べ、

それぞれに、可能な限り小さいディスク番号を、彼らのスタート時間割り当てる列挙戦略を定義します。

ソーティングサブルーチン qsort が、モデルの最初で、uses ステートメントでロードされるモジュール

mmsystem によって定義されることに注意してください。 

 

 declarations 

  Strategy: array(range) of cpbranching 

  FSORT: array(FILES) of integer 

 end-declarations 

  

 qsort(SYS_DOWN, SIZE, FSORT)       ! Sort files in decreasing order of size 

 forall(f in FILES) 

  Strategy(f):=assign_var(KALIS_SMALLEST_MIN, KALIS_MIN_TO_MAX, 

                          {getstart(file(FSORT(f)))})  

 cp_set_branching(Strategy)  

 

 if not cp_minimize(diskuse) then writeln("Problem infeasible"); end-if 

variable selection criterion (assign_var の最初のアーギュメント)の選択は、ここでは、必ずしも、重要

であるというわけではありません。なぜなら、戦略 Strategyf は、すべて、ただ一つの変数に適用す

るだけで、したがって、選択は、一切、行われないからです。以下のように書いても同等です。 

 declarations 

  Strategy: cpbranching 

  FSORT: array(FILES) of integer 

  LS: cpvarlist 

 end-declarations 

  

 qsort(SYS_DOWN, SIZE, FSORT)       ! Sort files in decreasing order of size 

 forall(f in FILES) 

  LS+=getstart(file(FSORT(f))) 

 Strategy:=assign_var(KALIS_INPUT_ORDER, KALIS_MIN_TO_MAX, LS)  

 cp_set_branching(Strategy)  

 

 if not cp_minimize(diskuse) then writeln("Problem infeasible"); end-if  

このサーチで、最初に見つかったソリューションは 3 枚のフロッピーディスクを使用するソリューション

でした。すなわち、すぐに、最適解を見つられたということです。全体のサーチは 17 ノードで終わり、掛

かった時間は、ディフォルトのスケジュール作成戦略、最小化戦略で必要であった時間の、ほんのご

く一部ですみました。 

 

5.7.1.2 cp_schedule を使う 

ス ケ ジ ュ ー ル 作 成 サ ー チ は 、 前 処 理 (shaving) フ ェ ー ズ と 、 2 つ の 主 な フ ェ ー ズ

(KALIS_INITIAL_SOLUTIONと KALIS_OPTIMAL_SOLUTION)から成ります。ユーザは、対応するフェーズ

選択で、cp_set_schedule_searchcp をコールすることで、_列挙戦略を指定できます。'initial solution'

フェーズは、すばやく良いソリューションを提供することを目的とし、'optimal solution'フェーズは、最適

性を証明することを目的とします。最大ノード数などのサーチ・リミットは、それぞれのフェーズに、別

個に適用されますが、全体的な時間制限(parameter MAX_COMPUTATION_TIME)は、最終フェーズのみ

に適用されます。 

scheduling search の、変数ベースの branching scheme の定義は、前に見た cp_set_strategy を

cp_set_schedule_strategy で置き換え、cp_set_strategy を cp_minimize や cp_maximize による

標準サーチに関して見た方法と、まったく同じ方法で行います。 

cp_set_schedule_branching(KALIS_INITIAL_SOLUTION, Strategy)  

 if cp_schedule(diskuse)=0 then writeln("Problem infeasible"); end-if 

このサーチ戦略で、最適解は、'initial solution で、たったの 8 つのノードの後に見つけられました。そし

て、列挙は、そこに止まりました。なぜなら、前処理フェーズで、最適解の値である 3 枚の lower bound

が証明されたからです。 

NB： スケジュール作成サーチの個々のフェーズからのアウトプット・ログを得るためには、コントロー

ル・パラメタ VERBOSE_LEVEL を 2 に設定してください。すなわち、ソリューションアルゴリズムの始まり

の前に、以下のラインをモデルに追加してください。 

setparam("VERBOSE_LEVEL", 2) 

 

5.7.2 タスクベースの列挙（Task-based enumeration） 

タスクベースの列挙戦略は、列挙されるタスクを選ぶ選択戦略、タスク処理時間のための value 

selection heuristic、および、タスク開始時間の value selection heuristic の定義から成っています。 

問題の典型的な定義を考えて見ましょう。ここに与えられたジョブの集合があり、それぞれのジョブは、

固定された順序で、所与の機械の集合で処理される必要があります。機械は、一度に、1 つのジョブ

を処理します。生産タスクの処理時間(すなわち、機械でのジョブの処理)、および、ジョブ別の機会処

理の順番（6X6 ケース）は Table5.3 に示されています。目的関数は、スケジュールの makespan(最も遅

い完成時間)を最小にすることです。 

Table: 6X6 job-shop instance from FT63 

Job    Machines        Durations 

1 3  1  2  4  6  5   1  3  6  7  3  6   

2 2  3  5  6  1  4   8  5  10  10  10  4   

3 3  4  6  1  2  5   5  4  8  9  1  7   

4 2  1  3  4  5  6   5  5  5  3  8  9   

5 3  2  5  6  1  4   9  3  5  4  3  1   

6 2  4  6  1  5  3   3  3  9  10  4  1  

 

5.7.2.1 モデルの定式化 

JOBS でジョブの集合を、そして、MACH (MACH = {1,...,NM)で機械の集合を意味するとします。すべて

のジョブ j は、一連のタスク taskjm に従って生産されて、ここで、taskjm は、taskj,m+1 がスタートする前

に完了することが必要です。タスク taskjm は、機械 RESjm によって処理され、固定した処理時間 DURjm

を持っています。 

下のモデルは、この job-shop scheduling problem の定式化です。 

resources { }( )NMMACHmresm ,...,1=Î  

tasks ( )MACHmJOBSjtask jm ÎÎ ,  

minimize finish  
=finish maximum JOBSjÎ ( )endtask NMj .,  

1.: =Î" capacityresMACHm m  

{ }MAXTIMEendtaskstarttaskMACHmJOBSj jmjm ,...,0.,.:, ÎÎÎ"  

{ } { }1,.:1,...,1, +=-ÎÎ" mjjm tasksuccessorstaskNMmJOBSj  

jmjm DURdurationtaskMACHmJOBSj =ÎÎ" .:,  

1.:, =ÎÎ"
jmRESjm trequirementaskMACHmJOBSj  

 

5.7.2.2 インプリメンテーション 

以下の Mosel モデルは、job-shop scheduling 問題を実行し、問題を解くためのタスクベースのブラン

チ戦略を定義します。スタート変数の中から、残っているドメインが最も小さいタスクを選択し、この変

数に、最も小さい可能な値から列挙を行います。タスクベースのブランチ戦略は、Xpress-Kalis の中で、

ファンクション task_serialize により定義され、このファンクションは、アーギュメントとして、user task 

selection、処理時間変数とスタート変数のための値の選択戦略、および、適用するタスクの集合を取

ります。このようなタスクベースのブランチ戦略は、まったく自由に、どんな変数ベースのブランチ戦略

にも結合できます。 

 

model "Job shop (CP)" 

 uses "kalis", "mmsystem" 

 

 parameters 

  DATAFILE = "jobshop.dat" 

  NJ = 6                                 ! Number of jobs 

  NM = 6                                 ! Number of resources 

 end-parameters 

 

 forward function select_task(tlist: cptasklist): integer 

 

 declarations 

  JOBS = 1..NJ                           ! Set of jobs 

  MACH = 1..NM                           ! Set of resources 

  RES: array(JOBS,MACH) of integer       ! Resource use of tasks 

  DUR: array(JOBS,MACH) of integer       ! Durations of tasks 

 

  res: array(MACH) of cpresource         ! Resources 

  task: array(JOBS,MACH) of cptask       ! Tasks 

 end-declarations 

 

 initializations from "Data/"+DATAFILE 

  RES DUR 

 end-initializations 

 

 HORIZON:= sum(j in JOBS, m in MACH) DUR(j,m) 

 forall(j in JOBS) getend(task(j,NM)) <= HORIZON  

 

! Setting up the resources (capacity 1)   

 forall(m in MACH)     

  set_resource_attributes(res(m), KALIS_UNARY_RESOURCE, 1) 

! Setting up the tasks (durations, resource used)  

 forall(j in JOBS, m in MACH)    

  set_task_attributes(task(j,m), DUR(j,m), res(RES(j,m))) 

 

! Precedence constraints between the tasks of every job 

 forall (j in JOBS, m in 1..NM-1) 

!  getstart(task(j,m)) + DUR(j,m) <= getstart(task(j,m+1))  

  setsuccessors(task(j,m), {task(j,m+1)}) 

 

! Branching strategy 

 Strategy:=task_serialize("select_task", KALIS_MIN_TO_MAX,  

            KALIS_MIN_TO_MAX, 

            union(j in JOBS, m in MACH | exists(task(j,m))) {task(j,m)}) 

 cp_set_branching(Strategy) 

 

! Solve the problem 

 starttime:= gettime 

 

 if not cp_minimize(getmakespan) then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

  

! Solution printing 

 cp_show_stats 

 write(gettime-starttime, "sec ") 

 writeln("Total completion time: ", getsol(getmakespan)) 

 forall(j in JOBS) do 

  write("Job ", strfmt(j,-2)) 

  forall(m in MACH | exists(task(j,m))) 

   write(strfmt(RES(j,m),3), ":", strfmt(getsol(getstart(task(j,m))),3),   

         "-", strfmt(getsol(getend(task(j,m))),2)) 

  writeln 

 end-do 

  

!******************************************************************* 

! Task selection for branching 

 function select_task(tlist: cptasklist): integer  

  declarations 

   Tset: set of integer 

  end-declarations 

 

 ! Get the number of elements of "tlist" 

  listsize:= getsize(tlist)   

 

 ! Set of uninstantiated tasks 

  forall(i in 1..listsize)  

   if not is_fixed(getstart(gettask(tlist,i))) then  

    Tset+= {i}      

   end-if 

  

  returned:= 0 

   

  ! Get a task with smallest start time domain 

  smin:= min(j in Tset) getsize(getstart(gettask(tlist,j)))  

  forall(j in Tset) 

   if getsize(getstart(gettask(tlist,j))) = smin then   

    returned:=j; break 

   end-if 

 

 end-function 

 

end-model 

 

5.7.2.3 結果 

この問題の最適解には、55 の makespan があります。ディフォルトスケジュール作成戦略との比較で

は、このブランチ戦略は、300 ノードを超えていたもの 105 ノードまで減少させました。しかし、より大き

いインスタンスでは、このブランチ戦略よりも、ディフォルトのスケジュール作成戦略の方が優れてい

つようです。ディフォルトの最小化戦略は、何分かのランタイムでは、この問題のソリューションを見つ

けられませんでした。 

Figure 5.7: Solution graph in IVE 

 

5.7.2.4 代替的なサーチ戦略 

上で見たものに似ていますが、スケジュール作成サーチのためのユーザタスク選択戦略を定義する

こ と も で き ま す 。 上 の モ デ ル に 必 要 な 唯 一 の 変 更 は 、 cp_set_branching を

cp_set_schedule_strategy で置き換え、cp_minimize を cp_schedule で置き換えることだけです。

ユーザのタスク選択ファンクション select_task の定義は変わりません。 

 

! Branching strategy 

 Strategy:=task_serialize("select_task", KALIS_MIN_TO_MAX,  

            KALIS_MIN_TO_MAX, 

            union(j in JOBS, m in MACH | exists(task(j,m))) {task(j,m)}) 

 cp_set_schedule_strategy(KALIS_INITIAL_SOLUTION, Strategy) 

 

! Solve the problem 

 if cp_schedule(getmakespan)=0 then 

  writeln("Problem is infeasible") 

  exit(1) 

 end-if 

 

この戦略は、ユーザタスク選択による標準サーチよりも、列挙を完了するためのノードが、一層、少な

いノードですみます。 

ユーザ定義のタスク選択ファンクション select_task ではなく、下記の、事前に定義されたタスク選択

評価基準の 1 つを使うこともできます。 

KALIS_SMALLEST_EST / KALIS_LARGEST_EST: 

choose task with the smallest/largest lower bound on its start time ('earliest start time'),  

KALIS_SMALLEST_LST / KALIS_LARGEST_LST:   

choose task with the smallest/largest upper bound on its start time ('latest start time'),  

KALIS_SMALLEST_ECT / KALIS_LARGEST_ECT:   

choose task with the smallest/largest lower bound on its completion time ('earliest completion 

time'),  

KALIS_SMALLEST_LCT / KALIS_LARGEST_LCT:   

choose task with the smallest/largest upper bound on its completion time ('latest completion 

time').  

現在の例では、最も良い選択は KALIS_SMALLEST_LCTであることが判明しています( cp_schedule、

および、cp_minimize の両方を使い、約 60 ノードでサーチを終了)。 

 

 

 Strategy:=task_serialize(KALIS_SMALLEST_LCT, KALIS_MIN_TO_MAX,  

            KALIS_MIN_TO_MAX, 

            union(j in JOBS, m in MACH | exists(task(j,m))) {task(j,m)}) 

 

5.7.3 propagation algorithm の選択 

スケジュール作成問題のためのサーチアルゴリズムのパーフォーマンスは、列挙戦略の定義のみに

依存するのではありません。resource constraint のプロパゲーションアルゴリズムの選択も、大き

く影響します。 

プロパゲーション・タイプは、下記のように、手順 set_resources_attributes の最後のオプショナ

ルナアーギュメントでセットされます。 

 set_resource_attributes(res, KALIS_DISCRETE_RESOURCE, CAP, ALG) 

 

ここで、ALG は、propagation algorithm choice です。 

unary resource のためには、KALIS_TASK_INTERVALS、KALIS_DISJUNCTIONS から選択します。前者は、

大きいコンピュータの計算オーバーヘッドというコストを支払うことで、より強い刈り込みを行うので、小

さい問題から、中程度の問題までは、KALIS_DISJUNCTIONS の選択のほうがよいでしょう。前のセク

ションの jobshop 問題の例を取れば、KALIS_DISJUNCTIONS を使用した場合、ディフォルトのスケジュ

ール作成サーチは、KALIS_TASK_INTERVALS よりも約 4 倍速いですが、サーチされるノードの数は、

よりわずかに大きい結果となりました。 

cumulative resources の た め の プ ロ パ ゲ ー シ ョ ン ア ル ゴ リ ズ ム の オ プ シ ョ ン は 、

KALIS_TASK_INTERVALS と KALIS_TIMETABLING で す 。 フ ィ ル タ リ ン グ ア ル ゴ リ ズ ム

KALIS_TASK_INTERVALS は、より強力ですが、比較的遅いので、このことは、難しい、中型の問題のた

めのよい選択ですが、計算オーバーヘッドが問題になる大型の問題には、KALIS_TIMETABLING が向

い て い る で し ょ う 。 例 を 挙 げ れ ば 、 Section 5.4 、 5.7.1 で 議 論 し た binpacking 問 題 で は 、

KALIS_TASK_INTERVALS の 方 が 、 ( デ ィ フ ォ ル ト ス ケ ジ ュ ー ル 作 成 サ ー チ を 使 用 し た ) 

KALIS_TIMETABLING よりも、3 倍速く、問題を解きました。 

Appendix A 

トラブル・シューティング 

· No license found: Kalis for Mosel、licensing system、および、Xpress-Kalis モジュールがイン

ストールされていなければなりません。Dash から受け取ったライセンスファイルを、

Xpress-MP installation directory にコピーし、environment variable XPRESSDIR が、このディ

レクトリをポイントするように設定する必要があります。 

· The Xpress-Kalis module is not found: Mosel のディレクトリ dso に、ファイ

ル kalis.dso がインストールされていないときは、environment variable 

MOSEL_DSO を、このファイルのロケーションで、定義しなければなりません。 



 132 

Appendix B 

Glossary of CP terms 

CP 用語のグロッサリーは、英文をお読み下さい。 

